



Mercurial > hg > index.cgi

view **dod.s** @ **53:bb39e4af7e21** tip

log
graph
tags
bookmarks
branches
changeset
browse

Fix reference to NMI vector

The initializer sets the NMI vector, not the FIRQ vector. It still makes no sense, but there you go.

author William Astle <lost@l-w.ca>

date Sun, 28 Dec 2014 10:48:13 -0700 (2014-12-28)

parents [47ffee6789b1](#)

children

file
latest
diff
comparison
annotate
file log
raw
help

line source

line wrap: on

```

1 ; Dungeons of Daggorath
2 ;
3 ; Original game copyright Â© 1982 by Dyna Micro
4 ;
5 ; The code contained in this file is not the original source code to Dungeons of Daggorath. It was
6 ; constructed by William Astle in 2013 by disassembling the Dungeons of Daggorath ROM.
7 ;
8 ; According to a web page retrieved from http://frodpod.tripod.com/lisence.html on May 24, 2013,
9 ; this endeavour is permitted. In case the web page becomes unavailable, and because it contains what
10 ; I believe to be important credit information, I have reproduced the text of it below:
11 ;
12 ;*****
13 ;* Grant of license to reproduce Dungeons of Daggorath
14 ;*
15 ;* My name is Douglas J. Morgan. I was the president of DynaMicro, Inc. (since dissolved), the company
16 ;* which conceived, created and wrote Dungeons of Daggorath, a best selling Radio Shack Color Computer
17 ;* adventure game.
18 ;*
19 ;* I have examined the contract I signed with Radio Shack for their license of the game. The contract
20 ;* provides that Radio Shack shall have an exclusive license to manufacture and produce the game, but
21 ;* that said exclusive license shall revert to a non-exclusive license should Radio Shack cease to
22 ;* produce and sell the game. To the best of my knowledge, they have not produced the game for many
23 ;* years. Thus, it is my belief that the right to grant a license for the game has reverted to me.
24 ;*
25 ;* I hereby grant a non-exclusive permanent world-wide license to any and all Color Computer site
26 ;* administrators, emulator developers, programmers or any other person or persons who wish to develop,
27 ;* produce, duplicate, emulate, or distribute the game on the sole condition that they exercise every

```

```

28 ;* effort to preserve the game insofar as possible in its original and unaltered form.
29 ;*
30 ;* The game was a labor of love.  Additional credits to Phillip C. Landmeier - who was my partner and
31 ;* who originally conceived the vision of the game and was responsible for the (then) state of the art
32 ;* sounds and realism, to April Landmeier, his wife - the artist who drew all the creatures as well as
33 ;* all the artwork for the manual and game cover, and to Keith Kiyohara - a gifted programmer who helped
34 ;* write the original game and then contributed greatly to compressing a 16K game into 8K so that it
35 ;* could be carried and produced by Radio Shack.
36 ;*
37 ;* The game did very well for us.  I give it to the world with thanks to all who bought it, played it
38 ;* or enjoyed it.
39 ;*
40 ;* There is one existing copy of the original source code.  Anyone willing to pay for the copying of
41 ;* the listing (at Kinko's) and shipment to them, who intends to use it to enhance or improve the emulator
42 ;* versions of the game is welcome to it.
43 ;*
44 ;* Verification of this license grant or requests for the listing can be made by contacting Louis Jordan,
45 ;* Thank you.
46 ;*****
47 ;
48 ; Louis Jordan's email address is given as louisgjordan@yahoo.com in a hyperlink in the above statement.
49 ;
50 ; It is my belief that this endeavor to disassemble Dungeons of Daggorath is in compliance with the above
51 ; license grant. I have done so for my own amusement and for the challenge. I have also done so because
52 ; I failed to elicit a response from Louis Jordan as described in the license grant. I am not surprised
53 ; that I received no reply given that the page above was put online during or prior to 2006.
54
55 ; some utility macros
56 dod                macro noexpand
57                     swi
58                     fcb \1
59                     ifeq \1-$1B
60                     fcb \2
61                     endc
62                     endm
63
64 skip2              macro noexpand
65                     fcb $8c
66                     endm
67
68
69 ; macros for color basic ROM calls
70 romcall            macro noexpand
71                     swi2
72                     fcb \1
73                     endm

```

```
74
75 ; set lighting for render
76 setlighting      macro noexpand
77                  dod S00
78                  endm
79 ; draw a graphic with graphic data at (X)
80 drawgraphic      macro noexpand
81                  dod S01
82                  endm
83 ; render a packed string (immediate data)
84 renderstrimp      macro noexpand
85                  dod S02
86                  endm
87 ; render a packed string from (X)
88 renderstr         macro noexpand
89                  dod S03
90                  endm
91 ; render character in A
92 renderchar        macro noexpand
93                  dod S04
94                  endm
95 ; decode a 5 bit packed string from (X) to stringbuf
96 decodestrsb       macro noexpand
97                  dod S05
98                  endm
99 ; decode a 5 bit packed string from (X) to (U)
100 decodestr         macro noexpand
101                  dod S06
102
103                  endm
103 ; generate an 8 bit random number in A
104 getrandom         macro noexpand
105                  dod S07
106                  endm
107 ; clear graphics screen currently visible; return parameter pointer in U
108 cleargfx1         macro noexpand
109                  dod S08
110                  endm
111 ; clear graphics screen currently used for drawing; return parameter pointer in U
112 cleargfx2         macro noexpand
113                  dod S09
114                  endm
115 ; clear the status line
116 clearstatus       macro noexpand
117                  dod S0A
118                  endm
119 ; clear the command entry area
```

```
120 clearcommand    macro noexpand
121                 dod S0B
122                 endm
123 ; check for death, fainting, or recovery, and calculate how long before next
124 ; damage reduction tick
125 checkdamage      macro noexpand
126                 dod S0C
127                 endm
128 ; update the inventory on the status line
129 updatestatus     macro noexpand
130                 dod S0D
131                 endm
132 ; update dungeon display
133 updatedungeon    macro noexpand
134                 dod S0E
135                 endm
136 ; do a newline, show prompt, and cursor
137 showprompt       macro noexpand
138                 dod S0F
139                 endm
140
141 ; do a delay for about 1.33 seconds
142 delay            macro noexpand
143                 dod S10
144                 endm
145
146 ; set a block of memory (from X to U-1) to $00
147 clearblock       macro noexpand
148                 dod S11
149                 endm
150 ; set a block of memory (from X to U-1) to $ff
151 setblock         macro noexpand
152                 dod S12
153                 endm
154 ; fade in the image at (X) with sound effects at scale 1.0, clear status and command area
155 fadeinclrst      macro noexpand
156                 dod S13
157                 endm
158 ; fade in the image at (X) with sound effects at scale 1.0, clear command area
159 fadein           macro noexpand
160                 dod S14
161                 endm
162 ; fade out image at (X) with sound effects, clear command area
163 fadeout          macro noexpand
164                 dod S15
165                 endm
```

```
166 ; display the PREPARE! screen
167 showprepare      macro noexpand
168                 dod S16
169                 endm
170 ; create object of type in A
171 createobject      macro noexpand
172                 dod S17
173                 endm
174 ; set object specs (object pointer in U)
175 setobjectspecs    macro noexpand
176                 dod S18
177                 endm
178 ; reset display and show dungeon
179 resetdisplay      macro noexpand
180                 dod S19
181                 endm
182 ; generate a level
183 createlevel       macro noexpand
184                 dod S1A
185                 endm
186 ; play a sound number from immediate data at full volume
187 playsoundimm      macro noexpand
188                 dod S1B,\1
189                 endm
190 ; play sound specified in A, volume in B
191 playsound         macro noexpand
192                 dod S1C
193                 endm
194
195 ; ROM call numbers
196 POLCAT            equ 0
197 CSRDON            equ 4
198 BLKIN             equ 6
199 BLKOUT            equ 8
200 WRTLDR            equ 12
201
202 ROMTAB            equ $A000
203
204 BLKTYP            equ $7c
205 BLKLEN            equ $7d
206 CBUFAD            equ $7e
207
208 RESVEC            equ $A027
209
210 ; SWI routines
211 S00               equ 0
212
```

```

212 S01          equ 1
213 S02          equ 2
214 S03          equ 3
215 S04          equ 4
216 S05          equ 5
217 S06          equ 6
218 S07          equ 7
219 S08          equ 8
220 S09          equ 9
221 S0A          equ $0A
222 S0B          equ $0B
223 S0C          equ $0C
224 S0D          equ $0D
225 S0E          equ $0E
226 S0F          equ $0F
227 S10          equ $10
228 S11          equ $11
229 S12          equ $12
230 S13          equ $13
231 S14          equ $14
232 S15          equ $15
233 S16          equ $16
234 S17          equ $17
235 S18          equ $18
236 S19          equ $19
237 S1A          equ $1A
238 S1B          equ $1B
239 S1C          equ $1C

240
241 PIA0          equ $ff00
242 PIA1          equ $ff20
243 SAMREG        equ $ffc0
244 TOPRAM        equ $4000
245 STACK         equ $1000
246
247 ; the direct page
248              org $200
249 zero          rmb 2          ; initialized to $0000
250 V202          rmb 1          ; apparently unused
251 allones       rmb 2          ; initialized to $ffff
252 horizcent     rmb 2          ; center coordinate for scaled graphics (X)
253 vertcent      rmb 2          ; center coordinate for scaled graphics (Y)
254 screenvis     rmb 2          ; pointer to the parameter block of the currently shown screen
255 screendraw     rmb 2          ; pointer to the parameter block of the screen to use for
    drawing
256 demoseqptr    rmb 2          ; pointer to demo game command sequence
    ---

```

```

257 objectfree      rmb 2                ; pointer to next free object data slot
258 linebuffptr     rmb 2                ; line input buffer pointer
259 playerloc       rmb 2                ; current player position in maze
260 carryweight     rmb 2                ; how much weight the player is currently carrying (for
movement cost)
261 ; powerlevel, magicoff, magicdef, physoff, physdef, and damagelevel must remain in the same specific order
262 ; with the same spacing between them in order to match the same structure used by the creature
263 ; data.
264 powerlevel      rmb 2                ; player power
265 magicoff        rmb 1                ; magical attack value (player)
266 magicdef        rmb 1                ; magical defense value (player)
267 physoff         rmb 1                ; physical attack value (player)
268 physdef         rmb 1                ; physical defense value (player)
269 lefthand        rmb 2                ; pointer to object carried in left hand
270 righthand       rmb 2                ; pointer to object carried in right hand
271 damagelevel     rmb 2                ; player damage level
272 facing          rmb 1                ; the direction the player is facing
273 curtorch        rmb 2                ; pointer to currently mounted torch
274 baselight       rmb 2                ; base light level in dungeon
275 nokeyboard      rmb 1                ; set if no keyboard operations should be done during IRQ
276 backpack        rmb 2                ; pointer to first item in backpack
277 creaturefreeze  rmb 1                ; nonzero means creatures are frozen
278 levbgmask       rmb 1                ; the current level background colour mask
279 lightlevel      rmb 1                ; the current light level, $ff means dark
280 lightcount      rmb 1                ; counter between pixels when drawing lines
281 ybeg            rmb 2                ; start Y coord for line drawing
282 xbeg            rmb 2                ; start X coord for line drawing
283 yend            rmb 2                ; end Y coord for line drawing
284 xend            rmb 2                ; end X coord for line drawing
285 xcur            rmb 3                ; current X coordinate when drawing line
286 ycur            rmb 3                ; current Y coordinate when drawing line
287 xpstep          rmb 3                ; difference in X coordinate between pixels when drawing line
288 ypstep          rmb 3                ; difference in Y coordinate between pixels when drawing line
289 pixelcount      rmb 2                ; number of pixels to draw in a line
290 xbstep          rmb 1                ; the offset to add to pointer when moving to new byte (line
drawing)
291 xystep          rmb 1                ; the offset to add to pointer when moving to new row (line
drawing)
292 drawstart       rmb 2                ; start address of drawing area (line drawing)
293 drawend         rmb 2                ; end address of drawing area (line drawing)
294                rmb 4                ; *unused*
295 horizscale      rmb 1                ; horizontal scaling factor for rendering
296 vertscale       rmb 1                ; vertical scaling factor for rendering
297 polyfirst       rmb 1                ; for rendering images - set if this is the first vertex
298 lastunscalex    rmb 2                ; most recent unscaled X coordinate for rendering
299 lastunscaley    rmb 2                ; most recent unscaled Y coordinate for rendering

```

```

300 soundseqseed      rmb 2          ; sound: sequence generator seed
301                  rmb 1          ; *unused*
302 sndtemp           rmb 1          ; sound: temporary storage for dac value
303                  rmb 1          ; *unused*
304 sndampmult         rmb 2          ; sound: amplitude multiplier for volume slides (MSB is used)
305 sndampstep         rmb 2          ; sound: amplitude step for volume slides
306 soundrepeat       rmb 1          ; sound: repeat counter
307                  rmb 1          ; *unused*
308 soundvol           rmb 1          ; sound: volume multiplier for sound playing
309 soundrepeat2       rmb 1          ; sound: repeat counter for scorpion, wraith, and viper sounds
310 sndlowtonedel     rmb 2          ; sound: low tone wave delay for dual tone generator
311 sndhitonedel      rmb 2          ; sound: high tone wave delay for dual tone generator
312                  rmb 4          ; *unused*
313 randomseed        rmb 3          ; random number generator seed value
314 effectivelight     rmb 1          ; effective light level in dungeon
315 effectivemlight    rmb 1          ; effective magical light level in dungeon (also used for
    alternativelightschemes)
316 savedefflight     rmb 1          ; effective light level saved during fainting
317                  rmb 2          ; *unused*
318 movehalf          rmb 1          ; set if rendering half step for MOVE
319 movebackhalf      rmb 1          ; set if rendering half step for MOVE BACK
320 rendermagic       rmb 1          ; set if rendering should use magical illumination
321                  rmb 1          ; *unused*
322 waitnewgame       rmb 1          ; set if waiting for keypress to start new game
323 kwmatch           rmb 1          ; for tracking if a match is found when looking up a keyword
324 kwcount           rmb 1          ; counter during keyword list lookup
325                  rmb 1          ; *unused*
326 kwexact           rmb 1          ; set if keyword lookup matched exactly

327 temploc           rmb 2          ; working coordinates during various processing steps
328 genpathlen        rmb 1          ; number of steps to dig during maze generation
329                  rmb 2          ; *unused*
330 currentlevel       rmb 1          ; currently playing dungeon level
331 creaturecntptr     rmb 2          ; pointer to creature count table for the current level
332                  rmb 2          ; *unused*
333 holetabptr         rmb 2          ; pointer to the hole table for this level
334 gencurcoord        rmb 2          ; current coordinates when generating maze
335 curdir            rmb 1          ; current direction we're processing
336 renderdist        rmb 1          ; distance from "camera" for rendering
337 objectcount        rmb 1          ; number of objects of specific type to create when
    initializing
338 objectlevel        rmb 1          ; starting level (minimum) of object being created
339 parseobjtype       rmb 1          ; the type of object parsed from command
340 parseobjtypegen     rmb 1          ; the generic type of the object parsed
341 parsegenobj        rmb 1          ; nonzero if a generic object was parsed
342 objiterstart       rmb 1          ; for iterating over object list, zero means start, nonzero
    means underway

```



```

343 objiterptr      rmb 2      ; current pointer during object list iteration
344 showseer        rmb 1      ; nonzero means to show creatures when displaying a scroll
345 clockctrs       rmb 1      ; 60Hz tick (triggers 10 times per second)
346                 rmb 1      ; 10Hz tick (triggers once per second)
347                 rmb 1      ; 1Hz tick (triggers once per minute)
348                 rmb 1      ; 1 minute tick (triggers once per hour)
349                 rmb 1      ; 1 hour tick (triggers once per day)
350                 rmb 1      ; 1 day tick (overflows once every 256 days)
351 disablesched    rmb 1      ; set to nonzero to disable timer handling
352 dofadesound     rmb 1      ; nonzero if we're doing the fade sound effect thing
353 fadesoundval    rmb 1      ; the DAC value to use for the fade sound (complemented every
    tick)
354 enablefadesound rmb 1      ; nonzero means fading will be done with the sound effect
355 schedlists      rmb 2      ; notional "not active" queue?
356                 rmb 2      ; 60Hz tick event list
357                 rmb 2      ; 10Hz tick event list
358                 rmb 2      ; 1Hz tick event list
359                 rmb 2      ; 1 minute tick event list
360                 rmb 2      ; 1 hour tick event list
361                 rmb 2      ; the "ready" list (also 1 day tick event list)
362 hidestatus      rmb 1      ; nonzero will cause the command processor to clear and reset
    on input (status line hidden)
363 heartctr        rmb 1      ; number of ticks until next heart beat
364 heartticks      rmb 1      ; number of ticks between heart beat
365 heartstate      rmb 1      ; zero = contracted heart, ff = expanded heart
366 enableheart     rmb 1      ; nonzero means heartbeat is running
367 displayptr      rmb 2      ; pointer to routine to display the main dungeon area
368 pageswap        rmb 1      ; nonzero means we're ready to swap graphics screens during
    IRQ
369 dungeonchg      rmb 1      ; nonzero if the dungeon display should be updated
370 columnctr       rmb 1      ; column counter/tracker for displaying inventory list
371 textother       rmb 1      ; nonzero means nonstandard text location
372 loadsaveflag    rmb 1      ; load/save flag - <0 = ZLOAD, >0 = ZSAVE, 0 = regular init
373 schedtabfree    rmb 2      ; pointer to next free entry in the scheduling table
374 readylistchg    rmb 1      ; nonzero if the ready list processing should be restarted
375 keybufread      rmb 1      ; keyboard buffer read offset
376 keybufwrite     rmb 1      ; keyboard buffer write offset
377                 rmb 3      ; *unused*
378 accum0          rmb 3      ; temporary 24 bit accumulator
379 accum1          rmb 3      ; temporary 24 bit accumulator
380 accum2          rmb 1      ; temporary 8 bit accumulator
381                 rmb 9      ; *unused*
382 keybuf          rmb 32     ; keyboard ring buffer
383 linebuff        rmb 32     ; line input buffer
384 linebuffend     equ *      ; end of line input buffer
385                 rmb 2      ; *unused*
386 wordbuff        rmb 32     ; buffer used for parsing words from the line

```

```

387 wordbuffend    equ *                ; end of word buffer
388                rmb 2                ; *unused*
389 stringbuf      rmb 34                ; temporary buffer used for decoding immediate packed strings
390 fontbuf        rmb 10                ; temporary buffer used for decoding font data
391 cmddecodebuff   rmb 31                ; buffer used for decoding commands
392 ; These are descriptors used for controlling the rendering engine. Each one is 8 bytes long and is defined as
    follows:
393 ; 0-1   start address of text area (memory)
394 ; 2-3   number of character cells in text area
395 ; 4-5   current cursor position within text area
396 ; 6     background colour mask for the text area
397 ; 7     whether to render text on the secondary screen
398 ;
399 ; There are three text areas:
400 ; infoarea    this is text rendered in the main dungeon display area
401 ; statusarea  this is text rendered on the status line
402 ; commandarea this is text rendered in the command area
403 infoarea      rmb 2                ; screen start address of info text area
404                rmb 2                ; number of character cells in info text area
405                rmb 2                ; current cursor position in info text area
406                rmb 1                ; background colour mask for info text area
407                rmb 1                ; nonzero if info text area should not be rendered on
    secondary screen
408 statusarea    rmb 2                ; screen start address of status line area
409                rmb 2                ; number of character cells in status line area
410                rmb 2                ; cursor position in status line area
411                rmb 1                ; background colour mask for status line area
412                rmb 1                ; nonzero if status line text should not be rendered on
    secondary screen
413 commandarea   rmb 2                ; start offset of the command entry area
414                rmb 2                ; number of character cells in command entry area
415                rmb 2                ; current cursor position in entry area
416                rmb 1                ; background colour mask of background area
417                rmb 1                ; nonzero if main text should not be rendered on secondary
    screen
418 creaturecounts rmb 12                ; creature counts for level 1
419                rmb 12                ; creature counts for level 2
420                rmb 12                ; creature counts for level 3
421                rmb 12                ; creature counts for level 4
422                rmb 12                ; creature counts for level 5
423 creaturetab    rmb 32*17            ; the creatures currently active on this level
424 mazedata       rmb $400             ; the actual room data for the current level
425 neighbourbuff  rmb 9                ; buffer for calculating neighbors in maze generation
426 schedtab      rmb $10a             ; scheduler entries
427 emptyhand      rmb 14                ; "object" information for empty hand

```

```

428 objecttab      rmb 72*14                ; the object data table (room for 72 entries)
429 datatop        equ *
430
431
432                org $c000
433 START          ldu #dodemo                ; point to the demo setup routine
434                bra LC008                  ; go handle the game
435 LC005          ldu #dogame                ; point to the real game setup routine
436 LC008          lds #STACK                ; put the stack somewhere safe
437                ldx #PIA0                  ; point at PIA0
438                ldd #$34fa                ; initializers for the PIA
439                sta 3,x                    ; set data mode, interrupts disabled (side B)
440                sta 1,x                    ; set data mode, interrupts disabled (side A)
441                ldx #PIA1                  ; point at PIA1
442                sta 1,x                    ; set data mode, interrupts disabled (side A)
443                clr 3,x                    ; set direction mode for side B
444                stb 2,x                    ; set VDG and single bit sound to output
445                lda #$3c                  ; flags for data mode, no interrupts, sound enabled
446                sta 3,x                    ; set side B for data mode
447                ldd #$2046                ; SAM value for "pmode 4" graphics, screen at $1000
448                jsr setSAM                ; go set the SAM register
449                lda #$f8                  ; value for "pmode 4", color set 1
450                sta 2,x                    ; set VDG mode
451                ldx #zero                  ; point to start of variables
452 LC030          clr ,x+                    ; clear a byte
453                cmpx #TOPRAM              ; are we at the top of memory?
454                blo LC030                  ; brif not
455                stu ,--s                  ; set return address to the game initialization routine
456
457                lda #zero/256              ; point to MSB of direct page
458                tfr a,dp                  ; set DP appropriately
459                setdp zero/256            ; tell the assembler about DP
460 LC041          ldy #LD7E8                ; point to variable initialization table
461                lda ,y+                    ; fetch number of bytes in this initializer
462                beq LC086                  ; brif zero - we're done
463                ldx ,y++                  ; get address to copy bytes to
464                bsr LC04B                  ; go copy bytes
465 LC04B          bra LC041                  ; go handle another initializer
466                ldb ,y+                    ; fetch source byte
467                stb ,x+                    ; stow at destination
468                deca                      ; are we done yet?
469                bne LC04B                  ; brif not
470                rts                      ; return to caller
471 LC053          pshs cc,a,b,x,y,u          ; save registers and interrupt status
472                orcc #$10                 ; disable IRQ
473                ldx #schedlists           ; point to start of variables to clear for level creation
474                clr ,x+                   ; clear a byte

```

```

474      cmpx #hidestatus                                ; are we finished clearing things?
475      blo LC05A                                       ; brif not
476      ldx #schedtab                                  ; start of the scheduling list
477      stx schedtabfree                               ; mark that as the end of it
478 LC066      clr ,x+                                   ; clear more data
479      cmpx #emptyhand                                ; end of area to clear?
480      blo LC066                                       ; brif not
481      ldy #LD7DC                                      ; point to list of entries to schedule
482      dec readylistchg                               ; mark ready list as modified
483      ldd #12                                         ; set tick count to 0 and list to "ready" list
484 LC076      ldx ,y++                                  ; get routine address
485      beq LC084                                       ; brif end of list
486      jsr LC25C                                       ; create scheduler entry
487      stx 3,u                                         ; save routine address in scheduling entry
488      jsr LC21D                                       ; add to "ready" list
489      bra LC076                                       ; go look for another routine
490 LC084      puls cc,a,b,x,y,u,pc
491 LC086      bsr LC053                                ; initialize new level data
492      ldu #LDA91                                      ; point to objects to create for game
493      clra                                           ; initialize object type
494 LC08C      ldb ,u                                   ; get object info
495      andb #$0f                                       ; get low nibble
496      stb objectcount                               ; save number to create (total)
497      ldb ,u+                                         ; get object info again
498      lsrb                                           ; get high nibble (object starting level)
499      lsrb
500      lsrb
501      lsrb
502      stb objectlevel                               ; save starting level
503 LC09A      createobject                             ; create an object
504      dec 5,x                                         ; mark as equipped or carried
505      incb                                           ; bump level
506      cmpb #5                                         ; was it level 5?
507      ble LC0A5                                       ; brif yes
508      ldb objectlevel                               ; get level back
509 LC0A5      dec objectcount                           ; created enough objects?
510      bne LC09A                                       ; brif not
511      inca                                           ; move to next object type
512      cmpu #LDA91+18                                 ; end of table?
513      blo LC08C                                       ; brif not - go to next entry
514      ldu #statusarea                               ; point to text parameters for status area
515      dec textother                                  ; indicate nonstandard text area
516      clearstatus                                    ; blank status line (where we'll put the copyright notice)
517      renderstrimpp                                  ; display copyright message
518      fcb $f8,$df,$0c,$c9                           ; packed string "COPYRIGHT DYNA MICRO MCMLXXXII"
519      fcb $27,$45,$00,$02

```

```

520          fcb $65,$c1,$03,$52
521          fcb $39,$3c,$00,$68
522          fcb $da,$cc,$63,$09
523          fcb $48
524          clr textother          ; reset text rendering to standard mode
525          rts                    ; transfer control to correct game loop
526  dodemo    dec waitnewgame      ; flag demo game
527          bsr enablepiairq       ; set up interrupts and sound
528          ldx #img_wizard        ; point to wizard image
529          dec enablefadesound    ; enable fade sound effect
530          fadein                 ; fade the wizard in
531          renderstrimmp          ; display <CR>"I DARE YE ENTER..."<CR>
532          fcb $9f,$d2,$02,$06    ; packed "\rI DARE YE ENTER...\r" string
533          fcb $45,$06,$4a,$02
534          fcb $ba,$85,$97,$bd
535          fcb $ef,$80
536          renderstrimmp          ; display "...THE DUNGEONS OF DAGGORATH!!!"
537          fcb $f7,$bd,$ea,$20    ; packed "...THE DUNGEONS OF DAGGORATH!!!" string
538          fcb $a0,$25,$5c,$72
539          fcb $bd,$d3,$03,$cc
540          fcb $02,$04,$e7,$7c
541          fcb $83,$44,$6f,$7b
542          delay                  ; * wait for about 2.6 seconds
543          delay                  ; *
544          fadeout                ; fade the wizard out
545          cleargfx2              ; clear second graphics screen
546          dec pageswap           ; flag graphics swap ready
547          sync                   ; wait for swap to happen
548          lda #2                 ; create maze for level 3
549          ldu #startobjdemo      ; point to demo game object list
550          bra LC131              ; go start demo running
551  enablepiairq  ldd #$343c        ; set up initializers for PIAs
552          sta PIA1+1             ; set data mode, no interrupts, cassette off
553          stb PIA1+3             ; set data mode, no interrupts, sound on
554          inca                   ; adjust to enable interrupt
555          sta PIA0+3             ; set data mode, enable VSYNC, clear MSB of analog MUX
556          cwai #$ef             ; enable IRQ and wait for one
557          rts                    ; return to caller
558  dogame        bsr enablepiairq  ; set up interrupts and sound
559          ldd #$100b             ; maze position (11,16)
560          std playerloc          ; set start position there
561          clr powerlevel         ; reset power level to new game level (keeps LSB of default
                                value)
562          clra                   ; create maze for level 1
563          ldu #startobj          ; point to new game backpack object list
564  LC131        showprepare       ; show the PREPARE! screen

```

```

565         createlevel                ; create maze
566         ldy #backpack              ; point to backpack list head pointer
567 LC139    lda ,u+                   ; get object to create
568         bmi LC14F                  ; brif end of list
569         createobject               ; create requested object
570         inc 5,x                    ; mark object as in inventory or equipped
571         exg x,u                    ; swap object pointer and list pointer
572         setobjectspecs              ; set the specs properly (as in fully revealed)
573         exg x,u                    ; swap pointers back
574         clr 11,x                   ; mark object revealed
575         stx ,y                     ; save new object in backpack list
576         tfr x,y                    ; move list pointer to object just created
577         bra LC139                  ; go look for another object to create
578 LC14F    tst waitnewgame           ; are we doing a demo game?
579         beq LC166                  ; brif not
580         dec disablesched            ; disable scheduling events
581         ldx #displayscroll         ; set to scroll display
582         stx displayptr             ;
583         dec showseer               ; set to show creatures on map
584         updatedungeon              ; show the dungeon map
585         delay                      ; delay for about 2.5 seconds
586         delay                      ;
587         clr disablesched            ; enable scheduling events
588         sync                       ; wait a couple of ticks
589         sync                       ;
590 LC166    resetdisplay              ; clear command and status areas and show the dungeon
591         showprompt                 ; show command prompt
592         jmp LC1F5                  ; go to main loop
593 LC16D    stx CBUFAD                ; set address to read to
594         romcall BLKIN              ; read a block
595         tsta                       ; is it the end of the file?
596         lbne RESVEC                ; brif so - premature end, fail with a reset
597         ldb BLKTYP                 ; get type of block
598         rts
599 disablepiairq ldu #PIA0            ; point to PIA0
600         ldd #$343c                 ; set up initializers for PIA
601         sta 3,u                    ; disable VSYNC interrupt, clear analogue mux msb
602         sta PIA1+3                 ; disable interrupts on PIA1, cassette off
603         stb PIA1+1                 ; disable interrupts on PIA1, sound on
604         rts
605 busywait  ldx zero                 ; get long delay constant
606 busywait000 leax -1,x              ; have we reached 0?
607         bne busywait000            ; brif not
608         rts
609 LC192    bsr disablepiairq         ; set up PIA for cassette I/O
610         bsr busywait               ; delay for a bit

```

```

611      bsr busywait
612      romcall WRTLDR                ; write a file header
613      romcall BLKOUT
614      bsr busywait                ; delay for a bit
615      romcall WRTLDR                ; write a leader for data area
616      ldx #zero                    ; point to start of game state
617  LC1A6      ldd #$0180              ; set block type to data, size to 128 bytes
618      std BLKTYP
619      stx CBUFAD                    ; set start of buffer to write
620      romcall BLKOUT                ; write a data block
621      cmpx #datatop                ; have we reached end of state?
622      blo LC1A6                    ; brif not
623      stu BLKTYP                    ; write trailing block
624      romcall BLKOUT
625      bsr busywait                ; delay for a bit
626      bra LC1EC                    ; go init things and restart main loop
627  LC1C1      bsr disablepiairq      ; set up PIA for cassette I/O
628      romcall CSRDON                ; start tape
629  LC1C6      ldu screendraw          ; point to drawing area
630      ldx ,u                        ; get pointer to screen data - use as a read buffer
631      bsr LC16D                    ; read a block
632      bne LC1C6                    ; brif data block
633      ldx ,u                        ; get buffer pointer
634      ldu #wordbuff                ; point to requested file name
635      ldb #8                        ; 8 characters in file name
636  LC1D5      lda ,x+                ; does character match?
637      cmpa ,u+
638      bne LC1C1                    ; brif not - look for another header
639      decb                          ; end of file name?
640      bne LC1D5                    ; brif not - check another
641      romcall CSRDON                ; start tape
642      ldx #zero                    ; point to game state area
643  LC1E4      bsr LC16D                ; read a block
644      bpl LC1E4                    ; brif it was still a data block
645      lds #STACK                    ; reset stack pointer
646  LC1EC      jsr enablepiairq        ; make sure PIAs are set right
647      clr loadsaveflag              ; flag regular operations
648      resetdisplay                  ; clear command and status areas, update appropriately
649      showprompt                    ; show command prompt
650  LC1F5      ldu #schedlists+12      ; point to ready list head
651      clr readylistchg              ; mark ready list restart not needed
652  LC1FA      tfr u,y                ; save ready list pointer
653  LC1FC      tst loadsaveflag        ; are we loading or saving?
654      bgt LC192                    ; brif saving
655      bmi LC1C1                    ; brif loading
656      ldu ,u                        ; are we at the end of the ready list?

```

```

657          beq LC1F5                      ; brif so
658          pshs y,u                      ; save registers
659          jsr [3,u]                     ; call the registered routine
660          puls y,u                      ; restore registers
661          tst readylistchg              ; do we need to restart the ready list processing?
662          bne LC1F5                      ; brif so
663          cmpb #12                      ; are we leaving the routine in the ready list?
664          beq LC1FA                      ; brif so - check next entry
665          bsr LC238                      ; remove this event from the ready list
666          bsr LC21D                      ; reschedule in requested queue for requested number of ticks
667          tfr y,u                      ; move current pointer to previous pointer
668          bra LC1FC                      ; go check next entry
669 LC21D      pshs cc,a,b,x                ; save flags and registers
670          orcc #$10                     ; disable IRQ
671          sta 2,u                      ; reset tick count
672          ldx #schedlists               ; pointer to routine base
673          abx                          ; add offset
674          clra                          ; NULL pointer
675          clrb
676          std ,u                      ; mark this timer unused
677 LC22B      cmpd ,x                    ; are we at a NULL pointer (end of list)?
678          beq LC234                      ; brif so
679          ldx ,x                      ; point to next entry
680          bra LC22B                      ; go check if we're at the end yet
681 LC234      stu ,x                    ; move this timer entry to the end of the list
682          puls cc,a,b,x,pc              ; restore registers, interrupt status, and return
683 LC238      pshs cc,x                  ; save interrupt status and registers
684          orcc #$10                     ; disable IRQ
685          ldx ,u                      ; get next pointer from this entry
686          stx ,y                      ; save it in previous entry
687          puls cc,x,pc                  ; restore interrupts, registers, and return
688 LC242      pshs b,x,y,u                ; save registers
689          tst disablesched              ; are we handling timers?
690          bne LC25A                      ; brif not
691 LC248      tfr u,y                    ; save timer pointer
692          ldu ,u                      ; get pointer to timer info
693          beq LC25A                      ; brif nothing doing for timer
694          dec 2,u                      ; has this timer record expired?
695          bne LC248                      ; brif not - check next one
696          bsr LC238                      ; go process timer record
697          ldb #12                      ; offset to "ready" list
698          bsr LC21D                      ; go move event entry to "ready" list
699          bra LC248                      ; go process next timer record
700 LC25A      puls b,x,y,u,pc              ; restore registers and return
701 LC25C      pshs x                      ; save registers
702          ldu schedtabfree              ; get open slot for scheduling

```



```

703          leax 7,u                ; point to next open slot
704          stx schedtabfree        ; save new open slot for scheduling
705          puls x,pc               ; restore registers and return
706 ; Set the SAM video mode and display offset register to the value in D. Starting at the lsb of
707 ; D, the SAM bits are programmed from FFC0 upward. This sets bits 9-0 of the SAM register
708 ; to match the value in D.
709 setSAM      pshs x,b,a            ; save registers
710          ldx #SAMREG              ; point to SAM register
711 setSAM000    lsra                 ; * shift the bit value to set to carry
712          rorb                    ; *
713          bcs setSAM001            ; brif bit set
714          sta ,x                  ; clear the bit
715          skip2                   ; skip next instruction
716 setSAM001    sta 1,x              ; set the bit
717          leax 2,x                ; move to next SAM register bit
718          cmpx #SAMREG+$14         ; are we at the end of the register?
719          blo setSAM000            ; brif not
720          puls a,b,x,pc           ; restore registers and return
721 ; IRQ service routine
722 irqsvc      ldx #PIA1             ; point to PIA1
723          lda -29,x                ; get interrupt status
724          lbpl LC320               ; brif not VSYCN
725          lda #zero/256            ; point to direct page MSB
726          tfr a,dp                ; make sure DP is set correctly
727          tst pageswap             ; do we have a screen swap to do?
728          beq LC29D               ; brif not
729          ldd screenvis            ; get currently visible screen pointer
730          ldu screendraw           ; get newly drawn screen pointer
731          std screendraw           ; save current screen as screen to draw
732          stu screenvis            ; save drawn screen as current
733          ldd 4,u                  ; get the SAM value for the new screen
734          bsr setSAM               ; go program the SAM
735          clr pageswap             ; flag no swap needed
736 LC29D      tst dofadesound        ; are we doing the "fade buzz" thing?
737          beq LC2A9               ; brif not
738          com fadesoundval         ; invert the bits of of the value
739          lda fadesoundval         ; fetch new value
740          lsla                     ; * align to DAC bits
741          lsla                     ; *
742          sta ,x                  ; set DAC output
743 LC2A9      tst enableheart        ; is the heart beating?
744          beq LC2DC               ; brif not
745          dec heartctr             ; count down ticks till beat
746          bne LC2DC               ; brif not expired
747          lda heartticks           ; fetch ticks till next beat
748          sta heartctr             ; reset counter

```

```

749         ldb 2,x                ; fetch single bit sound register
750         eorb #2                ; invert single bit sound output
751         stb 2,x                ; set new sound output
752         tst hidestatus         ; is the status line shown?
753         beq LC2DC              ; brif not - don't update heart
754         ldu #statusarea       ; point to status line text area descriptor
755         ldx 4,u                ; fetch current text position
756         ldd #15                ; position for centring heart
757         std 4,u                ; save output position
758         lda #$20               ; code for contracted heart (left)
759         com heartstate         ; invert heart state
760         beq LC2D1              ; brif contracted
761         lda #$22               ; code for expanded heart (left)
762 LC2D1      jsr LCA17            ; render left half of heart
763         inc 5,u                ; bump character position
764         inca                   ; code for right half of heart
765         jsr LCA17              ; render right half of heart
766         stx 4,u                ; save original text position
767 LC2DC      ldu #schedlists+2   ; point to timer lists
768         jsr LC242              ; check if any records expired at 60Hz
769         ldx #clockctrs         ; point to timer records
770         ldy #LC324             ; point to timer max values
771 LC2E9      inc ,x              ; bump timer value
772         cmpx #clockctrs+5      ; end of timer record?
773         beq LC2FF              ; brif so
774         lda ,x                 ; fetch new value
775         cmpa ,y+               ; has timer maxed out?
776         blt LC2FF              ; brif not
777         clr ,x+                ; reset timer record
778         leau 2,u               ; move to next timer list
779         jsr LC242              ; see if any events have expired
780         bra LC2E9              ; go handle next timer
781 LC2FF      tst nokeyboard      ; are we polling the keyboard?
782         bne LC320              ; brif not
783         tst waitnewgame        ; are we running a demo/waiting for keypress for game start?
784         beq LC318              ; brif not
785         clr PIA0+2             ; strobe all keyboard columns
786         lda PIA0               ; fetch row data
787         anda #$7f              ; mask off comparator input
788         cmpa #$7f              ; did we have any keys down?
789         beq LC320              ; brif not
790         ldx #LC005             ; pointer to game start routine
791         stx 10,s               ; set return to game start routine
792 LC318      romcall POLCAT       ; poll the keyboard
793         tsta                   ; was a key down?
794         beq LC320              ; brif not

```

```

795          bsr writekeybuf          ; go process keyboard input
796 LC320      lda PIA0+2              ; clear interrupt status
797          rti
798 ; These are the rollover points for the timers. Each timer only ticks if the previous
799 ; timer has rolled over.
800 LC324      fcb 6                    ; tick 10 times per second
801          fcb 10                    ; tick 1 time per second
802          fcb 60                    ; tick 1 time per minute
803          fcb 60                    ; tick 1 time per hour
804          fcb 24                    ; tick 1 time per day
805 readkeybuf  pshs cc,b,x            ; save registers and interrupt status
806          orcc #$10                 ; disable IRQ
807          clra                      ; flag no key pressed
808          ldx #keybuf               ; point to keyboard ring buffer
809          ldb keybufread             ; get buffer read offset
810          cmpb keybufwrite           ; same as buffer write offset?
811          beq readkeybuf000          ; brif so - no key available
812          lda b,x                   ; fetch key from buffer
813          incb                      ; bump buffer pointer
814          andb #$1f                 ; wrap around if needed
815          stb keybufread            ; save new buffer read offset
816 readkeybuf000 puls cc,b,x,pc        ; restore registers and interrupts
817 ; Add a keypress to the keyboard buffer. NOTE: this does not check for buffer overflow
818 ; which means when the buffer gets full, it just rolls over and overwrites the previous
819 ; data.
820 writekeybuf  pshs cc,b,x            ; save registers and interrupt status
821          orcc #$10                 ; disable IRQ
822          ldx #keybuf               ; point to keyboard ring buffer
823          ldb keybufwrite            ; get buffer write offset
824          sta b,x                   ; stash new key
825          incb                      ; bump buffer pointer
826          andb #$1f                 ; wrap around if needed
827          stb keybufwrite            ; save new buffer write offset
828          puls cc,b,x,pc            ; restore registers and interrupts
829 ; SWI handler
830 swisvc      andcc #$ef              ; re-enable IRQ - SWI disables it
831          ldx 10,s                  ; get return address
832          lda ,x+                    ; get operation code
833          stx 10,s                  ; save new return address
834          ldx #LC384                 ; point to first SWI routine
835          ldu #LC995                 ; point to routine offset table
836 LC360      ldb ,u+                  ; get length of previous routine
837          abx                        ; add to routine pointer
838          deca                       ; are we at the right routine?
839          bpl LC360                  ; brif not
840          stx ,--s                  ; save routine address

```

```

841                ldd 3,s                ; restore D register
842                ldx 6,s                ; restore X register
843                ldu 10,s               ; restore U register
844                jsr [,s++]             ; call the routine
845                rti                   ; return to caller
846 ; SWI2 handler
847 swi2svc         clrb                  ;* restore direct page for ROM call
848                tfr b,dp              ;*
849                ldu 10,s               ; get return address
850                ldb ,u+                ; get ROM routine offset
851                stu 10,s               ; save new return address
852                ldu #ROMTAB            ; point to ROM vector table
853                jsr [b,u]              ; call the ROM routine
854                sta 1,s                ;* save return values
855                stx 4,s                ;*
856                rti                   ; return to caller
857 ; SWI 0 routine
858 ; Calculate base light level in dungeon. Note that "magic lighting" is also used for simulating
859 ; the fadeout and fade in during fainting.
860 LC384           lda effectivelight    ; fetch effective light level in dungeon
861                tst rendermagic        ; are we checking for special lighting conditions?
862                beq LC38E              ; brif not
863                lda effectivemlight    ; get magical light level
864                clr rendermagic        ; undo special light level checking
865 LC38E           clrb                  ; default to full bright
866                suba #7                ; adjust level based on the order of the table used
867                suba renderdist        ; subtract render distance from light level
868                bge LC39F              ; brif adjusted light level >= 0 - means we can see everything
869                decb                  ; change to dark default
870                cmpa #$f9              ; are we in a partial visible range?
871                ble LC39F              ; brif not - use the default value (dark)
872                ldx #LCB96             ; point to end of table of pixel masks, used for powers of two
873                levels
874                ldb a,x                ; fetch value from pixel mask (1, 2, 4, 8, 16, 32)
875 LC39F           stb lightlevel         ; save new light level (full bright, dark, or partial)
876                rts                   ; return to caller
877 ; SWI 1 routine
878 ;*****
879 ; This routine renders a line graphic from the specification stored at (X).
880 ;
881 ; The data at (X) is a series of operations as follows:
882 ;
883 ; if (X) is < $FA, then the two bytes at (X) are an absolute Y and X coordinate. If polyfirst is clear, this
884 ; is
885 ; the first vertex in a polygon and the coordinates are simply recorded. Otherwise, a line is drawn from
886 ; the previous coordinates to the new coordinates. These coordinates have the Y coordinate first.

```

```

885 ;
886 ; If (X) is >= $FA, it is a special operation defined as follows:
887 ;
888 ; FA: return from a "subroutine" to the previous flow
889 ; FB: call a subroutine at the memory address in the next two bytes
890 ; FC: draw a series of points using relative motion. The following byte is split into nibbles with the
891 ;     upper nibble being the Y displacement and the lower nibble being the X displacement. These values
892 ;     are signed and will be doubled when applied to the drawing. This gives a range of -32 to +30 in
893 ;     steps of 2 for each direction. If both displacements are zero (a zero byte), this is the end of
894 ;     the relative sequence. The end of one of these sequences is the end of a polygon.
895 ; FD: like FB but doesn't record the previous location
896 ; FE: flags the end of the input and causes a return to the caller. Do not use this in a subroutine as
897 ;     the stack will have been used to record the return data location.
898 ; FF: mark the next coordinates as the start of a new polygon.
899 ;
900 ; In all cases, the X and Y coordinates actually used have a scale factor applied to them based on the
901 ; distance from the defined centre of the graphics area which is stored in (horizcent,vertcent). The
    horizontal
902 ; scale factor is at horizscale and the vertical scaling factor is at vertscale. A factor of 128 serves as a
    scale
903 ; factor of 1. 192 would be 1.5 and 64 would be 0.5.
904 ;
905 ; Variables used:
906 ;
907 ; horizcent    the horizontal centre point for rendering graphics and scaling
908 ; vertcent     the vertical centre point for rendering graphics and scaling
909 ; lightlevel   the light level with respect to rendering the graphic
910 ; horizscale   the horizontal scaling factor (binary point to the right of bit 7)
911 ; vertscale    the vertical scaling factor (binary point to the right of bit 7)
912 ; polyfirst    nonzero if this is not the first coordinate in a polygon
913 ; lastunscalex the most recent absolute unscaled X coordinate
914 ; lastunscaley the most recent absolute unscaled Y coordinate
915 LC3A2          clr polyfirst                ; mark input as start of polygon
916              lda lightlevel                ; fetch dungeon light level
917              inca                          ; is it $ff (dark)?
918              beq LC3F6                      ; brif so - skip rendering
919 LC3A9          ldb ,x                      ; fetch input data
920              subb #$fa                      ; adjust for operation codes
921              blo LC3CF                      ; brif not operation code
922              leax 1,x                      ; move on to next image data byte
923              ldy #LC3B9                    ; point to jump table for operation codes
924              ldb b,y                      ; get offset to operation routine
925              jmp b,y                      ; execute operation routine
926 LC3B9          fcb LC3C9-LC3B9              ; (FA) return from a "subroutine"
927              fcb LC3BF-LC3B9              ; (FB) call a "subroutine"
928              fcb LC417-LC3B9              ; (FC) polygon

```

```

929          fcb LC3C6-LC3B9          ; (FD) jump to a new "routine"
930          fcb LC3F6-LC3B9          ; (FE) end of input - return to caller
931          fcb LC3CB-LC3B9          ; (FF) next coordinates are start of new polygon
932 LC3BF      ldd ,x++                ; get address of "subroutine" to call
933          stx ,--s                  ; save return address
934          tfr d,x                    ; set new "execution" address
935          skip2                      ; skip next instruction
936 LC3C6      ldx ,x                  ; get address of "routine" to jump to
937          skip2                      ; skip next instruction
938 LC3C9      ldx ,s++                ; get back saved input location
939 LC3CB      clr polyfirst           ; reset polygon start flag to start
940          bra LC3A9                  ; go process more input
941 LC3CF      tst polyfirst           ; is this the first coordinate in a polygon?
942          bne LC3D9                 ; brif not
943          bsr LC3E2                 ; fetch input coordinates and save them
944          dec polyfirst             ; flag as not first coordinate
945          bra LC3A9                  ; go process more input
946 LC3D9      bsr LC3E0               ; set up coordinates to draw a line
947          jsr drawline              ; draw the line
948          bra LC3A9                  ; go process more input
949 LC3E0      bsr LC3F7               ; move last end coordinates to line start
950 LC3E2      ldb ,x+                  ; get the next Y coordinate and move pointer forward
951          stb lastunscaley          ; save unscaled Y coordinate
952          bsr LC400                 ; scale the Y coordinate
953          addd vertcent             ; add in base Y coordinate
954          std yend                  ; save scaled end coordinate for line
955          ldb ,x+                  ; get the next X coordinate and move pointer forward
956          stb lastunscalex         ; save unscaled X coordinate
957          bsr LC406                 ; scale the X coordinate
958          addd horizcent            ; add in base X coordinate
959          std xend                  ; save scaled X coordinate for line
960 LC3F6      rts                     ; return to caller
961 LC3F7      ldd yend                ; fetch last Y coordinate
962          std ybeg                  ; save as begining of new line segment
963          ldd xend                  ; fetch last X coordinate
964          std xbeg                  ; save as beginning of new line segment
965          rts                       ; return to caller
966 LC400      lda vertscale           ; get desired vertical scaling factor
967          subb vertcent+1           ; find difference from Y base coordinate
968          bra LC40A                 ; go finish calculating scale
969 LC406      lda horizscale          ; get desired horizontal scale factor
970          subb horizcent+1         ; find difference from X base coordinate
971 LC40A      bcs LC40F               ; brif negative difference
972          mul                      ; apply the scaling factor
973          bra LC414                 ; normalize to an integer in D and return
974 LC40F      negb                    ; make coordinate difference positive

```

```

975          mul                      ; apply the scaling factor
976          jsr LC499                ; negate coordinate value
977 LC414      jmp asrd7              ; normalize to an integer in D and return
978 LC417      lda ,x+                ; get next byte in input
979          beq LC3CB                ; brif NUL - end of values
980          bsr LC3F7                ; move last end coordinate to start coordinate for line
981          ldb -1,x                 ; get the relative movement specifications
982          asrb                     ; * fetch high nibble signed extended into B
983          asrb                     ; *
984          asrb                     ; *
985          asrb                     ; *
986          lslb                     ; and multiply by two
987          addb lastunscaley         ; add in previous Y coordinate
988          stb lastunscaley         ; save new Y coordinate
989          bsr LC400                ; go scale the Y coordinate
990          addd vertcent             ; add in the Y base coordinate
991          std yend                  ; save new ending Y coordinate
992          ldb -1,x                 ; get back the input byte again
993          andb #$0f                ; mask off the upper bits
994          bitb #8                  ; is bit 3 set?
995          beq LC438                ; brif not
996          orb #$f0                 ; sign extend to 8 bits
997 LC438      lslb                   ; multiply by two
998          addb lastunscalex         ; add in saved X coordinate
999          stb lastunscalex         ; save new X coordinate
1000          bsr LC406                ; go scale the X coordinate
1001          addd horizcent           ; add in base X coordinate
1002          std xend                  ; save new ending X coordinate
1003          jsr drawline             ; go draw a line
1004          bra LC417                ; look for another line segment
1005 ; swi 2 routine
1006 ; fetch a packed string immediately following the call and display it
1007 LC448      ldx 12,s               ; fetch return address - string address
1008          decodestrsb              ; go decode string
1009          stx 12,s                 ; save new return address - after string
1010          ldx #stringbuf           ; point to decoded string
1011          skip2                    ; skip the next instruction - nothing to display yet
1012 LC452      renderchar             ; display character in A
1013 ; swi 3 routine
1014 ; display an unpacked string pointed to by X
1015 LC454      lda ,x+                ; fetch byte from string
1016          bpl LC452                ; brif not end of string - display it
1017          rts                     ; return to caller
1018 ; swi 4 routine
1019 ; display character in A
1020 LC459      tst textother          ; are we looking for standard text mode?

```

```

1021                bne LC460                                ; brif not
1022                ldu #commandarea                        ; point to display state information
1023 LC460           ldx 4,u                                ; fetch current screen location
1024                jsr LC9B2                                ; actually display the appropriate character
1025                cmpx 2,u                                ; are we at the end of text area?
1026                blo LC46C                                ; brif not
1027                jsr LC9D4                                ; go scroll the text area
1028 LC46C           stx 4,u                                ; save new screen location
1029                rts                                     ; return to caller
1030 ; swi 5 routine - decode packed string at X to stringbuf
1031 LC46F           ldu #stringbuf                          ; point to output buffer
1032 ; swi 6 routine - decode a packed string at X to U
1033 ; the first value is the length of the string less one
1034 LC472           leay -1,u                              ; point to working data before buffer
1035                clr ,y                                  ; initialize value counter
1036                bsr LC48C                                ; fetch a value
1037                tfr b,a                                  ; save length
1038 LC47A           bsr LC48C                                ; fetch a value
1039                stb ,u+                                  ; save in output
1040                deca                                     ; done yet?
1041                bpl LC47A                                ; brif not
1042                sta ,u                                  ; flag end of string with $ff
1043                tst ,y                                  ; did we consume an even number of bytes?
1044                beq LC489                                ; brif so
1045                leax 1,x                                  ; move pointer forward
1046 LC489           stx 6,s                                  ; save pointer past end of input
1047                rts                                     ; return to caller
1048 LC48C           pshs a,u                              ; save registers
1049                lda ,y                                  ; get value counter
1050                ldu #LC4A2                              ; point to value handlers
1051                lda a,u                                  ; get offset to handler for this value
1052                jsr a,u                                  ; call the handler for this value
1053                lda ,y                                  ; get value counter
1054                inca                                     ; bump it
1055                anda #7                                  ; wrap it around - the pattern repeats every 8 values
1056                sta ,y                                  ; save new value counter
1057                andb #$1f                                ; values are only 5 bits - clear out extra bits
1058                puls a,u,pc                              ; restore registers and return
1059 LC4A2           fcb LC4AA-LC4A2                        ; value 0 handler
1060                fcb LC4B0-LC4A2                        ; value 1 handler
1061                fcb LC4B5-LC4A2                        ; value 2 handler
1062                fcb LC4B9-LC4A2                        ; value 3 handler
1063                fcb LC4BE-LC4A2                        ; value 4 handler
1064                fcb LC4C3-LC4A2                        ; value 5 handler
1065                fcb LC4C7-LC4A2                        ; value 6 handler
1066                fcb LC4CC-LC4A2                        ; value 7 handler

```



```

1067 ; value 0: upper 5 bits of current input byte
1068 LC4AA      ldb ,x          ; fetch input byte
1069          lsrb              ; * align in low bits of B
1070 LC4AD      lsrb              ; *
1071 LC4AE      lsrb              ; *
1072          rts                return to caller
1073 ; value 1: lower 3 bits of current input byte and upper 2 bits of next one
1074 ; consumes a byte
1075 LC4B0      ldd ,x+          ; fetch input data and consume a byte
1076          jmp asrd6          ; align in low bits of B
1077 ; value 2: bits 5...1 of current input byte
1078 LC4B5      ldb ,x          ; fetch input byte
1079          bra LC4AE          ; align in low bits of B
1080 ; value 3: bits 0 of current byte and upper 4 bits of next one
1081 ; consumes a byte
1082 LC4B9      ldd ,x+          ; fetch input data and consume a byte
1083          jmp asrd4          ; align in low bits of B
1084 ; value 4: low 4 bits of input byte and high bit of next one
1085 ; consumes a byte
1086 LC4BE      ldd ,x+          ; fetch input data and consume a byte
1087          jmp asrd7          ; align in low bits of B
1088 ; value 5: bits 6...2 of current input byte
1089 LC4C3      ldb ,x          ; fetch input data
1090          bra LC4AD          ; align in low bits of B
1091 ; value 6: low two bits of current input byte and high 3 bits of next one
1092 ; consumes a byte
1093 LC4C7      ldd ,x+          ; fetch input data and consume a byte
1094          jmp asrd5          ; align in low bits of B
1095 ; value 7: low 5 bits of current input byte
1096 ; consumes a byte
1097 LC4CC      ldb ,x+          ; fetch input data - already aligned
1098          rts                ; return to caller
1099 ; swi 7 routine
1100 ; Generate a pseudo random number based on seed in randomseed, return 8 bit value in A
1101 LC4CF      ldx #8           ; need to generate 8 bits
1102 LC4D2      clrb              ; initialize ls counter
1103          ldy #8              ; 8 bits in byte to count
1104          lda randomseed+2     ; get lsb of seed
1105          anda #$e1           ; drop bits 4-1 (keep 7,6,5,0)
1106 LC4DB      lsla              ; shift modified seed lsb left
1107          bcc LC4DF            ; brif no carry
1108          incb                 ; bump ls count
1109 LC4DF      leay -1,y          ; done 8 bits?
1110          bne LC4DB            ; brif not
1111          lsrb                 ; take bit 0 of the count
1112          rol randomseed       ; * and shift it into the seed value

```

```

1113         rol randomseed+1                ;*
1114         rol randomseed+2                ;*
1115         leax -1,x                        ; have we generated 8 bits?
1116         bne LC4D2                        ; brif not
1117         lda randomseed                   ; get msb of current seed value
1118         sta 3,s                          ; save 8 bit random value for return
1119         rts                             ; return to caller
1120 ; swi 8 routine - clear first graphics screen
1121 LC4F3         ldu screenvis               ; point to first screen parameter block
1122             skip2                       ; skip next instruction
1123 ; swi 9 routine - clear second graphics screen
1124 LC4F6         ldu screendraw              ; point to second screen parameter block
1125             ldb levbgmask                ; get current level background colour
1126             bsr LC517                    ; go clear the graphics area of the screen
1127             stu 10,s                     ; save pointer to parameter block for the caller
1128             rts                         ; return to caller
1129 ; swi 10 routine - clear the status line
1130 LC4FF         ldx #statusarea             ; point to text area parameters for the status line
1131             ldu #LD87C                   ; point to screen address table for the status line
1132             bra LC50D                    ; go clear the status line
1133 ; swi 11 routine - clear the command entry area
1134 LC507         ldx #commandarea            ; point to text area parameters for the command area
1135             ldu #LD888                   ; point to screen address table for the command area
1136 LC50D         clr 4,x                     ;* set current cursor to start of text area
1137             clr 5,x                       ;*
1138             ldb 6,x                       ; get background colour of text area
1139             bsr LC517                     ; go clear text area
1140             leau 6,u                      ; and repeat the process for the other graphics screen
1141 LC517         pshs a,b,x,y,u              ; save regsiters
1142             sex                          ; get background colour to A
1143             tfr d,y                       ; move it into Y too (4 bytes of background colour)
1144             leax ,u                       ; point to start of parameter area
1145             ldu 2,u                       ; get address of end of text area (+1)
1146 LC520         pshu a,b,y                 ; blast 4 background bytes to area
1147             cmpu ,x                       ; are we at the start of the area?
1148             bne LC520                     ; brif not
1149             puls a,b,x,y,u,pc             ; restore registers and return
1150 ; swi 12 routine
1151 ; Check for fainting or recovery from damage and handle the fading out and fading in as a result.
1152 ; Also check for death due to damage level exceeding power level.
1153 LC529         clr accum0                  ; mark high bits of 24 bit accumulator
1154             ldd powerlevel                ; get current power level
1155             std accum0+1                  ; save it in accumulator
1156             lda #6                       ; shift left 6 bits (times 64)
1157 LC531         lsl accum0+2                ;* do one left shift
1158             rol accum0+1                  ;*

```

```

1159          rol accum0          ;*
1160          deca                  ; done enough shifts?
1161          bne LC531              ; brif not
1162          clr accum1            ; clear high bits of 24 bit accumulator
1163          ldd damagelevel       ; get damage level
1164          std accum1+1          ; stow in accumulator
1165          lsl accum1+2          ;* shift left (times 2)
1166          rol accum1+1          ;*
1167          rol accum1            ;*
1168          ldd powerlevel        ; get current power level
1169          addd accum1+1         ; add in half damage level
1170          std accum1+1         ; save low word
1171          ldb accum1            ;* propagate carry
1172          adcb #0               ;*
1173          stb accum1            ;*
1174          clr accum2            ; initialize quotient
1175 LC554      ldd accum0+1        ; get low bits of powerlevel/64
1176          subd accum1+1        ; subtract (powerlevel + damagelevel * 2)
1177          std accum0+1         ; save low word
1178          lda accum0            ; fetch msb of powerlevel/64
1179          sbca accum1          ; finish subtracting with msb of (powerlevel + damagelevel *
2)
1180          sta accum0            ; save it in msb of result
1181          inc accum2            ; bump quotient
1182          bcc LC554             ; brif no carry from addition - we haven't got a result yet
1183          lda accum2            ; get division result
1184          suba #19              ; subtract 19

1185          sta heartticks       ; save number of ticks before redoing the calculation and also
how fast heart beats
1186          tst nokeyboard       ; are we blocking the keyboard?
1187          bne LC595             ; brif so
1188          cmpa #3               ; is number of ticks > 3?
1189          bgt LC5AE             ; brif so
1190          clearcommand          ; clear the command area
1191          lda effectivelight     ; fetch the effective light level
1192          sta savedefflight     ; save it
1193 LC578      dec effectivemlight  ; mark us as passed out
1194          jsr [displayptr]      ; update the main display area
1195          dec pageswap          ; set graphics swap required
1196          sync                  ; wait for swap to happen
1197          dec effectivelight     ; reduce effective light level
1198          lda effectivelight     ; fetch new light level
1199          cmpa #$f8             ; have we reached a minimum?
1200          bgt LC578             ; brif not
1201          cleargfx2             ; clear graphics

```

```

1202          dec pageswap                ; set graphics swap required
1203          dec nokeyboard              ; disable keyboard
1204          clr keybufread              ; * reset keyboard buffer - we passed out so clear any commands
1205          clr keybufwrite             ; *
1206          bra LC5AE                   ; get on with things
1207 LC595      cmpa #4                   ; have we recovered enough to wake up?
1208          ble LC5AE                   ; brif not
1209 LC599      jsr [displayptr]          ; update the main display area
1210          dec pageswap                ; set graphics swap required
1211          sync                        ; wait for swap to happen
1212          inc effectivemlight          ; mark as not passed out
1213          inc effectivelight          ; bump effective light level
1214          lda effectivelight          ; fetch new light level
1215          cmpa savedefflight          ; are we at old intensity?
1216          ble LC599                  ; brif not
1217          clr nokeyboard              ; re-enable keyboard
1218          showprompt                  ; show the prompt
1219 LC5AE      ldx powerlevel             ; get current power level
1220          cmpx damagelevel            ; is it less than damage level?
1221          blo LC5B5                   ; brif so - we're dead!
1222          rts                         ; returnt o caller
1223 ; This routine handles player death
1224 LC5B5      ldx #img_wizard           ; point to wizard image
1225          dec enablefadesound         ; neable fade sound effect
1226          fadeinclrst                ; fade in the wizard
1227          renderstrimmp              ; display "YET ANOTHER DOES NOT RETURN..."
1228          fcb $ff,$c1,$92,$d0         ; packed "YET ANOTHER DOES NOT RETURN..." string
1229          fcb $01,$73,$e8,$82
1230          fcb $c8,$04,$79,$66
1231          fcb $07,$3e,$80,$91
1232          fcb $69,$59,$3b,$de
1233          fcb $f0
1234          clr nokeyboard              ; enable keyboard polling in IRQ
1235          dec waitnewgame             ; set up to wait for keypress to start new game
1236 LC5D7      bra LC5D7                 ; wait forever (or until the IRQ does something)
1237 ; swi 13 routine
1238 LC5D9      ldu #statusarea           ; point to parameters for status line
1239          dec textother               ; set to nonstandard text area
1240          lda levbgmask               ; get current level background
1241          coma                        ; invert it for status line
1242          sta 6,u                     ; set up for displaying status line
1243          clra                        ; set position to start clearing (start of line)
1244          clrb
1245          bsr LC609                   ; clear half the line
1246          std 4,u                     ; save display position
1247          ldx lefthand                ; fetch object in left hand

```

```

1248          bsr LC617                ; get name of object
1249          renderstr                ; display left hand object
1250          ldd #17                   ; set position to start clearing
1251          bsr LC609                 ; go clear half the line
1252          ldx righthand             ; fetch object in right hand
1253          bsr LC617                ; get name of object
1254          tfr x,y                   ; save start pointer
1255          ldd #$21                  ; set up offset for displaying right justified
1256 LC5FD      decb                   ; move cursor point left
1257          tst ,y+                   ; end of string yet?
1258          bpl LC5FD                 ; brif not - keep moving left
1259          std 4,u                   ; save render position
1260          renderstr                 ; display the right hand object
1261          clr textother             ; reset to standard text rendering
1262          rts
1263 LC609      pshs a,b               ; save registers
1264          std 4,u                   ; save the start position
1265          ldd #15                   ; set up for a space (code 0) 15 times
1266 LC610      renderchar             ; render a space
1267          decb                      ; done yet?
1268          bne LC610                 ; brif not
1269          puls a,b,pc               ; restore registers and return
1270 LC617      pshs a,b,y,u           ; save registers
1271          leay ,x                   ; point to object data
1272          bne LC622                 ; brif there is object data
1273          ldx #LC650                ; point to "EMPTY" string
1274          bra LC63C                 ; return result
1275 LC622      ldu #wordbuff          ; point to word buffer
1276          tst 11,y                  ; has it been revealed?
1277          bne LC632                 ; brif not
1278          lda 9,y                   ; fetch sub type
1279          ldx #kw_supreme            ; point to first "adjective" keyword
1280          bsr LC63E                 ; copy correct string into buffer
1281          clr -1,u                  ; make a space after adjective
1282 LC632      lda 10,y                ; get base type
1283          ldx #kw_flask              ; point to first base type keyword
1284          bsr LC63E                 ; copy correct string into buffer
1285          ldx #wordbuff             ; point to start of string
1286 LC63C      puls a,b,y,u,pc         ; restore registers and return
1287 LC63E      pshs a,x                ; save registers
1288 LC640      decodestrbs             ; decode the current string into buffer
1289          deca                      ; are we there yet?
1290          bpl LC640                 ; brif not
1291          ldx #stringbuf+1           ; point to actual string (past object type)
1292 LC648      lda ,x+                 ; fetch character from decoded keyword
1293          sta ,u+                   ; save in output buffer

```

```

1294          bpl LC648                      ; brif not end of string yet
1295          puls a,x,pc                    ; restore registers and return
1296 LC650      fcb $05,$0d,$10,$14,$19,$ff ; unpacked string "EMPTY"
1297 ; swi 14 routine
1298 LC656      tst nokeyboard                ; is keyboard disabled?
1299          bne LC65F                      ; brif so - return, don't update display
1300          bsr LC660                      ; go update the display
1301          dec pageswap                   ; flag graphics swap required
1302          sync                          ; wait for swap to happen
1303 LC65F      rts
1304 LC660      pshs a,b,x,y,u                ; save registers
1305          ldd baselight                  ; get dungeon base lighting
1306          ldu curtorch                   ; is there a torch lit?
1307          beq LC66C                      ; brif not
1308          adda 7,u                       ; add in physical light from torch
1309          addb 8,u                       ; add in magical light from torch
1310 LC66C      std effectivelight            ; save effective light level for dungeon
1311          jsr [displayptr]               ; update the main display area
1312          puls a,b,x,y,u,pc
1313 ; swi 15 routine
1314 LC674      ldx #LC67A                   ; point to newline followed by prompt
1315          renderstr                      ; go display the newline and prompt
1316          rts                          ; return to caller
1317 LC67A      fcb $1f,$1e                 ; unpacked string CR PERIOD UNDERSCORE BS (including
          following)
1318 LC67C      fcb $1c,$24,$ff             ; unpacked string UNDERSCORE BS
1319
1320 ; swi 16 routine
1321 ; delay for 81 ticks (1.3 seconds)
1322 LC67F      ldb #$51                    ; fetch delay tick count
1323 LC681      sync                      ; wait for a tick
1324          decb                          ; are we done yet?
1325          bne LC681                      ; brif not
1326          rts
1327 ; these two routine clear an area to 0 (black) or $ff (white) starting at X and
1328 ; ending at U
1329 ; swi 17 routine
1330 LC686      clra                        ; set area to $00 (clear to black)
1331          skip2                          ; skip next byte
1332 ; swi 18 routine
1333 LC688      lda #$ff                    ; set area to $FF (clear to white)
1334 LC68A      sta ,x+                      ; clear a byte
1335          cmpx 10,s                      ; are we done yet?
1336          bne LC68A                      ; brif not
1337          rts                          ; return to caller
1338 ; This looks like a leftover from earlier development which had the

```

```

1339 ; rom calls as a SWI call instead of using SWI2. This routine cannot
1340 ; be reached through the SWI mechanism and it cannot be called directly
1341 LC691      clrb                      ;* reset direct page for ROM call
1342           tfr b,dp                  ;*
1343           ldu 12,s                  ; fetch return address
1344           ldb ,u+                   ; fetch rom call wanted
1345           stu 12,s                  ; save new return address
1346           ldu #ROMTAB               ; point to ROM vector table
1347           jsr [b,u]                 ; call the routine
1348           sta 3,s                   ;* save return values
1349           stx 6,s                   ;*
1350           rts
1351 ; swi 19 routine
1352 ; fade in the image specified by (X) with sound effects, clear status line and command area
1353 LC6A4      clr enableheart          ; disable heartbeat
1354           clearstatus              ; clear the status area
1355 ; swi 20 routine
1356 ; fade in the image specified by (X) with sound effects, clear command area
1357 LC6A8      clearcommand             ; clear the command area
1358           ldd #$8080               ;* set X and Y scale values to 1.0
1359           std horizscale           ;*
1360           ldb enablefadesound      ; are we doing sound effects on the fade?
1361           beq LC6B7                ; brif not
1362           ldb #$20                 ; set apparent lighting to 32 (less apparent)
1363           dec dofadesound          ; enable fade sound effect
1364 LC6B7      bsr LC6D7               ; go draw the image
1365           decb                    ;* reduce lighting count - make more apparent
1366           decb                    ;*
1367           bpl LC6B7                ; brif not done 16 steps
1368           clr dofadesound          ; disable fade sound effect
1369           clr enablefadesound      ; turn off fade sound effect
1370 LC6C1      playsoundimm $16        ; play sound effect
1371           rts                     ; return to caller
1372 ; swi 21 routine
1373 ; fade out the image specified by (X) with sound effects, clear command area
1374 LC6C5      clearcommand            ; clear the command entry area
1375           bsr LC6C1
1376           clrb                    ; set apparent illumination to fully lit
1377           dec dofadesound          ; enable the fade buzz sound effect
1378 LC6CC      bsr LC6D7               ; go draw the image
1379           incb                    ;* bump lighting count (make less apparent)
1380           incb                    ;*
1381           cmpb #$20                ; have we done 16 steps?
1382           bne LC6CC                ; brif not
1383           clr dofadesound          ; disable the fade buzz sound effect
1384           rts                     ; return to caller

```

```

1385 LC6D7      pshs x,u                ; save registers
1386          stb lightlevel            ; set illumination value for graphic rendering
1387          stb fadesoundval          ; save intensity for the fade sound
1388          cleargfx2                  ; clear second graphics screen
1389          drawgraphic                ; go draw graphic
1390          dec pageswap               ; flag graphics swap required
1391          sync                       ; wait for swap to happen
1392          puls x,u,pc                ; restore registers and return
1393 ; swi 22 routine - display the PREPARE! screen
1394 LC6E6      jsr LD489                ; clear second graphics screen and set up for text mode
1395          ldd #$12c                  ; * set cursor position to the middle of the screen
1396          std 4,u                    ; *
1397          renderstrimmp              ; display the PREPARE! message
1398          fcb $3c,$24,$58,$06        ; packed string "PREPARE!"
1399          fcb $45,$d8
1400          clr textother              ; reset to standard text rendering
1401          dec pageswap               ; set graphic swap required
1402          rts                        ; return to caller
1403 ; swi 23 routine
1404 ; Create a new object. Associate it with the level number in B. Object type in A.
1405 LC6FB      ldu objectfree            ; fetch free point in object table
1406          stu 6,s                    ; save pointer for return
1407          leax 14,u                  ; move to next entry in table
1408          stx objectfree             ; save as new free point in object table
1409          sta 9,u                    ; set object type to requested type
1410          stb 4,u                    ; set object level
1411          setobjectspecs              ; set up object specs from data tables
1412          ldb 10,u                   ; fetch object general type
1413          ldx #LC719                 ; point to modifier table
1414          lda b,x                    ; get modified type entry
1415          bmi LC718                  ; brif no modification
1416          ldb 11,u                   ; get reveal strength of original object type
1417          setobjectspecs              ; set up object data from replacement type
1418          stb 11,u                   ; restore reveal strength
1419 LC718      rts
1420 LC719      fcb $ff                  ; flasks do not get a replacment
1421          fcb $ff                    ; rings do not get a replacement
1422          fcb $ff                    ; scrolls do not get a replacement
1423          fcb $10                    ; shields default to leather shield specs
1424          fcb $11                    ; swords default to wooden sword specs
1425          fcb $0f                    ; torches default to pine torch specs
1426 ; swi 24 routine
1427 LC71F      lsla                      ; four bytes per object specs entry
1428          lsla
1429          ldx #objspecs               ; point to object data table
1430          leay a,x                    ; point to correct entry in table

```



```

1431             leax 10,u                ; point to location in data table
1432             lda #4                   ; four bytes to copy
1433             jsr LC04B                 ; copy data into new object
1434             ldx #objextraspecs-4     ; point to extra object data
1435 LC730        leax 4,x                ; move to next entry
1436             lda ,x                   ; is it end of table?
1437             bmi LC742                 ; brif so
1438             cmpa 3,s                 ; is this entry for the object type we're creating?
1439             bne LC730                 ; brif not - try another
1440             ldd 1,x                   ; copy the ring charges and defensive values
1441             std 6,u
1442             lda 3,x
1443             sta 8,u
1444 LC742        rts                    ; return to caller
1445 ; swi 25 routine
1446 LC743        clearstatus             ; clear the status line
1447             clearcommand             ; clear the command area
1448             checkdamage              ; update damage information
1449             inc heartctr             ; bump count until next heart beat
1450             dec hidestatus           ; set command processing to proceed normally
1451             dec enableheart          ; enable heartbeat
1452             updatestatus             ; update status line to current information
1453 cmd_look     ldx #LCE66              ; standard dungeon view routine
1454             stx displayptr           ; restore display to standard dungeon view
1455             updatedungeon            ; update display
1456             rts                    ; return to caller
1457 ; swi 26 routine
1458 LC759        sta currentlevel        ; save current dungeon level
1459             ldb #12                  ; number of entries in creature count table
1460             mul                      ; calculate offset to creature counts for this level
1461             addd #creaturecounts      ; point to correct creature count table for this level
1462             std creaturecntptr       ; save pointer to creature count table
1463             ldb currentlevel         ; get back current level number
1464             ldx #holetab             ; point to hole/ladder table
1465 LC768        stx holetabptr          ; save hole/ladder data pointer
1466 LC76A        lda ,x+                 ; fetch flag
1467             bpl LC76A                 ; brif we didn't consume a flag
1468             decb                     ; are we at the right set of data for the level?
1469             bpl LC768                 ; brif not - save new pointer and search again
1470             ldx #creaturetab         ; get start address to clear
1471             ldu #mazedata            ; get end address to clear
1472             clearblock               ; go clear area to zeros
1473             jsr LC053                 ; initialize data for new level
1474             jsr createmaze           ; create the maze
1475             ldu creaturecntptr       ; point to creature counts for this level
1476             lda #11                  ; offset for wizard

```

```

1477 LC783      ldb a,u                ; get number of creatures required
1478                      beq LC78D      ; brif none
1479 LC787      jsr LCFA5              ; create a creature
1480                      decb           ; created enough creatures?
1481                      bne LC787      ; brif not
1482 LC78D      deca                    ; move on to next creature
1483                      bpl LC783      ; brif not finished all creatures
1484                      ldu #creaturetab-17 ; point to creature table
1485                      clr objiterstart ; set to iterate from beginning of object table
1486 LC795      jsr LCF63              ; go fetch object
1487                      beq LC7B6      ; brif no more objects
1488                      tst 5,x        ; is object carried?
1489                      bpl LC795      ; brif so - fetch another
1490 LC79E      leau 17,u              ; move to next creature entry
1491                      cmpu #mazedata  ; are we at the end of the creature table?
1492                      blo LC7AA      ; brif not - use this creature
1493                      ldu #creaturetab ; point to start of creature table
1494 LC7AA      tst 12,u               ; is creature alive?
1495                      beq LC79E      ; brif not
1496                      ldd 8,u        ; get existing creature inventory
1497                      stx 8,u        ; put this object at start of creature inventory
1498                      std ,x         ; now put remaining inventory in the "next" pointer
1499                      bra LC795      ; go place another object
1500 LC7B6      lda currentlevel       ; get current level
1501                      anda #1        ; set to "1" for odd, "0" for even
1502                      nega           ; negate - set to 00 for even, ff for odd
1503                      sta levbgmask  ; set level background mask
1504                      sta commandarea+6 ; set background mask for command area
1505                      sta infoarea+6 ; set background mask for text area
1506                      coma          ; invert mask
1507                      sta statusarea+6 ; set background mask for status line
1508                      rts            ; return to caller
1509 ; From here until the SWI routine jump table is the sound handling system. Any frequencies listed in the
1510 ; descriptions of these routines are for illustrative purposes as they are almost certainly wrong to a
1511 ; greater or lesser degree.
1512 ;
1513 ; swi 27 routine
1514 ; play a sound specified by the immediate identifier
1515 LC7C8      ldx 12,s                ; fetch return address
1516                      lda ,x+        ; fetch immediate data
1517                      stx 12,s        ; update return address
1518                      ldb #$ff       ; set to maximum volume
1519 ; swi 28 routine
1520 ; play a sound specified by the value in A
1521 LC7D0      stb soundvol            ; set the volume for the sound playing routine
1522                      ldx #LC7DC     ; point to sound routine jump table

```

```

1523         lsla                                ; two bytes per jump table entry
1524         jsr [a,x]                            ; call the sound generator routine
1525         clr PIA1                             ; turn off sound output
1526         rts                                 ; return to caller
1527 ; the jump table for sound routines
1528 LC7DC         fdb LC82B                       ; sound 0 - spider sound
1529             fdb LC850                       ; sound 1 - viper sound
1530             fdb LC951                       ; sound 2 - club giant sound
1531             fdb LC83C                       ; sound 3 - blob sound
1532             fdb LC8E2                       ; sound 4 - knight sound
1533             fdb LC955                       ; sound 5 - axe giant sound
1534             fdb LC84A                       ; sound 6 - scorpion sound
1535             fdb LC8DE                       ; sound 7 - shield knight sound
1536             fdb LC84D                       ; sound 8 - wraith sound
1537             fdb LC959                       ; sound 9 - galdrog sound
1538             fdb LC877                       ; sound 10 - wizard's image sound
1539             fdb LC877                       ; sound 11 - wizard sound
1540             fdb LC80A                       ; sound 12 - flask sound
1541             fdb LC811                       ; sound 13 - ring sound
1542             fdb LC827                       ; sound 14 - scroll sound
1543             fdb LC8DA                       ; sound 15 - shield sound
1544             fdb LC8A6                       ; sound 16 - sword sound
1545             fdb LC8B2                       ; sound 17 - torch sound
1546             fdb LC93F                       ; sound 18 - attack hit (player)
1547             fdb LC8E6                       ; sound 19 - attack hit (creature)
1548             fdb LC872                       ; sound 20 - walk into wall sound
1549             fdb LC86D                       ; sound 21 - creature death
1550             fdb LC88A                       ; sound 22 - wizard fade sound
1551 ; sound 12 - flask
1552 LC80A         ldu #LC823                     ; point to 410Hz base tone
1553             lda #4                          ; repeat sound 4 times
1554             bra LC816                       ; go do the sound
1555 ; sound 13 - ring
1556 LC811         ldu #LC81F                     ; point to 780Hz base tone
1557             lda #10                         ; repeat sound 10 times
1558 LC816         sta soundrepeat                ; set repeat counter
1559 LC818         jsr ,u                         ; make a sound
1560             dec soundrepeat                 ; have we done enough of them?
1561             bne LC818                       ; brif not
1562             rts
1563 ; These routines produce a "sliding" tone starting at the base frequency. The specified base
1564 ; frequency is a rough estimate. The tones are created using square waves. After each full wave,
1565 ; the delay is reduced by one which increases the frequency. The last cycle is with the delay
1566 ; equal to 1 which yields an approximate frequency of 9520Hz. Because the delays become progressively
1567 ; shorter, the lower frequency range lasts longer than the higher frequency range.
1568 ;

```

```

1569 ; The "fcb $10" instructions turn the following LDX into LDY, effecting skipping them.
1570 LC81F          ldx #$40          ; set low frequency of sliding tone to ~782Hz
1571              fcb $10
1572 LC823          ldx #$80          ; set low frequency of sliding tone to ~413Hz
1573              fcb $10
1574 ; sound 14 - scroll
1575 LC827          ldx #$100         ; set low frequency of sliding tone to ~212HzHz
1576              fcb $10
1577 ; sound 0 - spider
1578 LC82B          ldx #$20          ; set low frequency of sliding tone to ~1416Hz
1579 LC82E          bsr onesquarewave ; do one square wave
1580              leax -1,x           ; reduce delay (increase frequency)
1581              bne LC82E           ; brif not yet reached maximum frequency
1582              rts
1583 ; Output a square wave with wave time defined by delay in X.
1584 ; The frequency of the wave is per the following table, which is calculated based on the the
1585 ; clock rate of 894886.25 cycles per second and the total time taken for this routine to
1586 ; execute. The total time for this routine to execut is 120+16X cycles where X is the value
1587 ; in X. So, the table is as follows. The X values are in hexadecimal. The frequency values
1588 ; are in decimal.
1589 ;
1590 ; X      Frequency
1591 ; 0001   6580Hz
1592 ; 0020   1416Hz
1593 ; 0040   782Hz
1594 ; 0080   413Hz
1595 ; 0100   212Hz
1596 ; FFFF   0.8533Hz
1597 onesquarewave  lda #$ff          ; (2~) hold DAC high for delay in X
1598              bsr setdacdel       ; (7~)
1599              clra                ; (2~) hold DAC low for delay in X
1600              bra setdacdel       ; (3~)
1601 ; sound 3 - blob
1602 ; Output a series of 16 ascending tones with a base frequency descending from 43.4Hz to 27.2Hz.
1603 LC83C          ldx #$500         ; set for an ascending tone from 43.4Hz
1604 LC83F          bsr onesquarewave ; go make the sound
1605              leax $30,x          ; decrease starting tone frequency by a bit
1606              cmpx #$800          ; have we reached 27.2Hz?
1607              blo LC83F           ; brif not
1608              rts
1609 ; sound 6 - scorpion
1610 LC84A          lda #2            ; two bits for scorpion
1611              skip2
1612 ; sound 8 - wraith
1613 LC84D          lda #1            ; one bit for wraith
1614              skip2

```

```

1615 ; sound 1 - viper
1616 ; This generates a sequence of sounds at notionally 5524Hz but it uses random amplituds so
1617 ; it's more of a random sound. The sound lasts about 35ms
1618 LC850      lda #10                      ; ten bits for viper
1619           sta soundrepeat2              ; save repeat count
1620 LC854      ldy #$c0                      ; number of iterations for tone generation
1621 LC858      bsr sndseqnext                ; (7~) get a sequence value
1622           bsr setdac                    ; (7~) set the dac
1623           leay -1,y                     ; (5~) done enough iterations?
1624           bne LC858                     ; (3~) brif not
1625           bsr LC8BA                      ; delay for 36.6 ms
1626           dec soundrepeat2              ; done repeats?
1627           bne LC854                     ; brif not
1628           rts                          ; return to caller
1629 ; This entry point takes a delay in X and programs the DAC with a value from the sequence generator.
1630 ; It exits after waiting out the X delay. It uses the MSB of the sequence value.
1631 setdacseqdel bsr sndseqnext              ; (7~) get a value from the sequence to set the DAC
1632 ; This entry point takes a delay in X and the DAC value in A. It programs the DAC and waits out
1633 ; the delay in X.
1634 setdacdel   bsr setdac                   ; (7~) program the DAC
1635           bra snddelay                    ; (3~) count down delay non-destructively
1636 ; sound 21 - creature death
1637 ; This does a slightly longer variation of the last sound for sound 22 below:
1638 ; A bust sliding from 622Hz to 162Hz, frequency shifting every 2.5 waves.
1639 ; This routine spins the sequence 640 times.
1640 LC86D      ldu #LDBDA                    ; point to creature death tone generator parameters
1641           bra LC893                      ; go process the sound
1642 ; sound 20 - walk into wall
1643 ; This one uses exactly the same tone as the first half of sound 22.
1644 ; That is a short burst sliding from 405Hz to 162Hz, frequency shifting every half wave
1645 ; This routine spins the sequence 104 times.
1646 LC872      ldu #LDBD2                    ; point to the generation specification for the sound
1647           bra LC893                      ; go generate the sound
1648 ; sound 10, sound 11 - wizard's image, wizard
1649 LC877      lda #8                        ; do 8 iterations of this scheme
1650           sta soundrepeat                ; set iteration counter
1651 LC87B      bsr sndseqnext                ; calculate new delay factor
1652           clra                          ; lose MSB
1653           lsr b                          ; double delay factor
1654           bne LC882                      ; brif not zero
1655           incb                          ; make sure don't do a massive delay
1656 LC882      tfr d,x                      ; put delay into correct register
1657           bsr LC82E                      ; do a sliding tone
1658           dec soundrepeat                ; have we done enough yet?
1659           bne LC87B                      ; brif not
1660 ; sound 22 - sound made just as a wizard fades out

```

```

1661 ; start with a short burst sliding from 405Hz to 162Hz, frequency shifting every half wave
1662 ; then, delay 36.6ms
1663 ; then, do a longer burst sliding from 622Hz to 162Hz, frequency shifting every two waves
1664 ; both bursts have semi-random amplitude derived from the sequence generator.
1665 ; For this sound, the sequence will be spun 616 times.
1666 LC88A          ldu #LDBD2                ; point to tone generator info
1667              bsr LC893                  ; process first pair
1668              bsr LC8BA                  ; delay for 36.6ms
1669              leau 4,u                  ; move to next pair of values
1670 LC893          ldx ,u                    ; get delay value (frequency)
1671 LC895          ldy 2,u                  ; get wave count for each frequency
1672 LC898          bsr setdacseqdel          ; set the dac for the first half-wave
1673              leay -1,y                  ; are we done yet?
1674              bne LC898                  ; brif not
1675              leax 2,x                    ; lengthen delay slightly (reduce frequency)
1676              cmpx #$150                 ; are we at the minimum frequency (163Hz)?
1677              bne LC895                  ; brif not - get wave count again and keep going
1678              rts                        ; return to caller
1679 ; sound 16 - sword
1680 ; Uses random amplitude on an ascending volumn scale (roughly 510 iterations)
1681 LC8A6          jsr LC931                ; * set for ascending volume from 0 to $ff with a step of 0.5
1682              fcb $80                    ; *
1683 LC8AA          bsr LC922                ; apply step and program DAC
1684              bcs LC8B2                  ; brif counter wrapped
1685              bsr setdac                 ; set the DAC
1686              bra LC8AA                  ; keep looping
1687 ; sound 17 - torch
1688 ; Uses a random amplitude on a descending volume scale (roughly 405 iterations)
1689 LC8B2          jsr LC92E                ; * set for descending volume from $ffff with a step of 0.625
1690              fcb $a0                    ; *
1691 LC8B6          bsr LC926                ; apply step, multiplier, and set the dac - will return to our
            caller when done
1692              bra LC8B6                  ; go apply another step
1693 LC8BA          ldx #$1000                ; delay factor for 36.6ms
1694 ; This routine counts X down nondestructively. It takes 16+8n cycles where
1695 ; n is the value in X.
1696 snddelay       pshs x                    ; (7~) save delay counter
1697 snddelay000    leax -1,x                ; (5~) has timer expired?
1698              bne snddelay000            ; (3~) brif not
1699 LC8C3          puls x,pc                 ; (9~) restore delay counter and return
1700 ; This routine programs the DAC with the intensity in A adjusted by the sound volume.
1701 ; This routine takes 27 cycles.
1702 setdac         ldb soundvol              ; (5~) fetch volume multiplier for sound
1703              mul                        ; (11~) multiply it by the value we're trying to set
1704              anda #$fc                  ; (2~) lose the non-DAC bits
1705              sta PIA1                  ; (5~) set DAC

```

```

1706             rts                ; (5~)
1707 ; This routine is a sequence generator with a period of 32768. soundseqseed is never initialized except
1708 sndseqnext    ldd soundseqseed    ; (5~) fetch current value
1709             lslb                ;* (2~) multiply by 4
1710             rola                ;* (2~)
1711             lslb                ;* (2~)
1712             rola                ;* (2~)
1713             addd soundseqseed      ; (6~) add to previous value
1714             incb                ; (2~) bump lsb
1715             std soundseqseed      ; (5~) save new value
1716             rts                ; (5~) return to caller
1717 ; sound 15 - shield
1718 ; Run a dual wave with a low wave of 955Hz and a high wave of 3020Hz
1719 LC8DA         bsr sndrundualwave
1720             fdb $6424
1721 ; sound 7 - shield knight
1722 ; Run a dual wave with a low wave of 1670Hz and a high wave of 3195Hz
1723 LC8DE         bsr sndrundualwave
1724             fdb $3212
1725 ; sound 4 - knight
1726 ; Run a dual wave with a low wave of 580Hz and a high wave of 1575Hz
1727 LC8E2         bsr sndrundualwave
1728             fdb $AF36
1729 ; sound 19 - attack hit against player
1730 ; Run a dual wave with a low wave of 2660Hz and a high wave of 4300Hz
1731 LC8E6         bsr sndrundualwave
1732             fdb $1909
1733 ; This routine runs essentially a dual tone. The "frequency" of the lower bits is determined by the value
1734 ; in sndlowtonedel. The frequency of the high bit is determined by the delay in sndhitonedel. The two
1735 ; frequencies run
1736 ; independently.
1736 LC8EA         bsr LC92E          ;* set up for descending volume with a step of 0.375
1737             fcb $60              ;*
1738 LC8ED         ldx sndlowtonedel    ; fetch low bits flip rate
1739             ldy sndhitonedel      ; fetch high bit flip rate
1740             clra                  ; initialize both "waves" to low
1741 LC8F3         leax -1,x            ; (5~) have we timed out on this level?
1742             bne LC8FD             ; (3~) brif not
1743             ldx sndlowtonedel      ; (5~) reset counter
1744             eora #$7f             ; (2~) flip all low bits of dac value
1745             bsr LC90A             ; (7~) apply step and scale - will return to our caller when
1746             things overflow (111~)
1746 LC8FD         leay -1,y           ; (5~) have run through the other sequence
1747             bne LC8F3             ; (3~) brif not - start again
1748             ldy sndhitonedel      ; (6~) reset counter
1749             eora #$80             ; (2~) flip high bit of dac value

```

```

1750          bsr LC90A          ; (7~) apply step and scale - will return to our caller whent
hings overflow (111~)
1751          bra LC8F3          ; (3~) go check things again
1752 LC90A      sta sndtemp      ; (4~) save dac value
1753          bsr LC97E          ; (7~) go calculate step and multiplier (53~)
1754          bls LC8C3          ; (3~) skip the caller to this routine and return to its
caller (PULS X,PC) if we wrapped
1755          bsr setdac          ; (7~) set the dac (28~)
1756          lda sndtemp        ; (4~) get back original dac value
1757          rts                ; (5~) return to caller
1758 ; this routine doesn't return to the caller but to the caller's caller
1759 sndrundualwave ldx ,s++      ; fetch location of parameters
1760          ldb ,x+              ; fetch delay constant for low wave
1761          clra                ; zero extend
1762          std sndlowtonedel    ; save it
1763          ldb ,x+              ; fetch delay constant for high wave
1764          std sndhitonedel    ; save it
1765          bra LC8EA          ; go process sound
1766 LC922      bsr sndseqnext    ; get a value from the sequence
1767          bra LC98D          ; apply step and multiplier (ascending)
1768 LC926      bsr sndseqnext    ; get value from sequence
1769 LC928      bsr LC97E          ; apply step and multiplier (descending)
1770          bls LC8C3          ; skip the caller to this routine and return to its caller if
we wrapped
1771          bra setdac          ; set the dac and return
1772 LC92E      ldx allones       ; set initial base value to $ffff
1773          fcb $10             ; go set up the step value
1774 LC931      ldx zero          ; set initial base value to $0000
1775 LC933      stx sndampmult    ; save initial base multiplier
1776          ldx ,s              ; get return address
1777          ldb ,x+              ; fetch step value
1778          clra                ; zero extend
1779          std sndampstep      ; save step value
1780          stx ,s              ; save return address to be after step value
1781          rts                ; return to caller
1782 ; sound 18 - attack hit against creature
1783 ; This is a sort of noisy square wave with a rough frequency of 4360Hz
1784 LC93F      bsr LC92E          ; * set up a countdown with step 0.375
1785          fcb $60             ; *
1786 LC942      jsr sndseqnext    ; get a sequence value
1787          lsra                ; make it in the low half of the range
1788          bsr LC928          ; apply step and multiplier (descending) (will return to
caller when overflow)
1789          jsr sndseqnext      ; get another sequence value
1790          ora #$80            ; force it high
1791          bsr LC928          ; apply the step and multiplier (descending) (will return to
our caller when overflow)

```



```

1792          bra LC942                ; keep looping
1793 ; These three are basically the same sound. However, the stronger creatures have longer sounds that take
      longer
1794 ; to reach full volume, and thus longer to complete. The axe giant is roughly twice as long as the club giant
      and
1795 ; the galdrog is roughly three times as long as the club giant.
1796 ; sound 2 - club giant
1797 LC951      ldx #$300                ; step value for club giant (3)
1798          fcb $10
1799 ; sound 5 - axe giant                ; step value for axe giant (2)
1800 LC955      ldx #$200
1801          fcb $10
1802 ; sound 9 - galdrog                ; step value for galdrog (1)
1803 LC959      ldx #$100
1804          stx sndampstep            ; save step value
1805          clra                      ; starting value at 0 (count up)
1806          clrb
1807          std sndampmult            ; set starting multiplier
1808 LC962      bsr LC922                ; get a value from the sequence and apply multiplier
1809          bcs LC971                ; brif we overflowed - done
1810          jsr setdac                ; set the dac
1811          ldx #$f0                  ; delay for roughly 200Hz
1812          jsr snddelay              ; go delay
1813          bra LC962                ; go run another half wave
1814 LC971      bsr LC92E                ; * set up for a count down with a step of 0.25
1815          fcb $40                  ; *
1816 LC974      bsr LC926                ; get a sequence value and apply the step (descending), will
      return to our caller when done
1817          ldx #$60                  ; get delay roughly equal to 1050Hz
1818          jsr snddelay              ; do the delay
1819          bra LC974                ; go do another half wave
1820 LC97E      pshs a                  ; (6~) save sequence value
1821          ldd sndampmult            ; (5~) get mulitplier base
1822          subd sndampstep           ; (6~) apply step value
1823 LC984      pshs cc                  ; (6~) save result of subtraction
1824          std sndampmult            ; (5~) save new multiplier base
1825          ldb 1,s                   ; (5~) get back dac value
1826          mul                      ; (11~) apply multiplier - use MSB in A
1827          puls cc,b,pc              ; (9~) restore registers and return
1828 LC98D      pshs a                  ; save sequence value
1829          ldd sndampmult            ; get multiplier base
1830          addd sndampstep           ; add step value
1831          bra LC984                ; go deal with multiplier
1832 ; this is the swi routine offset table - each byte is the difference between the entry point
1833 ; of the previous routine and itself
1834 LC995      fcb 0                    ; first routine has nothing before it
1835          fcb LC3A2-LC384

```

```

1836         fcb LC448-LC3A2
1837         fcb LC454-LC448
1838         fcb LC459-LC454
1839         fcb LC46F-LC459
1840         fcb LC472-LC46F
1841         fcb LC4CF-LC472
1842         fcb LC4F3-LC4CF
1843         fcb LC4F6-LC4F3
1844         fcb LC4FF-LC4F6
1845         fcb LC507-LC4FF
1846         fcb LC529-LC507
1847         fcb LC5D9-LC529
1848         fcb LC656-LC5D9
1849         fcb LC674-LC656
1850         fcb LC67F-LC674
1851         fcb LC686-LC67F
1852         fcb LC688-LC686
1853         fcb LC6A4-LC688
1854         fcb LC6A8-LC6A4
1855         fcb LC6C5-LC6A8
1856         fcb LC6E6-LC6C5
1857         fcb LC6FB-LC6E6
1858         fcb LC71F-LC6FB
1859         fcb LC743-LC71F
1860         fcb LC759-LC743
1861         fcb LC7C8-LC759
1862         fcb LC7D0-LC7C8
1863 ;*****
1864 ; The following code handles displaying text on the screen. It works as follows.
1865 ;
1866 ; The graphics screen is divided into a grid of character cells 32 columns wide by 24 rows high. Each cell
1867 ; is 8 pixels wide by 8 pixels high. Text can be rendered anywhere on the screen as long as it fits within
1868 ; a character cell. The cells line up on even bytes which makes actually rendering the characters fast.
1869 ;
1870 ; Characters are encoded in 5 bits as follows: A through Z are given codes 1 through 26. 0 is a space. 27
1871 ; is the exclamation point, 28 is the underscore, 29 is the question mark, and 30 is the period. Code 31
1872 ; is used as a carriage return. Codes 32 and 33 are the left and right parts of the contracted heart symbol
1873 ; while 34 and 35 are the left and right parts of the expanded heart symbol. 36 is backspace.
1874 ;
1875 ; Glyphs for codes 0 through 30 are encoded using the packed five bit encoding and are located at LDB1B. They
1876 ; are encoded in a 5 by 7 bitmap which is shifted to be offset one pixel from the left of the character cell
1877 ; upon decoding.
1878 ;
1879 ; The glyphs for the heart codes are in unpacked encoding and are located at LDBB6 and occupy the entire
1880 ; 8 bit width of the character cell.
1881 ;

```

```

1882 ; These routines expect a pointer to the text configuration parameters in U. At offset 0 is the start address
1883 ; of the scrollable area of the screen (memory address). At offset 2 is the ending character cell address of
1884 ; the scrollable area of the screen. At offset 4 is the current printing position. At offset 6 is a mask with
1885 ; all pixels set to the background colour. At offset 7 a flag which when nonzero inhibits rendering text to
1886 ; the secondary graphics screen area. For the ordinary command entry area at the bottom of the screen, this
1887 ; will point to commandarea.
1888 LC9B2      cmpa #$24      ; is it backspace?
1889           beq LC9BF      ; brif so
1890           cmpa #$1f      ; vertical spacer?
1891           beq LC9CA      ; brif so
1892           bsr LCA17      ; go handle a glyph
1893           leax 1,x        ; move to next character position
1894           rts            ; return to caller
1895 LC9BF      leax -1,x      ; move display pointer back one
1896           cmpx allones   ; did we wrap around negative?
1897           bne LC9C9      ; brif not
1898           ldx 2,u        ; get end of text area
1899           leax -1,x      ; move back one position to be in the text area
1900 LC9C9      rts            ; return to caller
1901 LC9CA      leax $20,x     ; move pointer forward one character row
1902           exg d,x        ; move pointer so we can do math
1903           andb #$e0      ; force pointer to the start of the line
1904           exg d,x        ; put pointer back where it belongs
1905           rts            ; return to caller
1906 LC9D4      pshs a,b,x,y  ; save registers
1907           ldx ,u         ; get start of screen address
1908           ldd 2,u        ; get end of text area
1909           subd #$20      ; knock one character row off it
1910           std 2,s        ; save new display location
1911           bsr LCA10      ; multiply by 8 - 8 pixel rows per cell
1912           tfr d,y        ; save counter
1913 LC9E3      ldd $100,x    ; get bytes 8 pixel rows ahead
1914           tst 7,u        ; do we need to skip the second screen?
1915           bne LC9EF      ; brif so
1916           std $1800,x    ; save scroll data on second screen
1917 LC9EF      std ,x++      ; save scroll data and move pointer ahead
1918           leay -2,y      ; are we done yet?
1919           bne LC9E3      ; brif not
1920           ldb 6,u        ; fetch current background colour
1921           sex            ; and make A match
1922           ldy #$100      ; number of bytes to blank bottom row
1923 LC9FC      tst 7,u        ; are we doing second screen too?
1924           bne LCA04      ; brif not
1925           std $1800,x    ; blank pixels in second screen
1926 LCA04      std ,x++      ; blank pixels and move pointer forward
1927           leay -2,y      ; are we done yet?

```

```

1928                bne LC9FC                        ; brif not
1929                puls a,b,x,y,pc                    ; restore registers and return
1930 LCA0C            lslb                             ; * enter here to shift D left 5 bits
1931                rola                               ; *
1932                lslb                             ; *
1933                rola                               ; *
1934 LCA10            lslb                             ; * enter here to shift D left 3 bits
1935                rola                               ; *
1936 LCA12            lslb                             ; * enter here to shift D left 2 bits
1937                rola                               ; *
1938                lslb                             ; *
1939                rola                               ; *
1940                rts
1941 LCA17            pshs a,b,x,y,u                    ; save registers
1942                cmpa #$20                          ; is it a printing character?
1943                blo LCA29                          ; brif so
1944                suba #$20                          ; mask off printing characters
1945                ldb #7                             ; 7 bytes per font table entry
1946                mul                                ; get offset in table
1947                addd #LDBB6                        ; add in base address of table
1948                tfr d,x                            ; put font pointer somewhere useful
1949                bra LCA44                          ; go draw glyph
1950 LCA29            ldb #5                             ; 5 bytes per font table entry
1951                mul                                ; get offset in table
1952                addd #LDB1B                        ; add in base address of table
1953                tfr d,x                            ; put pointer somewhere useful
1954                ldu #fontbuf                        ; point to buffer to decode glyph data
1955                decodestr                          ; go decode a packed string
1956                ldx #fontbuf+7                    ; point one past end of buffer
1957 LCA39            lsl ,-x                          ; * centre glyph data in byte
1958                lsl ,x                             ; *
1959                cmpx #fontbuf                      ; at start of buffer?
1960                bhi LCA39                          ; brif not - keep centring
1961                ldu 6,s                            ; get back U value
1962 LCA44            ldd 4,u                          ; get display address location
1963                bsr LCA10                          ; multiply by 8 - gets start of row in 11..8
1964                lsr b                              ; * and divide lsb by 8 again to get offset within
1965                lsr b                              ; * the row to bits 4..0
1966                lsr b                              ; * and force to top of character cell
1967                addd ,u                            ; add in start of text area
1968                tfr d,y                            ; put pointer somewhere useful
1969                ldb #7                             ; seven bytes to copy
1970 LCA51            lda ,x+                          ; get byte from font data
1971                eora 6,u                          ; merge with background colour
1972                sta ,y                             ; save it on the screen
1973                tst 7,u                            ; do we need to update second screen?

```

```

1974          bne LCA5F          ; brif not
1975          sta $1800,y        ; save pixels on second screen
1976 LCA5F      leay $20,y        ; move display pointer down one pixel row
1977          decb              ; are we done yet?
1978          bne LCA51          ; brif not - do another
1979          puls a,b,x,y,u,pc   ; restore registers and return
1980 ; This routine divides a 16 bit unsigned value in D by a 16 bit unsigned value in X. The result
1981 ; will be in D with the binary point to the right of A.
1982 LCA67      pshs a,b,x        ; make hole for result and save divisor
1983          clr ,s              ; * initialize quotient
1984          clr l,s              ; *
1985          clr accum0          ; use accum0 for extra precision on dividend
1986          std accum0+1        ; save dividend
1987          beq LCA97           ; brif dividend is zero - nothing to do
1988          cmpd 2,s            ; is dividend equal to divisor?
1989          bne LCA7C           ; brif not
1990          inc ,s              ; set quotient to 1.0
1991          bra LCA97           ; go return
1992 LCA7C      ldx #16           ; we need to do 16 iterations
1993 LCA7F      lsl accum0+2      ; * shift dividend
1994          rol accum0+1        ; *
1995          rol accum0          ; *
1996          lsl l,s            ; = shift quotient
1997          rol ,s              ; =
1998          ldd accum0          ; get dividend high word
1999          subd 2,s            ; subtract out divisor
2000          bcs LCA93           ; brif it doesn't go
2001          std accum0          ; save new dividend residue
2002          inc l,s            ; record the fact that it went
2003 LCA93      leax -1,x         ; have we done all 16 bits?
2004          bne LCA7F           ; brif not
2005 LCA97      puls a,b,x,pc     ; fetch result, restore registers, and return
2006 LCA99      coma             ; * do a one's complement of D
2007          comb              ; *
2008          addd #1            ; adding 1 turns it into negation
2009          rts                ; return to caller
2010 LCA9F      pshs a,b,x        ; save registers
2011          ldx pixelcount      ; get number of pixels to draw
2012          ldd ,s              ; get the difference
2013          bpl LCAAE          ; brif positive
2014          bsr LCA99          ; negate difference
2015          bsr LCA67          ; divide by number of pixels
2016          bsr LCA99          ; negate the result
2017          skip2              ; skip next instruction
2018 LCAAE      bsr LCA67          ; divide by number of pixels
2019          std ,s             ; save step value

```

```

2020          puls a,b,x,pc          ; restore registers and return
2021 LCAB4          jmp LCB8A          ; go return from the line drawing routine
2022 ; Draw a line from (xbeg,ybeg) to (xend,yend) respecting the light level in the dungeon (lightlevel)
2023 ; which is used as a step count between when to actually draw pixels.
2024 ;
2025 ; Variables used:
2026 ; lightlevel    the current light level in the dungeon
2027 ; lightcount    how many pixels left before we actually draw another
2028 ; ybeg          input start Y
2029 ; xbeg          input start X
2030 ; yend          input end Y
2031 ; xend          input end X
2032 ; xcur          X coordinate of pixel to be drawn (24 bits with 8 bits after binary point)
2033 ; ycur          Y coordinate of pixel to be drawn (24 bits with 8 bits after binary point)
2034 ; xpstep        24 bit X coordinate difference (per pixel)
2035 ; ypstep        24 bit Y coordinate difference (per pixel)
2036 ; pixelcount    number of pixels to draw in the line
2037 ; xstep         the offset for when X coordinate goes to a new byte
2038 ; ystep         the offset for when Y coordinate goes to a new line
2039 ; drawstart     the start address of the graphics screen area the line is within
2040 ; drawend       the end address of the graphics screen area the line is within
2041 ; accum0        a temporary scratch variable
2042 ;
2043 ; Note: ypstep+1 and xpstep+1 are also used as temporary holding values for the
2044 ; integer difference in the Y and X coordinates respectively.
2045 drawline        pshs a,b,x,y,u          ; save registers
2046                inc lightlevel          ; are we completely dark?
2047                beq LCAB4                ; brif so - we can short circuit drawing entirely
2048                lda lightlevel          ; get light level in dungeon
2049                sta lightcount          ; save in working count (skip count for pixel drawing)
2050                ldd xend                ; get end X coordinate
2051                subd xbeg                ; subtract start X coordinate
2052                std xpstep+1            ; save coordinate difference
2053                bpl LCACB                ; brif positive difference
2054                bsr LCA99                ; negate the difference
2055 LCACB            std pixelcount          ; save absolute value of X difference as pixel count
2056                ldd yend                ; get end Y coordinate
2057                subd ybeg                ; subtract start Y coordinate
2058                std ypstep+1            ; save coordinate difference
2059                bpl LCAD7                ; brif positive difference
2060                bsr LCA99                ; negate the difference
2061 LCAD7            cmpd pixelcount          ; is the Y difference bigger than X?
2062                blt LCAE0                ; brif not
2063                std pixelcount          ; save Y difference as pixel count
2064                beq LCAB4                ; brif no pixels to draw - short circuit
2065 LCAE0            ldd xpstep+1            ; get X difference

```

```

2066          bsr LCA9F                ; calculate X stepping value
2067          std xstep+1              ; save X stepping value
2068          tfr a,b                  ; save msb of difference
2069          sex                      ; sign extend it
2070          ldb #1                   ; X stepping value - 1 for ascending
2071          sta xstep                ; sign extend stepping difference to 24 bits
2072          bpl LCAF0                ; brif positive
2073          negb                     ; set stepping value to -1
2074 LCAF0      stb xstep              ; save X byte stepping value
2075          ldd ypstep+1             ; get Y difference
2076          bsr LCA9F                ; calculate Y step value
2077          std ypstep+1            ; save result
2078          tfr a,b                  ; save msb of difference
2079          sex                      ; sign extend it
2080          ldb #$20                 ; Y byte stepping value - 32 bytes per line, ascending
2081          sta ypstep              ; sign extend the difference to 24 bits
2082          bpl LCB02                ; brif positive
2083          negb                     ; negate the difference - -32 bytes per line, descending
2084 LCB02      stb xstep              ; save Y byte stepping value
2085          ldd xbeg                 ; get start X coordinate
2086          std xcur                 ; save in X coordinate counter
2087          ldd ybeg                 ; get start Y coordinate
2088          std ycur                 ; save in Y coordinate counter
2089          lda #$80                 ; value for low 8 bits to make the values ".5"
2090          sta xcur+2               ; set X coordinate to ".5"
2091          sta ycur+2               ; set Y coordinate to ".5"
2092          ldx 2,u                  ; get end of graphics area address
2093          stx drawend              ; save it for later
2094          ldx ,u                   ; get start of graphics area address
2095          stx drawstart            ; save it for later
2096          ldd ycur                 ; get Y coordinate for pixel
2097          jsr LCA0C                ; shift left 5 bits - 32 bytes per row
2098          leax d,x                 ; add to screen start address
2099          ldd xcur                 ; get X coordinate for pixel
2100          jsr asrd3                ; shift right 3 bits - 8 pixels per byte
2101          leax d,x                 ; add to row start address
2102          ldu #LCB8E               ; point to table of pixel masks
2103          ldy pixelcount           ; get number of pixels to draw
2104 LCB2E      dec lightcount         ; are we ready to draw another pixel (due to light level)?
2105          bne LCB54                ; brif not
2106          lda lightlevel           ; get light level
2107          sta lightcount           ; reset current "pixel delay"
2108          tst xcur                 ; is X coordinate off the right of the screen?
2109          bne LCB54                ; brif so
2110          cmpx drawstart           ; is the pixel address before the start of the graphics area?
2111          blo LCB54                ; brif so

```

```

2112      cmpx drawend                ; is the pixel address after the end of the graphics area?
2113      bhs LCB54                    ; brif so
2114      ldb xcur+1                   ; get X coordinate lsb
2115      andb #7                      ; mask off low 3 bits for offset in byte
2116      lda b,u                     ; get pixel mask to use
2117      tst levbgmask                ; currently using black background?
2118      beq LCB50                    ; brif so
2119      coma                         ; invert mask for white background
2120      anda ,x                      ; merge in existing graphics data
2121      skip2                        ; skip next instruction
2122 LCB50      ora ,x                 ; merge in existing graphics data (black background)
2123      sta ,x                       ; save new graphics data on the screen
2124 LCB54      lda xcur+1             ; get X coordinate lsb
2125      anda #$f8                   ; mask off the pixel offset in the byte
2126      sta accum0                  ; save it for later
2127      ldd xcur+1                   ; get X coordinate low bits
2128      addd xpstep+1               ; add in X difference
2129      std xcur+1                   ; save new low bits for X coordinate
2130      ldb xcur                     ; get X coordinate high bits
2131      adcb xpstep                  ; add in difference high bits
2132      stb xcur                     ; save new X coordinate high bits
2133      anda #$f8                   ; mask off pixel offset in data byte
2134      cmpa accum0                 ; are we in the same byte?
2135      beq LCB70                    ; brif so
2136      ldb xstep                    ; get byte X step value
2137      leax b,x                     ; move pointer appropriately
2138 LCB70      ldd ycur+1             ; get Y coord low bits
2139      sta accum0                  ; save screen Y coordinate
2140      addd ypstep+1               ; add in Y step value low bits
2141      std ycur+1                   ; save new low bits
2142      ldb ycur                     ; get Y coord high bits
2143      adcb ypstep                  ; add in Y step value high bits
2144      stb ycur                     ; save new Y coord high bits
2145      cmpa accum0                 ; are we on the same screen row?
2146      beq LCB86                    ; brif so
2147      ldb ystep                    ; get Y byte step value
2148      leax b,x                     ; move pointer appropriately
2149 LCB86      leay -1,y              ; have we drawn all the pixels?
2150      bne LCB2E                    ; brif not - draw another
2151 LCB8A      dec lightlevel          ; compensate for "inc" above
2152      puls a,b,x,y,u,pc            ; restore registers and return
2153 LCB8E      fcb $80,$40,$20,$10    ; pixels 0, 1, 2, 3 (left to right) in byte
2154      fcb $08,$04,$02,$01          ; pixels 4, 5, 6, 7 (left to right) in byte
2155 LCB96      pshs a,x,u             ; save registers
2156      ldx linebuffptr             ; get input buffer/line pointer
2157      ldu #wordbuff               ; point to word buffer

```



```

2158 LCB9D      lda ,x+                ; get character from input
2159           beq LCB9D                ; brif end of line
2160           bra LCBA5                ; get on with things
2161 LCBA3      lda ,x+                ; get new character from input
2162 LCBA5      ble LCBAF              ; brif not valid character
2163           sta ,u+                  ; save filename character
2164           cmpu #wordbuffend        ; are we at the end of the buffer?
2165           blo LCBA3                ; brif not - check another
2166 LCBAF      lda #$ff               ; put end of word marker
2167           sta ,u+
2168           stx linebuffptr          ; save new input pointer location
2169           tst wordbuff             ; set flags for whether we have a word
2170           puls a,x,u,pc            ; restore registers and return
2171 ; Parse an object from command line
2172 parseobj   clr parsegenobj        ; flag generic object type
2173           ldx #kwlist_obj          ; point to object type list
2174           bsr LCBEC                ; look up word in object type list
2175           bmi parseobj000          ; brif no match - try matching specific type
2176           beq badcommandret        ; brif no match - error out
2177           std parseobjtype         ; save object type matched
2178           rts                     ; return to caller
2179 parseobj000 dec parsegenobj        ; flag specific object type found
2180           ldx #kwlist_adj          ; point to specific object types
2181           bsr LCBE7                ; look up word in object type list
2182           ble badcommandret        ; brif no match
2183           std parseobjtype         ; save object type
2184           ldx #kwlist_obj          ; point to generic object types
2185           bsr LCBEC                ; look up keyword
2186           ble badcommandret        ; brif no match
2187           cmpb parseobjtypegen      ; did the object type match?
2188           bne badcommandret        ; brif not
2189           rts                     ; return to caller
2190 badcommandret leas 2,s             ; don't return to caller - we're bailing out
2191 badcommand renderstrimmp          ; display "???" for unknown command
2192           fcb $17,$7b,$d0          ; packed "???" string
2193           rts                     ; return to caller's caller
2194 LCBE7      pshs a,b,x,y,u          ; save registers
2195           clra                     ; initialize specific type to zero
2196           bra LCBF4                ; go look up keyword
2197 LCBEC      pshs a,b,x,y,u          ; save registers
2198           clra                     ; initialize specific type to zero
2199           clrb                     ; initialize generic type to zero
2200           bsr LCB96                ; parse a word from the input line
2201           bmi LCC2D                ; brif no word present
2202 LCBF4      clr kwmatch             ; flag no match
2203           clr kwexact              ; flag incomplete match

```

```

2204          ldb ,x+                ; fetch number of keywords in list
2205          stb kwcount            ; save it in temp counter
2206 LCBFC      ldu #wordbuff        ; point to decode buffer
2207          decodestrsb           ; decode the keyword string
2208          ldy #stringbuf+1       ; point to decoded keyword string (past the object code)
2209 LCC05      ldb ,u+              ; get a character from word string
2210          bmi LCC17              ; brif end of string
2211          cmpb ,y+              ; does it match?
2212          bne LCC22              ; brif not
2213          tst ,y                ; are we at the end of the keyword?
2214          bpl LCC05              ; brif not
2215          tst ,u                ; are we at the end of the word?
2216          bpl LCC22              ; brif not
2217 LCC15      dec kwexact          ; flag complete match
2218 LCC17      tst kwmatch          ; do we already have a match?
2219          bne LCC2B              ; brif so
2220          inc kwmatch            ; mark match found
2221          ldb stringbuf         ; get the keyword code
2222          std ,s                ; save keyword number and object code
2223 LCC22      inca                ; bump keyword count
2224          dec kwcount           ; have we reached the end of the list?
2225          bne LCBFC             ; brif not - check another keyword
2226          tst kwmatch           ; do we have a match?
2227          bne LCC2F             ; brif so
2228 LCC2B      ldd allones         ; flag error (-1)
2229 LCC2D      std ,s              ; save result
2230 LCC2F      puls a,b,x,y,u,pc    ; restore registers and return value, return
2231 LCC31      ldx #kwlist_dir     ; point to direction keywords

2232          bsr LCBEC             ; evaluate the specified keyword
2233          ble badcommandret      ; brif no matching keyword
2234          ldu #righthand        ; point to right hand contents
2235          cmpa #1               ; is it right hand wanted?
2236          beq LCC46             ; brif so - return pointer
2237          ldu #lefthand         ; point to left hand contents
2238          cmpa #0               ; is it left hand wanted?
2239          bne badcommandret      ; brif not - error
2240 LCC46      ldx ,u              ; fetch object pointer to X (and set Z if nothing)
2241          rts

2242 LCC49      pshs a,b,x,u        ; save coordinates and registers
2243          deca                  ; look at rooms to the NE, N, NW
2244          bsr LCC56
2245          inca                  ; look at rooms to the E, W, <here>
2246          bsr LCC56
2247          inca                  ; look at rooms to the SE, S, SW
2248          bsr LCC56
2249          puls a,b,x,u,pc       ; restore registers and return

```

```

2250 LCC56      pshs a,b                ; save coordinates
2251          decb                    ; look at room to W
2252          bsr LCC60
2253          incb                    ; look at room <here>
2254          bsr LCC60
2255          incb                    ; look at room E
2256          skip2                   ; skip next instruction
2257 LCC60      pshs a,b                ; save coordinates
2258          bsr LCC8E               ; did we fall off side of map?
2259          bne LCC6B               ; brif so
2260          bsr LCC7B               ; get pointer to room data
2261          lda ,x                  ; fetch room data
2262          skip2                   ; skip instruction
2263 LCC6B      lda #$ff               ; flag no tunnel
2264          sta ,u+                 ; save data for this room
2265          puls a,b,pc              ; save registers and return
2266 LCC71      getrandom              ; get a random number
2267          anda #$1f               ; convert it to 0-31
2268          tfr a,b                 ; save it
2269          getrandom              ; get another random number
2270          anda #$1f               ; also convert it to 0-31
2271 LCC7B      pshs a,b                ; save coordinates
2272          anda #$1f               ; force coordinates to range 0-31
2273          andb #$1f
2274          tfr d,x                 ; save coordinates for later
2275          ldb #32                  ; 32 rooms per row
2276          mul                     ; calculate row offset
2277          addd #mazedata          ; convert to absolute pointer
2278          exg d,x                 ; get pointer to X, get back coordinates
2279          abx                     ; add offset within row
2280          puls a,b,pc              ; restore coordinates and return pointer in X
2281 LCC8E      pshs a,b                ; save coordinates
2282          anda #$1f               ; modulo the Y coordinate
2283          cmpa ,s                  ; does it match?
2284          bne LCC9A               ; brif not - fell off side
2285          andb #$1f               ; modulo the X coordinate
2286          cmpb l,s                 ; does it match? (set flags)
2287 LCC9A      puls a,b,pc              ; return Z set if not falling off side of map
2288 ; This routine creates a maze for the specified level number.
2289 createmaze  ldx #mazedata          ; get start address to set to $ff
2290          ldu #mazedata+1024        ; get end address
2291          setblock                  ; go set block to $ff
2292          ldx #levelseeds           ; point to level seeds table
2293          ldb currentlevel          ; fetch current level
2294          abx                      ; offset into table (the seeds overlap!)
2295          ldd ,x++                 ; fetch first two bytes of level seed

```

```

2296          std randomseed                ; set random seed
2297          lda ,x                        ; fetch third byte of level seed
2298          sta randomseed+2              ; set random seed
2299          ldy #500                      ; dig out 500 rooms
2300          jsr LCC71                     ; fetch a random starting point
2301          std temploc                   ; save starting pointer
2302 LCCBB      getrandom                    ; get random number
2303          anda #3                       ; select only 4 directions
2304          sta curdir                     ; save direction we're going
2305          getrandom                      ; get random number
2306          anda #7                       ; convert to value from 1-8
2307          inca
2308          sta genpathlen                 ; save number of steps to dig out
2309          bra LCCD2                      ; go dig the tunnel
2310 LCCCA      ldd gencurcoord              ; get current coordinate
2311          std temploc                   ; save it as starting position
2312          dec genpathlen                 ; have we gone far enough?
2313          beq LCCBB                     ; brif so - select a new direction
2314 LCCD2      ldd temploc                  ; fetch maze coordinates
2315          jsr LD11B                      ; apply direction to coordinates
2316          bsr LCC8E                      ; did we fall off the side of the map?
2317          bne LCCBB                     ; brif so - select a new direction
2318          std gencurcoord                ; save new coordinate
2319          tst ,x                         ; is this room open?
2320          beq LCCCA                      ; brif so - move to next
2321          ldu #neighbourbuff             ; point to temporary storage area
2322          jsr LCC49                      ; set bytes to FF or 00 depending on whether the rooms in the
          3x3 area are open
2323          lda 3,u                        ; get W room
2324          adda ,u                        ; add data for NW room
2325          adda 1,u                       ; add data for N room
2326          beq LCCBB                     ; brif all open - get new direction
2327          lda 1,u                        ; get data for N room
2328          adda 2,u                       ; add data for NE room
2329          adda 5,u                       ; add data for E room
2330          beq LCCBB                     ; brif all open - get new direction
2331          lda 5,u                        ; get data for E room
2332          adda 8,u                       ; add data for SE room
2333          adda 7,u                       ; add data for S room
2334          beq LCCBB                     ; brif all open - get new direction
2335          lda 7,u                        ; get data for S room
2336          adda 6,u                       ; add data for SW room
2337          adda 3,u                       ; add data for W room
2338          beq LCCBB                     ; brif all open - get new direction
2339          clr ,x                         ; mark this room open
2340          leay -1,y                      ; have we dug out enough rooms?

```

```

2341          bne LCCCA                      ; brif not - keep digging
2342          clr temploc                     ; set coordinates to top left
2343          clr temploc+1
2344 LCD11      ldd temploc                     ; get current coordinates
2345          jsr LCC7B                         ; convert to pointer
2346          lda ,x                           ; get room data
2347          inca                             ; is ot open?
2348          beq LCD41                         ; brif not
2349          ldd temploc                     ; get coordinates
2350          ldu #neighbourbuff               ; point to temp area
2351          jsr LCC49                         ; calculate neighbors
2352          lda ,x                           ; get room data at current room
2353          ldb #$ff                         ; data for "no room"
2354          cmpb 1,u                         ; is there a room N?
2355          bne LCD2D                         ; brif so
2356          ora #3                           ; flag as no exit N
2357 LCD2D      cmpb 3,u                       ; is there a room W?
2358          bne LCD33                         ; brif so
2359          ora #$c0                         ; flag as no exit W
2360 LCD33      cmpb 5,u                       ; is there a room E
2361          bne LCD39                         ; brif so
2362          ora #$0c                         ; flag as no exit E
2363 LCD39      cmpb 7,u                       ; is there a room S?
2364          bne LCD3F                         ; brif so
2365          ora #$30                         ; flag as no exit S
2366 LCD3F      sta ,x                         ; save adjusted room data
2367 LCD41      ldb #32                       ; 32 rooms per row
2368          inc temploc+1                   ; bump X coordinate
2369          cmpb temploc+1                   ; did we wrap?
2370          bne LCD11                         ; brif not
2371          clr temploc+1                   ; reset to left edge
2372          inc temploc                     ; bump Y coordinate
2373          cmpb temploc                     ; did we wrap?
2374          bne LCD11                         ; brif not - fix another room's exits
2375          ldb #70                           ; create 70 doors
2376          ldu #doormasks                   ; pointer to routine to make a normal door
2377 LC056      bsr LCD6D                       ; go create a door
2378          decb                             ; are we done yet?
2379          bne LC056                         ; brif not
2380          ldb #$2d                         ; create 45 magic doors
2381          ldu #mdoormasks                   ; pointer to routine to make a magic door
2382 LCD60      bsr LCD6D                       ; go create a door
2383          decb                             ; done yet?
2384          bne LCD60                         ; brif not
2385          ldb clockctrs+2                   ; get number of times to spin the random number generator

```

(cycles once/minute)

```

2386 LCD67      getrandom                ; fetch a random number
2387           decb                      ; have we done enough randoms?
2388           bne LCD67                 ; brif not, do another
2389           rts                      ; return to caller
2390 LCD6D      pshs a,b,x,y,u           ; save registers
2391           ldy #dirmasks             ; point to direction masks
2392 LCD73      jsr LCC71                 ; get a random location
2393           std gencurcoord            ; save coordinates
2394           ldb ,x                    ; get room data at location
2395           cmpb #$ff                 ; is there a room?
2396           beq LCD73                 ; brif not - try again
2397           getrandom                 ; get random number
2398           anda #3                   ; normalize to direction
2399           sta curdir                ; save direction
2400           bitb a,y                  ; is there a door or wall at that direction?
2401           bne LCD73                 ; brif so - try again
2402           orb a,u                   ; mark the direction as having a door of desired type
2403           stb ,x                    ; save new room data
2404           ldd gencurcoord            ; get back coordinates
2405           jsr LD11B                 ; get pointer to neighbor
2406           ldb curdir                ; get direction back
2407           addb #2                   ; calculate opposite direction
2408           andb #3
2409           lda ,x                    ; get data at neighboring room
2410           ora b,u                   ; set it to the right type of door
2411           sta ,x                    ; save new neighbor data
2412           puls a,b,x,y,u,pc         ; restore data and return
2413 ; These are the random seeds for the level mazes. Note that the seeds overlap by two
2414 ; bytes. The actual seed values are:
2415 ; Level 1: 73c75d
2416 ; Level 2: c75d97
2417 ; Level 3: 5d97f3
2418 ; Level 4: 97f313
2419 ; Level 5: f31387
2420 levelseeds  fcb $73,$c7,$5d,$97,$f3,$13,$87
2421 dirmasks    fcb $03,$0c,$30,$c0      ; direction masks
2422 doormasks   fcb $01,$04,$10,$40      ; direction masks to create doors
2423 mdoormasks  fcb $02,$08,$20,$80      ; direction masks to create magic doors
2424 ; This routine draws the display for a scroll.
2425 ;
2426 ; If showseer is set to nonzero, it displays creature and object information (SEER SCROLL)
2427 ; otherwise it shows only the maze, holes, and player location (VISION SCROLL).
2428 ;
2429 ; temploc is used as a temporary scratch counter for displaying the maze itself.
2430 displayscroll ldu screendraw          ; point to screen we're using to draw on
2431           ldd #$1fff                 ; maximum X and Y coordinates

```

```

2432          std temploc          ; save coordinates
2433 LCDB9      ldd temploc          ; fetch current coordinates
2434          bsr LCE11              ; calculate absolute pointer to screen location
2435          jsr LCC7B              ; fetch pointer to room data
2436          clrb                  ; initialize to black
2437          lda ,x                ; fetch room data
2438          inca                  ; is it an empty room?
2439          bne LCDC7              ; brif not
2440          decb                  ; set to white
2441 LCDC7      lda #6              ; set 6 rows
2442 LCDC9      stb ,y              ; set a row
2443          leay $20,y            ; move to next row
2444          deca                  ; done all rows?
2445          bne LCDC9              ; brif not
2446          dec temploc+1          ; move left one space
2447          bpl LCDB9              ; brif not at left yet
2448          lda #$1f              ; max right coord
2449          sta temploc+1          ; reset X coordinate to far right
2450          dec temploc            ; move back a row
2451          bpl LCDB9              ; brif still in map
2452          tst showseer           ; are we showing creatures and objects?
2453          beq LCE2B              ; brif not
2454          clr objiterstart       ; start iteration from scratch
2455 LCDE3      jsr LCF63            ; go fetch object
2456          beq LCDF7              ; brif no more objects
2457          tst 5,x                ; is the object equipped/carried?
2458          bne LCDE3              ; brif so
2459          ldd 2,x                ; get coordinates of object
2460          bsr LCE11              ; get absolute address of location
2461          ldd #8                 ; object location symbol
2462          bsr LCE1D              ; display symbol
2463          bra LCDE3              ; go check another object
2464 LCDF7      ldx #creaturetab-17 ; point to creature table
2465 LCDFA      leax $11,x           ; move to next creature
2466          cmpx #mazedata         ; are we at the end of the creature table?
2467          beq LCE2B              ; brif so
2468          tst 12,x               ; is creature alive?
2469          beq LCDFA              ; brif not
2470          ldd 15,x               ; get current creature location
2471          bsr LCE11              ; turn location into pointer
2472          ldd #$1054             ; symbol for creature
2473          bsr LCE1D              ; go display symbol
2474          bra LCDFA              ; go check another creature
2475 LCE11      tfr d,y              ; save requested coordinates
2476          ldb #$c0               ; calculate row offset based on display height of 6 px
2477          mul                    ; now we have the offset from the start of the screen

```

```

2478          addd ,u                ; now D has the absolute address of the start of the line
2479          exg d,y                ; put pointer in Y and get back coordinates
2480          leay b,y                ; offset in the X direction for real pointer
2481          rts                    ; return to caller
2482 LCE1D      sta $20,y             ; set top row of symbol
2483          stb $40,y             ; set second row of symbol
2484          stb $60,y             ; set third row of symbol
2485          sta $80,y             ; set bottom row of symbol
2486 LCE2A      rts                    ; return to caller
2487 LCE2B      ldd playerloc        ; get current player position
2488          bsr LCE11             ; calculate absolute address
2489          ldd #$2418            ; bit patterns to create a *
2490          bsr LCE1D             ; go mark the player position
2491          ldx holetabptr        ; point to the hole table for this level
2492          bsr LCE38             ; go display holes going up then fall through for holes going
                                down
2493 LCE38      lda ,x+              ; get hole type flag
2494          bmi LCE2A             ; brif end of this table (return)
2495          ldd ,x++              ; get coordinates
2496          bsr LCE11             ; calculate absolute address
2497          ldd #$3c24            ; symbol for displaying a hole
2498          bsr LCE1D             ; go display symbol
2499          bra LCE38             ; go check another entry
2500 LCE47      pshs a,x             ; save registers
2501          ldx #LCF48            ; point to lighting level constants
2502          tst movehalf          ; is this a half step render?
2503          bne LCE5C             ; brif not
2504          leax >1,x             ; move ahead in the render scale constants
2505          tst movebackhalf      ; is it a half step back?
2506 LCE56      bne LCE5C             ; brif not
2507          leax >-11,x           ; move to backstep levels
2508 LCE5C      lda renderdist       ; get distance to render
2509          lda a,x               ; get scale factor for the distance
2510          sta horizscale        ; save horizontal scaling factor
2511          sta vertscale         ; save vertical scaling factor
2512          puls a,x,pc           ; restore registers and return
2513 ; This is the routine that shows the regular dungeon view.
2514 LCE66      cleargfx2            ; clear the graphics area
2515          clr renderdist        ; set render distance to immediate
2516          ldd playerloc         ; get player location
2517          std temploc           ; save current render location
2518 LCE6E      bsr LCE47            ; calculate scaling factor for current render location
2519          ldd temploc           ; fetch render location
2520          jsr LCC7B             ; get maze pointer
2521          lda ,x                ; get maze data for current location
2522          ldu #neighbourbuff    ; point to neighbor calculation buffer

```



```

2523                ldx #4                ; check four directions
2524 LCE7D          tfr a,b                ; save door info
2525                andb #3                ; check low 2 bits
2526                stb 4,u                ;= save twice so we can handle rotation without special cases
2527                stb ,u+                ;=
2528                lsra                    ;* shift room data to next direction
2529                lsra                    ;*
2530                leax -1,x                ; have we done all four directions?
2531                bne LCE7D                ; brif not
2532                ldb facing                ; get the direction we're facing
2533                ldu #neighbourbuff        ; point to neighbor table
2534                leau b,u                ; offset neighbor table
2535                ldy #LDBDE                ; point to direction rendering table (pointers to graphic
                elements)
2536 LCE96          lda ,y+                ; get table entry flag/direction
2537                bmi LCED8                ; brif end of table
2538                ldb a,u                ; get direction data
2539                lslb                    ; two bytes per door type
2540                cmpb #4                ; is it a magic door?
2541                bne LCEA9                ; brif not
2542                ldx b,y                ; fetch graphic pointer
2543                dec rendermagic            ; flag to render to magic light
2544                bsr LCECE                ; go draw the magic door
2545                ldb #6                ; change type to wall (invisible magic door)
2546 LCEA9          ldx b,y                ; get graphic
2547                bsr LCECE                ; draw the graphic
2548                leay 8,y                ; move to next table entry
2549                bra LCE96                ; go handle another direction

2550 LCEB1          rts                    ; return to caller
2551 LCEB2          tfr x,y                ; save graphic pointer
2552                tst b,u                ; is there a door in that direction?
2553                bne LCEB1                ; brif so
2554                addb facing                ; calculate real direction
2555                stb curdir                ; save real direction
2556                ldd temploc                ; fetch render location
2557                jsr LD11B                ; get new coordinates and room pointer
2558                jsr LCF82                ; get creature in room
2559                beq LCEB1                ; brif no creature in room
2560                exg x,y                ; save creature pointer in Y, get original graphic pointer

                back
2561 LCEC8          tst 2,y                ; is creature magical?
2562                beq LCECE                ; brif not - use physical ight
2563                dec rendermagic            ; render magic light
2564 LCECE          pshs u                ; save registers
2565                setlighting                ; set light level
2566                ldu screendraw            ; point to drawing screen

```

```

2567          drawgraphic          ; draw the selected graphic
2568          puls u,pc             ; restore registers and return
2569 LCED8      ldd temploc         ; get render location
2570          jsr LCF82              ; get creature in room
2571          beq LCEEB             ; brif no creature
2572          tfr x,y               ; save creature pointer
2573          ldb 13,y              ; get creature tpe
2574          lslb                  ; double it
2575          ldx #LDAA3             ; point to creature graphics table
2576          ldx b,x               ; get graphic data
2577          bsr LCEC8              ; go render graphic
2578 LCEEB      ldb #3              ; right hand
2579          ldx #LDCB0             ; point to graphic
2580          bsr LCEB2              ; go render graphic if there's a door
2581          ldb #1                ; left hand
2582          ldx #LDCB9             ; point to graphic
2583          bsr LCEB2              ; go render graphic if there's a door
2584          ldx #LDD3C             ; point to graphic
2585          ldd temploc           ; get current location
2586          jsr LCFE1              ; get hole information
2587          bmi LCF09              ; brif no hole
2588          ldx #LDCC2             ; point to graphic table for holes
2589          lsla                  ; two bytes per pointer entry
2590          ldx a,x               ; get actual graphic for the hole present
2591 LCF09      bsr LCECE            ; go render the graphic
2592          clr objiterstart       ; reset object iterator
2593 LCF0D      ldd temploc         ; get current room
2594          jsr LCF53              ; fetch next object on floor
2595          beq LCF24              ; brif no more objects
2596          lda 10,x               ; get object type
2597          lsla                  ; double it - two bytes per pointer entry
2598          ldx #LD9EE             ; point to object images
2599          ldx a,x               ; get correct graphic image
2600          dec rendermagic        ; set to render magic light
2601          bsr LCECE              ; render object in magic light (why??)
2602          bsr LCECE              ; render object in physical light
2603          bra LCF0D              ; go handle another object in the room
2604 LCF24      tst ,u              ; any door looking forward?
2605          bne LCF3D              ; brif so
2606          lda facing             ; get direction facing
2607          sta curdir             ; save direction going
2608          ldd temploc           ; get current direction
2609          jsr LD11B              ; get pointer in that direction
2610          std temploc           ; save new location
2611          inc renderdist         ; bump render distance (next room going forward)
2612          lda renderdist         ; get distance

```

```

2613      cmpa #9                                ; is it 9 steps out?
2614      lble LCE6E                              ; brif 9 or less - render another room
2615 LCF3D      rts                              ; return to caller
2616 ; These are the scale factors used for rendering rooms.
2617      fcb $c8,$80,$50,$32,$1f,$14,$0c,$08,$04,$02
2618 LCF48      fcb $ff,$9c,$64,$41,$28,$1a,$10,$0a,$06,$03,$01
2619 LCF53      bsr LCF63                        ; fetch next object in iteration
2620      beq LCF62                              ; brif no object
2621      cmpd 2,x                              ; is object at desired location
2622      bne LCF53                              ; brif not - try again
2623      tst 5,x                              ; is object in inventory?
2624      bne LCF53                              ; brif so - not in room
2625      andcc #$fb                            ; clear Z for found
2626 LCF62      rts                              ; return to caller
2627 LCF63      pshs a                          ; save register
2628      lda currentlevel                      ; fetch current level
2629      ldx objiterptr                        ; fetch object pointer
2630      tst objiterstart                      ; are we starting at beginning?
2631      bne LCF72                              ; brif not
2632      ldx #objecttab-14                     ; point to start of table
2633      dec objiterstart                      ; mark not at beginning any more
2634 LCF72      leax 14,x                       ; move to next object
2635      stx objiterptr                        ; save object pointer for iteration
2636      cmpx objectfree                      ; are we at the end of the table?
2637      beq LCF80                              ; brif so - return
2638      cmpa 4,x                              ; is the object on this level?
2639      bne LCF72                              ; brif not - look for another object
2640      andcc #$fb                            ; turn off Z flag for object found
2641 LCF80      puls a,pc                       ; restore registers and return
2642 LCF82      ldx #creaturetab-17             ; point to creature table
2643 LCF85      leax $11,x                      ; move to next entry
2644      cmpx #mazedata                       ; end of table?
2645      beq LCF96                              ; brif so
2646      cmpd 15,x                            ; is the creature in the desired maze location
2647      bne LCF85                              ; brif not - check another
2648      tst 12,x                              ; is the creature alive?
2649      beq LCF85                              ; brif not - check another
2650 LCF96      rts                              ; return to caller, Z clear if we found a creature
2651 LCF97      pshs a,b,x                      ; save registers
2652 LCF99      jsr LCC71                       ; get a starting point for the creature
2653      std ,s                                ; save resulting location
2654      lda ,x                                ; fetch room data at location
2655      inca                                  ; is it a room?
2656      beq LCF99                              ; brif not - try again
2657      puls a,b,x,pc                         ; restore registers, return value, and return
2658 ; Create a creature

```

```

2659 LCFA5      pshs a,b,x,y,u      ; save registers
2660 LCFA7      ldu #creaturetab-17  ; point to creature table
2661 LCFAA      leau $11,u          ; move to next entry
2662           tst 12,u              ; is creature alive?
2663           bne LCFAA             ; brif not - look for another entry
2664           dec 12,u              ; mark creature alive
2665           sta 13,u              ; set creature type as requested
2666           ldb #8                ; 8 bytes per creature data
2667           mul                   ; get offset into creature data table
2668           addd #LDABB           ; now we have a pointer to this creatures data
2669           tfr d,y               ; put creature data pointer in Y (source pointer)
2670           tfr u,x               ; put creature slot into X (destination pointer)
2671           lda #8                ; there are 8 bytes for each creature info
2672           jsr LC04B             ; copy data into this creature slot
2673 LCFC4       bsr LCF97           ; get a location to start the creature in
2674           bsr LCF82             ; check if there's already a creature there
2675           bne LCFC4             ; brif so - try again
2676           std 15,u              ; put the creature there
2677           tfr u,x               ; save creature pointer
2678           jsr LC25C             ; get scheduling entry
2679           stx 5,u               ; save creature pointer in scheduling entry
2680           ldd #LD041            ; creature scheduling handler
2681           std 3,u               ; set handler for this entry
2682           lda 6,x               ; get scheduling ticks for creature
2683           ldb #4                ; put in 10Hz list
2684           jsr LC21D             ; go add to scheduling list
2685           puls a,b,x,y,u,pc      ; restore registers and return
2686 LCFE1       pshs a,b,x,u       ; save registers
2687           ldu holetabptr        ; point to hole table for this level (going up)
2688           bsr LCFF2             ; see if there is a hole for this room
2689           tsta                  ; is there a hole?
2690           bpl LCFEE             ; brif so - return info
2691           bsr LCFF2             ; check for this level going down
2692           adda #2                ; flag the hole as down
2693 LCFEE       sta ,s              ; save hole information for return
2694           puls a,b,x,u,pc        ; restore registers and return
2695 LCFF2       lda ,u+             ; fetch hole flags
2696           bmi LCFFC             ; brif end of table entries
2697           ldx ,u++              ; get location for the hole
2698           cmpx 2,s               ; does it match the current location?
2699           bne LCFF2             ; brif not - try another entry
2700 LCFFC       rts                 ; return to caller
2701 ; This is the "hole/ladder" table. Each entry is suffixed by $80. Each set specifies the
2702 ; holes and ladders between two levels. The first is between levels 1 and 2. The second is
2703 ; between levels 2 and 3. And so on. You will not that the table includes references to
2704 ; level 0 (above the dungeon) and level 6 (below the dungeon) - they are simply empty

```

```

2705 ; table entries which prevents having to have special cases to handle them.
2706 holetab          fcb $80                      ; marker for end of "level 0" to level 1
2707                fcb 1,0,23
2708                fcb 0,15,4
2709                fcb 0,20,17
2710                fcb 1,28,30
2711                fcb $80                      ; marker for end of level 1-2
2712                fcb 1,2,3
2713                fcb 0,3,31
2714                fcb 0,19,20
2715                fcb 0,31,0
2716                fcb $80                      ; marker for end of level 2-3
2717                fcb $80                      ; marker for end of level 3-4
2718                fcb 0,0,31
2719                fcb 0,5,0
2720                fcb 0,22,28
2721                fcb 0,31,16
2722                fcb $80                      ; marker for end of level 4-5
2723                fcb $80                      ; marker for end of level 5-6
2724 ; This is the routine that adjusts the creature counts for handling retreats. It is called every
2725 ; five minutes. If there are less than 32 creatures on the current level, it will pick a random
2726 ; creature (club giants through galdrogs) and bump the count that will be spawned the next time
2727 ; the level is entered. This *only* applies to the level currently being played.
2728 ;
2729 ; It's worth noting that this can ONLY affect levels 1, 2, and 3 because there is no way to return
2730 ; to levels 4 (no holes up from 5) which means level 5 can only be entered once.
2731 LD027             ldx creaturecntptr          ; point to creature counts for this level
2732                 ldb #11                      ; maximum creature number
2733                 clra                          ; initialize count
2734 LD02C             adda b,x                    ; add the number of this creature
2735                 decb                          ; at end of creature list?
2736                 bpl LD02C                     ; brif not
2737                 cmpa #32                      ; do we have the maximum number of creatures yet?
2738                 bhs LD03D                     ; brif so
2739                 getrandom                     ; get a random value
2740                 anda #7                      ; only interested in spawning one of 8 creatures
2741                 adda #2                      ; offset above vipers
2742                 inc a,x                      ; bump creature count for that type
2743 LD03D             ldd #$0508                 ; reschedule for 5 minutes
2744                 rts                          ; return to caller
2745 ; This is the routine that handles creature movement, etc.
2746 LD041             ldy 5,u                    ; get creature data pointer
2747                 tst creaturefreeze           ; are creatures frozen (after the Wizard is beaten)?
2748                 bne LD06A                     ; brif so
2749                 ldb 12,y                     ; is the creature alive?
2750                 bne LD04D                     ; brif so

```

```

2751                rts                ; return to caller
2752 LD04D          lda 13,y            ; get the creature type
2753                cmpa #6             ; is it a scorpion?
2754                beq LD06D           ; brif so
2755                cmpa #10            ; is it the wizard's image or wizard?
2756                bge LD06D           ; brif so
2757                ldd 15,y            ; fetch room location
2758                clr objiterstart    ; reset object iterator
2759                jsr LCF53            ; fetch first object in room
2760                beq LD06D           ; brif no object in room
2761                ldd 8,y             ; get creature inventory pointer
2762                stx 8,y             ; save room object as head of inventory list
2763                std ,x              ; save inventory list as next item
2764                dec 5,x             ; mark object as carried
2765                updatedungeon       ; update the dungeon view
2766 LD06A          jmp LD103           ; go reschedule
2767 LD06D          ldd 15,y            ; get cerature location
2768                cmpd playerloc      ; is it in the room with the player?
2769                bne LD0B2           ; brif not
2770                lda 13,y            ; get creature type
2771                ldb #$ff            ; maximum sound volume
2772                playsound           ; go make the creature sound (always makes on attack)
2773                ldd #$8080          ; base defense modifiers
2774                ldx lefthand        ; get object in left hand
2775                bsr LD09E           ; set modifiers if shield
2776                ldx righthand       ; get object in right hand
2777                bsr LD09E           ; set modifiers if shield
2778                sta magicdef        ; save magical defense value for player
2779                stb physdef         ; save physical defense value for player
2780                tfr y,x             ; put the creature as the attacker
2781                ldu #powerlevel     ; put the player as defender
2782                jsr attack          ; calculate an attack
2783                bmi LD099           ; brif attack failed
2784                playsoundimm $13    ; play the hit sound
2785                jsr damage          ; go damage the player
2786 LD099          checkdamage         ; check damage levels
2787                jmp LD10F           ; go reschedule
2788 LD09E          pshs a,b,x          ; save registers
2789                beq LD0B0           ; brif no object
2790                lda 10,x            ; get object type
2791                cmpa #3             ; is it a shield?
2792                bne LD0B0           ; brif not
2793                ldx 6,x             ; get magical and physical defense values
2794                cmpx ,s             ; is it higher (magic has precedence)
2795                bhs LD0B0           ; brif so - less good
2796                stx ,s             ; save new defense multipliers

```

```

2797 LD0B0      puls a,b,x,pc      ; restore registers and return
2798 LD0B2      cmpa playerloc     ; are we in the same horizontal line as the player?
2799           bne LD0C3          ; brif not
2800           lda 16,y            ; get vertical coordinate for creature
2801           ldb #1             ; assume east
2802           suba playerloc+1     ; calculate distance to player
2803           bmi LD0D0          ; brif negative - player is east
2804           ldb #3             ; player is actually west
2805           bra LD0D0          ; go check movement
2806 LD0C3      ldd 15,y           ; get creature location
2807           cmpb playerloc+1    ; are we in the same column as the player?
2808           bne LD0E4          ; brif not
2809           ldb #2             ; assume south
2810           suba playerloc     ; calculate difference to player
2811           bmi LD0D0          ; brif player is south
2812           clrb              ; set north
2813 LD0D0      stb curdir         ; save direction
2814           ldd 15,y           ; get creature location
2815 LD0D4      bsr LD136          ; calculate new coordinates
2816           bne LD0E4          ; brif not a room
2817           cmpd playerloc     ; is the new room the player's place?
2818           bne LD0D4          ; brif not
2819           ldb curdir         ; get direction to move
2820           stb 14,y           ; set last movement direction to player direction
2821           clrb              ; select a last ditch direction
2822           bra LD101          ; go try the movement and continue
2823 LD0E4      ldx #LD114        ; point to direction selections
2824           getrandom          ; fetch a random value
2825           tsta              ; set flags
2826           bmi LD0EE          ; brif negative
2827           leax 3,x           ; select alternative direction sets
2828 LD0EE      anda #3           ; normalize direction to 0-3
2829           bne LD0F4          ; brif nonzero
2830           leax 1,x           ; move to next value
2831 LD0F4      lda #3            ; try 3 times for a movement
2832 LD0F6      ldb ,x+           ; get direction modifier
2833           bsr LD14F          ; go handle movement
2834           beq LD103          ; brif movement succeeded
2835           deca              ; have we tried enough times?
2836           bne LD0F6          ; brif not
2837           ldb #2            ; try one more last ditch option
2838 LD101      bsr LD14F          ; do movement
2839 LD103      lda 6,y           ; get movement tick rate
2840           ldx 15,y           ; get creature location
2841           cmpx playerloc     ; does it match the player?
2842           bne LD111          ; brif not - use movement rate

```

```

2843      updatedungeon                ; update the dungeon display immediately
2844      clr dungeonchg                ; mark dungeon update not required
2845 LD10F      lda 7,y                  ; get attack tick rate
2846 LD111      ldb #4                  ; and schedule for the 10Hz timer
2847      rts                          ; return to caller
2848 LD114      fcb $00,$03,$01,$00,$01,$03,$00 ; direction rotations for movement choices
2849 LD11B      pshs a,b                ; save coordinates
2850      ldb curdir                    ; get direction to move
2851      andb #3                      ; force it to 0-3
2852      lslb                          ; two bytes per direction adjuster
2853      ldx #LD12E                    ; point to direction adjusters
2854      ldd b,x                      ; get adjuster
2855      adda ,s+                      ; apply north/south adjustment
2856      addb ,s+                      ; apply east/west adjustment
2857      jmp LCC7B                    ; convert to pointer in X
2858 LD12E      fdb $ff00               ; move north (-1, 0)
2859      fdb 1                        ; move east (0, +1)
2860      fdb $100                     ; move south (+1, 0)
2861      fdb $ff                      ; move west (0, -1)
2862 LD136      pshs a,b,x,y,u          ; save registers
2863      bsr LD11B                    ; calculate new coordinates
2864      jsr LCC8E                    ; check if we fell off map
2865      bne LD14D                    ; brif so
2866      tfr d,u                      ; save coordinates for later
2867      lda ,x                      ; get data at the new location
2868      inca                          ; is it a room?
2869      beq LD14C                    ; brif not
2870      stu ,s                      ; save new coordinates for return
2871      stx 2,s                      ; save new room pointer
2872      lda #1                      ; set so we get Z=1 on return
2873 LD14C      deca                    ; set flags for success/fail
2874 LD14D      puls a,b,x,y,u,pc       ; restore registers and return
2875 LD14F      pshs a,b,x             ; save registers
2876      addb 14,y                   ; add selected rotation to current movement direction
2877      andb #3                      ; normalize to 0-3
2878      stb curdir                    ; save new direction
2879      ldd 15,y                     ; get creature location
2880      bsr LD136                    ; calculate new coordinates
2881      bne LD199                    ; brif no room there
2882      jsr LCF82                    ; get creature in room
2883      bne LD199                    ; brif there's a creature there - can't go
2884      std 15,y                     ; save new creature location
2885      ldb curdir                    ; get direction
2886      stb 14,y                     ; save as last moved direction
2887      ldd 15,y                     ; get new location
2888      suba playerloc                ; get distance from player (Y)

```



```

2889          bpl LD16F                      ; brif positive
2890          nega                            ; invert msb (absolute value)
2891 LD16F      subb playerloc+1              ; get distance from player (X)
2892          bpl LD174                      ; brif positive
2893          negb                            ; invert lsb (absolute value)
2894 LD174      stb accum0                   ; save X distance
2895          cmpa accum0                    ; is the Y distance more?
2896          bge LD17C                      ; brif so
2897          exg a,b                        ; use the Y distance then
2898 LD17C      sta accum0                   ; save calculated distance
2899          cmpa #8                        ; more than 8 rooms away in long distance?
2900          bgt LD198                      ; brif so
2901          cmpb #2                        ; more than 2 rooms away in short distance?
2902          bgt LD198                      ; brif so
2903          getrandom                       ; get a random value
2904          bita #1                        ; do we need to make a sound?
2905          beq LD196                      ; brif we won't make a sound
2906          lda accum0                     ; get distance
2907          ldb #$1f                       ; multiplier for distance
2908          mul                            ; calculate distance volume modifier
2909          comb                            ; invert it so closer is louder
2910          lda 13,y                       ; get creature number
2911          playsound                       ; go make the creature's sound
2912 LD196      dec dungeonchg               ; mark dungeon update required
2913 LD198      clra                         ; set Z for movement happened
2914 LD199      puls a,b,x,pc                ; restore registers and return
2915 ; This is the routine that ticks down the torch.
2916 LD19B      ldu curtorch                 ; get currently burning torch
2917          beq LD1BC                      ; brif no torch in use
2918          lda 6,u                         ; get remaining torch life
2919          beq LD1BC                      ; brif already empty
2920          deca                            ; reduce time remaining
2921          sta 6,u                         ; update object data
2922          cmpa #5                        ; is it 5 minutes left?
2923          bgt LD1B0                      ; brif more
2924          ldb #$18                       ; object type "DEAD TORCH"
2925          stb 9,u                         ; set torch to DEAD TORCH
2926          clr 11,u                       ; mark as fully revealed
2927 LD1B0      cmpa 7,u                     ; is time remaining less than physical light strength?
2928          bge LD1B6                      ; brif not
2929          sta 7,u                         ; tick down physical light strength
2930 LD1B6      cmpa 8,u                     ; is time remaining less than magical light strength?
2931          bge LD1BC                      ; brif not
2932          sta 8,u                         ; tick down magical light strength
2933 LD1BC      dec dungeonchg               ; mark update to dungeon required
2934          ldd #$0108                     ; reschedule for one minute

```

```

2935             rts                                ; return to caller
2936 ; This is the routine that periodically updates the dungeon display (or scroll). It does not update
2937 ; unless something has marked the display changed OR a scroll is being displayed. It is called twice
2938 ; per second.
2939 LD1C2         tst dungeonchg                    ; check if we need to update dungeon display
2940             bne LD1CD                          ; brif so
2941             ldx #displayscroll                  ; are we displaying a scroll?
2942             cmpx displayptr
2943             bne LD1D1                          ; brif not
2944 LD1CD         clr dungeonchg                    ; mark update not required
2945             updatedungeon                      ; update dungeon display
2946 LD1D1         ldd #$0304                        ; reschedule check for 0.5 seconds
2947             rts                                ; return to caller
2948
2949 LD1D5         clra                              ; set NULL value
2950             clrb
2951             subd damagelevel                    ; subtract it from the current damage level
2952             jsr asrd6                          ; shift right 6 bits (divide by 64)
2953             addd damagelevel                    ; reduce damage level by 1/64 of original damage level
2954             bgt LD1E2                          ; brif new damage level > 0
2955             clra                              ; minimize damage level at 0
2956             clrb
2957 LD1E2         std damagelevel                    ; save new damage level
2958             checkdamage                        ; check damage level and calculate ticks until next recovery
2959 run
2959             lda heartticks                      ; get ticks to reduce damage (heart rate)
2960             ldb #2                             ; requeue in the 60Hz ticker
2961             rts                                ; return to caller
2962 ; This routine handles the keyboard input.
2963 LD1EB         tst waitnewgame                    ; are we waiting for a new game?
2964             bne LD21B                          ; brif so
2965 LD1EF         jsr readkeybuf                    ; get a key from buffer
2966             tsta                              ; did we get something?
2967             beq LD248                          ; brif not
2968             tst nokeyboard                      ; is keyboard disabled?
2969             bne LD1EF                          ; brif so - keep draining buffer
2970             cmpa #$20                          ; is it a space?
2971             beq LD215                          ; brif so
2972             ldb #$1f                          ; value for CR
2973             cmpa #$0d                          ; is it CR?
2974             beq LD212                          ; brif so
2975             ldb #$24                          ; value for BS
2976             cmpa #8                            ; is it BS?
2977             beq LD212                          ; brif so
2978             clrb                              ; value for nothing (space)
2979             cmpa #$41                          ; is it a letter?

```

```

2980      blo LD212                                ; brif below uppercase alpha
2981      cmpa #$5a                                ; is it still a letter?
2982      bls LD215                                ; brif uppercase alpha
2983 LD212   tfr b,a                              ; save calculated code
2984      skip2                                    ; skip instruction
2985 LD215   anda #$1f                             ; normalize down to 0...31
2986      bsr LD24C
2987      bra LD1EF                                ; go handle another character
2988 LD21B   ldy demoseqptr                       ; fetch pointer to command sequence
2989      ldb ,y+                                  ; do we have a command to do?
2990      bpl LD229                                ; brif so
2991      delay                                    ; wait for a bit
2992      delay                                    ; wait for a bit more
2993      jmp START                                ; go start over again with the splash and demo
2994 LD229   ldx ,y++                              ; get pointer to the word
2995      ldu #cmddecodebuff                       ; point to command decode buffer
2996      decodestr                                ; decode the keyword
2997      leau 1,u                                  ; move past the "object type" flag
2998      delay                                    ; wait a bit
2999      skip2                                    ; skip next instruction
3000 LD235   bsr LD24C                              ; go handle input character
3001      lda ,u+                                  ; fetch a character from the decoded string
3002      bpl LD235                                ; brif not end of string
3003      clra                                      ; code for a space
3004      bsr LD24C                              ; go handle input character
3005      decb                                      ; have we consumed all the words in this command?
3006      bne LD229                                ; brif not - get another
3007      lda #$1f                                 ; code for carriage return
3008      bsr LD24C                                ; add character to buffer and process if needed
3009      sty demoseqptr                           ; save new command stream pointer
3010 LD248   ldd #$0102                            ; reschedule for next tick
3011      rts                                      ; return to caller
3012 LD24C   pshs a,b,x,y,u                       ; save registers
3013      tst hidestatus                           ; are we starting a new command string?
3014      bne LD256                                ; brif not
3015      resetdisplay                             ; clear command area, reset status, and redisplay dungeon
3016      showprompt                               ; show the prompt
3017 LD256   ldu linebuffptr                      ; get input buffer pointer
3018      cmpa #$1f                                 ; end of line?
3019      beq LD26F                                ; brif so
3020      cmpa #$24                                 ; BS?
3021      beq LD27D                                ; brif so
3022      renderchar                               ; render the character
3023      sta ,u+                                  ; save in buffer
3024      ldx #LC67C                              ; point to cursor string
3025      renderstr                               ; go render the cursor

```

```

3026          cmpu #linebuffend          ; is the buffer full?
3027          bne LD2B4                    ; brif not
3028 LD26F      clra                       ; make a space
3029          renderchar                   ; render it
3030          ldd allones                  ; get end of string marker
3031          std ,u++                     ; save in buffer
3032          ldu #linebuff                ; reset buffer pointer to start of line
3033          stu linebuffptr              ; save new buffer pointer
3034          bra LD292                    ; go process command
3035 LD27D      cmpu #linebuff             ; are we at the start of the line?
3036          beq LD2B4                    ; brif so - BS does nothing
3037          leau -1,u                    ; move buffer pointer back
3038          ldx #LD28C                   ; pointer to SPACE BS BS _ BS
3039          renderstr                     ; display the backspace string
3040          bra LD2B4                    ; get on with things
3041 LD28C      fcb $00,$24,$24,$1c,$24,$ff ; unpacked SPACE BS BS _ BS string
3042 LD292      ldx #kwlist_cmd            ; point to command list
3043          jsr LCBEC                     ; look up word in command list
3044          beq LD2A7                    ; brif nothing to match
3045          bpl LD2A1                    ; brif found
3046          jsr badcommand                ; show bad command string
3047          bra LD2A7                    ; go on with new command
3048 LD2A1      lslda                      ; two bytes per jump table entry
3049          ldx #LD9D0                   ; point to command jump table
3050          jsr [a,x]                     ; go handle command
3051 LD2A7      ldu #linebuff              ; start of command buffer
3052          tst hidestatus                 ; have we been told to start a new command stream?
3053          beq LD2B4                    ; brif so - don't display prompt
3054          tst nokeyboard                 ; is keyboard disabled?
3055          bne LD2B4                    ; brif so - no prompt
3056          showprompt                    ; show a new prompt
3057 LD2B4      stu linebuffptr            ; save new buffer pointer
3058          puls a,b,x,y,u,pc             ; restore registers and return
3059 cmd_attack jsr LCC31                  ; get pointer to specified hand
3060          ldu ,u                        ; fetch item in specified hand
3061          bne LD2C2                    ; brif item there
3062          ldu #emptyhand                ; point to data for empty hand
3063 LD2C2      tfr u,y                    ; save object data pointer
3064          lda 12,u                      ; fetch magical offense value
3065          sta magicoff                  ; save for combat calculations
3066          lda 13,u                      ; fetch physical offense value
3067          sta physoff                   ; save for combat calculations
3068          adda magicoff                 ; calculate sum of magical and physical damage
3069          rora                          ; * divide by 8
3070          lsra                          ; *
3071          lsra                          ; *

```

```

3072          ldx powerlevel          ; fetch current player power
3073          jsr applyscale          ; apply the scale factor calculated above
3074          addd damagelevel        ; apply the wielding cost to play damage
3075          std damagelevel        ; save new damage value
3076          lda 10,u               ; get object type
3077          adda #12                ; offset into sound table
3078          ldb #$ff               ; set full volume
3079          playsound              ; play the attack sound for the object
3080          lda 9,u                ; get object subtype
3081          cmpa #$13              ; is it less than "ENERGY"?
3082          blt LD2F7              ; brif so - not an expiring ring
3083          cmpa #$15              ; is it more than "FIRE"?
3084          bgt LD2F7              ; brif so - not an expiring ring
3085          dec 6,u                ; count down ring usages
3086          bne LD2F7              ; brif not used up
3087          lda #$16               ; type for "GOLD"
3088          sta 9,u                ; set to GOLD ring
3089          jsr LD638              ; update object stats appropriately
3090 LD2F7          ldd playerloc      ; get current location in dungeon
3091          jsr LCF82              ; find creature in the room
3092          beq LD375              ; brif no creature
3093          ldu #powerlevel        ; point to player power level
3094          exg x,u                ; swap player and creature pointers
3095          lda 10,y               ; fetch object type
3096          cmpa #1                ; is it a ring?
3097          beq LD31F              ; go do successful attack if so - rings never miss
3098          jsr attack             ; calculate if attack succeeds (attacker in X, defender in U)
3099          bmi LD375              ; brif attack fails

3100          ldy curtorch          ; do we have a torch burning?
3101          beq LD319              ; brif not
3102          lda 9,y                ; get torch type
3103          cmpa #$18              ; is it "DEAD"?
3104          bne LD31F              ; brif not
3105 LD319          getrandom         ; get random number
3106          anda #3                ; 1 in 4 chance of a hit in the dark
3107          bne LD375              ; brif we didn't hit
3108 LD31F          playsoundimm $12 ; play the "HIT" sound
3109          renderstrimm          ; display the "!!!" for a successful hit
3110          fcb $16,$f7,$b0        ; packed "!!!" string
3111          jsr damage             ; calculate damage, apply to victim
3112          bhi LD375              ; brif not dead
3113          leax 8,u               ; point to inventory head pointer
3114 LD32E          ldx ,x            ; get next inventory item
3115          beq LD33A              ; brif end of inventory
3116          clr 5,x               ; mark item as on the floor
3117          ldd 15,u               ; get location of creature

```

```

3118          std 2,x                      ; put the object there
3119          bra LD32E                     ; go process next inventory item
3120 LD33A      ldx creaturecntptr          ; point to creature count table for this level
3121          ldb 13,u                      ; get type of creature killed
3122          dec b,x                       ; reduce number of this creature type
3123          clr 12,u                      ; flag creature as dead
3124          updatedungeon                 ; update the dungeon display
3125          playsoundimm $15              ; play the "kill" sound
3126          ldd ,u                       ; fetch creature power level
3127          bsr asrd3                     ; divide by 8
3128          addd powerlevel               ; add gained power to current power level
3129          bpl LD351                     ; brif power level did not overflow
3130          lda #$7f                      ; maximize power level at 32767
3131 LD351      std powerlevel              ; save adjusted power level for player
3132          lda 13,u                      ; get the dead creature type
3133          cmpa #10                      ; is dead creature wizard's image?
3134          beq LD386                     ; brif so - do the annoyed wizard
3135          cmpa #11                      ; is dead creature the wizard?
3136          bne LD375                     ; brif not
3137          dec creaturefreeze           ; stop the creatures
3138          ldd #$713                     ; constants for physical light 7, magical light 19
3139          std baselight                 ; set base light level in dungeon
3140          ldx #objecttab+14             ; pointer to second object slot in object table
3141          stx objectfree                 ; mark end of object table at just past first object
3142          ldd zero                      ; NULL pointer
3143          std backpack                   ; mark backpack empty
3144          std curtorch                   ; mark no torch burning
3145          std righthand                 ; mark right hand empty
3146          std lefthand                  ; mark left hand empty
3147          resetdisplay                  ; reset display and show dungeon
3148 LD375      checkdamage                 ; update the damage situation
3149 ; The following are pointless in this routine - we're returning from a command and D is zero anyway!
3150 asrd7      asra                         ; enter here to do an arithmetic right shift 7 bits
3151          rorb
3152 asrd6      asra                         ; enter here to do an arithmetic right shift 6 bits
3153          rorb
3154 asrd5      asra                         ; enter here to do an arithmetic right shift 5 bits
3155          rorb
3156 asrd4      asra                         ; enter here to do an arithmetic right shift 4 bits
3157          rorb
3158 asrd3      asra                         ; enter here to do an arithmetic right shift 3 bits
3159          rorb
3160          asra
3161          rorb
3162          asra
3163          rorb

```

```

3164             rts                               ; return to caller
3165 LD386        ldx #img_wizard                   ; point to Wizard graphic
3166             fadeinclrst                         ; fade in the wizard
3167             renderstrimpp                       ; display "ENOUGH! I TIRE OF THIS PLAY..."
3168             fcb $ff,$c0,$57,$3e                 ; packed string "ENOUGH! I TIRE OF THIS PLAY..."
3169             fcb $a7,$46,$c0,$90
3170             fcb $51,$32,$28,$1e
3171             fcb $60,$51,$09,$98
3172             fcb $20,$c0,$e7,$de
3173             fcb $f0
3174             renderstrimpp                       ; also display "PREPARE TO MEET THY DOOM!!!"
3175             fcb $e8,$00,$08,$48                 ; packed string "PREPARE TO MEET THY DOOM!!!"
3176             fcb $b0,$0c,$8a,$0a
3177             fcb $3c,$0d,$29,$68
3178             fcb $0a,$23,$20,$23
3179             fcb $de,$dd,$ef,$60
3180             delay                               ; delay a bit
3181             ldu curtorch                        ; fetch current torch
3182             stu backpack                        ; put it in the backpack
3183             beq LD3C4                           ; brif no torch
3184             clra                               ; make sure the torch is the only thing in the backpack
3185             clrb
3186             std ,u
3187 LD3C4        ldd #200                          ; set player carry weight to 200
3188             std carryweight
3189             lda #3
3190             createlevel
3191             jsr LCF97
3192             std playerloc
3193             fadeout                             ; fade out the wizard
3194             resetdisplay
3195             rts
3196 ; Calculate the probability of a successful hit.
3197 ; Enter with the attacker info pointed to be X and the defender data pointed to by U.
3198 ;
3199 ; It first does the following calculation:
3200 ; MAX(15-(4(DPOW-DDAM)/APOW),0)
3201 ; 4(DPOW-DDAM)/APOW yields a fraction which is < 4 if the defender's remaining health is
3202 ; less than the attacker's power or > 4 if the defender's remaining health is greater
3203 ; than the attacker's power. This ranges from 0% to 375% in steps of 25%.
3204 ; This result is subtracted from 15 so that low numbers mean the attacker relatively weaker
3205 ; and higher numbers mean the attacker is relatively stronger. The final range is from 0
3206 ; (where the defender is much stronger than the attacker) to 15 where the attacker is very
3207 ; much stronger than the defender.
3208 ;
3209 ; These values are converted to a signed 16 bit number. Then an 8 bit unsigned random number

```

```

3210 ; is added to the result. Finally, 127 is subtracted. If the final result is < 0, then the
3211 ; attack fails. Otherwise, the attack succeeds.
3212 ;
3213 ; The following chart gives calculation results. V is the result of MAX(...) calculation
3214 ; above. Pb is the base value calculated by the routine. Rl is the low end of the range
3215 ; of the result once the random number is applied and the 127 is subtracted. Rh is the
3216 ; high end of the range. Finally, P% is the chance of a successful hit for that result.
3217 ;
3218 ; V      Pb      Rl      Rh      P%
3219 ; 0      -75     -202    53      21.1
3220 ; 1      -50     -177    78      30.9
3221 ; 2      -25     -152    103     40.6
3222 ; 3       0      -127    128     50.4
3223 ; 4       10     -117    138     54.3
3224 ; 5       20     -107    148     58.2
3225 ; 6       30     -97     158     62.1
3226 ; 7       40     -87     168     66.0
3227 ; 8       50     -77     178     69.9
3228 ; 9       60     -67     188     73.8
3229 ; 10      70     -57     198     77.7
3230 ; 11      80     -47     208     81.6
3231 ; 12      90     -37     218     85.5
3232 ; 13     100     -27     228     89.5
3233 ; 14     110     -17     238     93.4
3234 ; 15     120      -7     248     97.3
3235 ;
3236 ; As you can see, the lower 4 values are on a steeper slope than the remaining values.
3237 ; Otherwise, the scale is perfectly linear. Also, the worst chance of success, no matter
3238 ; how overmached, is 21.1%. The best chance, no matter how much stronger the attacker,
3239 ; is less than 100%.
3240 attack          pshs a,b,x,u          ; save registers
3241                  lda #15              ; maximum value of the V calculation
3242                  sta accum0           ; initialize V accumulator
3243                  ldd ,u               ; get victim power level
3244                  subd 10,u            ; get difference between that and victim damage level (health)
3245                  jsr LCA12            ; multiply difference by 4
3246 LD3E4           subd ,x              ; subtract attackers power
3247                  bcs LD3EC           ; brif we wrapped - we have our quotient
3248                  dec accum0          ; count down quotient
3249                  bne LD3E4           ; brif we haven't counted down to nothing
3250 LD3EC           ldb accum0          ; get result (V as above)
3251                  subb #3             ; one of first three values?
3252                  bpl LD3FB           ; brif not
3253                  negb                ; now 0 became 3, 1 became 2, and 2 became 1
3254                  lda #$19           ; * multiply by factor (25)
3255                  mul                 ; *

```



```

3256                jsr LCA99                      ; negate result (-75, -50, and -25)
3257                bra LD3FE                      ; calculate attack
3258 LD3FB          lda #10                       ; * multiply by factor (10) (all others are linear going up by
10
3259                mul                            ; * for each step
3260 LD3FE          std ,--s                        ; save probability base
3261                getrandom                      ; get a random value
3262                tfr a,b                        ; save random value
3263                clra                          ; zero extend
3264                addd ,s++                      ; add to probabilitly base
3265                subd #$7f                      ; subtract 127 so that >= 0 is a hit, < 0 is a miss
3266                puls a,b,x,u,pc                ; restore registers and return
3267 ; This routine calculates the damage done by an attack. Enter with the attacker info at X and the defender
3268 ; info at U.
3269 damage          pshs a,b,x,y,u                ; save registers
3270                tfr x,y                        ; save attacker pointer
3271                ldx ,y                        ; get attacker power
3272                lda 2,y                        ; get magical offense power
3273                bsr applyscale                 ; scale it
3274                tfr d,x                        ; save result
3275                lda 3,u                        ; get defender magical defense
3276                bsr applyscale                 ; scale it
3277                addd 10,u                      ; add in defenders current damage
3278                std 10,u                      ; save new defender damage
3279                ldx ,y                        ; get attacker power
3280                lda 4,y                        ; get physical offense power
3281                bsr applyscale                 ; scale it
3282                tfr d,x                        ; save it
3283                lda 5,u                        ; get defender's physical defense power
3284                bsr applyscale                 ; scale it
3285                addd 10,u                      ; add to current defender damage level
3286                std 10,u                      ; save new damage level
3287                ldx ,u                        ; get defender's power
3288                cmpx 10,u                      ; compare with new damage level
3289                puls a,b,x,y,u,pc              ; restore registers and return
3290 ; Multiply X by the value in A, where the binary point in A is to the left of bit 6. Return only the
3291 ; integer result in D (rounded down).
3292 applyscale       pshs a,b,x                    ; save parameters and registers
3293                clr accum0                     ; blank out temp storage area
3294                ldb 3,s                        ; get LSB of X
3295                mul                            ; multiply LSB
3296                std accum0+1                   ; save in scratch variable
3297                lda ,s                        ; fetch multiplier
3298                ldb 2,s                        ; fetch MSB of X
3299                mul                            ; multiply it
3300                addd accum0                    ; add in partial product

```

```

3301          lsl accum0+2          ;* shift product left so binary point is to the right of
3302          rolb                  ;* of the upper 16 bits - leave interger result in D.
3303          rola                  ;*
3304          std ,s                ; save integer result for return
3305          puls a,b,x,pc         ; clean up parameters, fetch product, and return
3306 cmd_climb          ldd playerloc ; get player location
3307          jsr LCFE1             ; fetch hole information
3308          bmi LD46F             ; brif no holes
3309          sta accum0           ; save hole info
3310          ldx #kwlist_dir      ; point to direction list
3311          jsr LCBEC            ; go parse direction
3312          ble LD46F            ; brif no direction
3313          ldb accum0           ; get hole info
3314          cmpa #4              ; is it up?
3315          beq LD472            ; brif so
3316          cmpa #5              ; is it down?
3317          bne LD46F            ; brif not
3318          lda #1               ; level goes up one if we descend
3319          bitb #2              ; is there a hole down?
3320          bne LD478            ; brif so
3321 LD46F             jmp badcommand ; complain about bad direction or no hole
3322 LD472             lda #$ff      ; level goes down one if we ascend
3323             cmpb #1            ; do we have a ladder?
3324             bne LD46F          ; brif not
3325 LD478             showprepare   ; show the scary PREPARE! screen
3326             adda currentlevel  ; calculate the new level number
3327             createlevel        ; build the new level
3328             resetdisplay       ; reset everything and show the maze
3329             rts                ; return to caller
3330 cmd_examine       ldx #LD495    ; pointer to the inventory display routine
3331             stx displayptr      ; set up the display update routine
3332             updatedungeon      ; update the display
3333             rts                ; return to caller
3334 LD489             cleargfx2     ; clear graphics
3335             ldx ,u              ; get current text area start
3336             ldu #infoarea       ; point to info text area descriptor
3337             stx ,u              ; set text area start to the same place
3338             dec textother       ; set to nonstandard text rendering
3339             rts                ; return to caller
3340 ; This is the dungeon display routine that handles showing the inventory list.
3341 LD495             bsr LD489      ; clear the graphics area and set up for text rendering
3342             clr columnctr       ; flag column zero in object list
3343             ldd #10             ;* set up to centre "IN THIS ROOM"
3344             std 4,u             ;* column 10, row 0
3345             renderstrimpp       ; show the "IN THIS ROOM" heading
3346             fcb $62,$5c,$0a,$21 ; packed string "IN THIS ROOM"

```

```

3347      fcb $33,$04,$9e,$f6
3348      fcb $fc
3349      ldd playerloc          ; get player location
3350      jsr LCF82              ; get creature at player location
3351      beq LD4C0              ; brif no creature there
3352      ldx 4,u                ; get current text position
3353      leax 11,x              ; move 11 over
3354      stx 4,u                ; save new position
3355      renderstrimmp          ; show the "!CREATURE!" string if a creature is present
3356      fcb $56,$c7,$22,$86    ; packed string "!CREATURE!"
3357      fcb $95,$91,$77,$f0
3358 LD4C0      clr objiterstart ; reset object iterator
3359 LD4C2      ldd playerloc     ; get player location
3360      jsr LCF53              ; fetch next object
3361      beq LD4CD              ; brif no more objects
3362      bsr LD505              ; display object
3363      bra LD4C2              ; go handle another object
3364 LD4CD      tst columnctr    ; are we at the start of a line?
3365      beq LD4D3              ; brif so
3366      bsr LD4FE              ; do a newline
3367 LD4D3      ldd #$1b20        ; set up for displaying a row of !!!!
3368 LD4D6      renderchar        ; display a !
3369      decb                  ; done enough of them?
3370      bne LD4D6              ; brif not
3371      ldx 4,u                ; get current text location
3372      leax 12,x              ; adjust for centering
3373      stx 4,u                ; save new text location
3374      renderstrimmp          ; display "BACKPACK" heading
3375      fcb $40,$82,$35,$c0    ; packed string "BACKPACK"
3376      fcb $23,$5f,$c0
3377      ldx #backpack          ; point to backpack head pointer
3378 LD4ED      ldx ,x            ; get next item in backpack
3379      beq LD4FB              ; brif nothing else in backpack
3380      cmpx curtorch          ; is the object the currently burning torch?
3381      bne LD4F7              ; brif not
3382      com 6,u                ; invert video if it is
3383 LD4F7      bsr LD505          ; display object name
3384      bra LD4ED              ; go display another object
3385 LD4FB      clr textother     ; reset to standard text rendering
3386      rts                  ; return to caller
3387 LD4FE      lda #$1f          ; character code for newline
3388      renderchar            ; go move to next line
3389      clr columnctr          ; flag column 1
3390      rts                  ; return to caller
3391 LD505      pshs a,b,x        ; save registers
3392      jsr LC617              ; fetch object name string (decoded)

```

```

3393         renderstr                                ; display object name
3394         lda levbgmask                             ; get current level mask
3395         sta 6,u                                   ; restore proper background
3396         com columnctr                             ; are we on column 1 or 2?
3397         beq LD51E                                 ; brif back at column 1
3398         ldd 4,u                                   ; get cursor position
3399         addd #$10                                 ; move right 16 cells
3400         andb #$f0                                 ; round down to multiple of 16
3401         std 4,u                                   ; save new cursor position
3402         skip2                                     ; move on with routine
3403 LD51E      bsr LD4FE                             ; do a newline
3404         puls a,b,x,pc                             ; restore registers and return
3405 cmd_get    bsr LD576                             ; go parse hand and return pointer to it
3406         bne LD573                                 ; brif no direction
3407         jsr parseobj                             ; go parse an object
3408         clr objiterstart                         ; reset object iterator
3409 LD52B      ldd playerloc                         ; get current dungeon location
3410         jsr LCF53                                 ; fetch next object
3411         beq LD573                                 ; brif no more objects
3412         tst parsegenobj                         ; did we get a generic object type?
3413         bne LD53C                                 ; brif not
3414         lda 10,x                                 ; get object type we're looking at
3415         cmpa parseobjtypegen                    ; does it match?
3416         bra LD540                                 ; go finish up
3417 LD53C      lda 9,x                               ; get specific object type
3418         cmpa parseobjtype                       ; does it match?
3419 LD540      bne LD52B                             ; brif not - try another
3420         stx ,u                                   ; put object in selected hand
3421         inc 5,x                                 ; mark as not on floor
3422         ldb 10,x                                 ; get object general type
3423         ldx #LD9FA                             ; point to weight table
3424         ldb b,x                                 ; get object weight
3425         clra                                    ; zero extend
3426         bra LD56B                                 ; go adjust carried weight
3427 cmd_drop  bsr LD576                             ; parse a hand and get pointer
3428         beq LD573                                 ; brif no hand
3429         clra                                    ; NULL Pointer
3430         clrb
3431         std ,u                                   ; empty the hand out
3432         clr 5,x                                 ; mark object as on floor
3433         ldd playerloc                         ; get dungeon location
3434         std 2,x                                 ; set object location
3435         lda currentlevel                       ; get current level
3436         sta 4,x                                 ; set object level
3437         ldb 10,x                                 ; get object general type
3438         ldx #LD9FA                             ; point to weight table

```

```

3439          ldb b,x                      ; get weight of object
3440          negb                          ; negate it for subtraction
3441          sex                          ; sign extend
3442 LD56B      addd carryweight            ; add weight adjustment to carried weight
3443          std carryweight              ; save new carried weight
3444          checkdamage                  ; go update the damage situation
3445          bra LD5B7                    ; update display and return
3446 LD573      jmp badcommand             ; complain about bad command
3447 LD576      jmp LCC31                  ; go parse a hand and return pointer
3448 cmd_stow   bsr LD576                  ; get pointer to object in requested hand
3449          beq LD573                    ; brif no object in the hand
3450 LD57D      ldd backpack               ; get first item in backpack
3451          std ,x                      ; make it the next item in the list
3452          stx backpack                ; make this item the first item in the backpack
3453          clra                         ; NULL pointer
3454          clrb
3455          std ,u                      ; mark selected hand empty
3456          bra LD5B7                    ; update status line, etc.
3457 cmd_pull   bsr LD576                  ; fetch pointer to object in specified hand
3458          bne LD573                    ; brif there is something in that hand
3459          jsr parseobj                 ; parse object name
3460          ldx #backpack                ; point to backpack head pointer
3461 LD593      tfr x,y                    ; save previous pointer location
3462          ldx ,x                      ; fetch pointer to next item
3463          beq LD573                    ; brif end of list
3464          tst parsegenobj              ; is a specific object type requested?
3465          bne LD5A3                    ; brif so
3466          lda 10,x                     ; get object type (general) requested
3467          cmpa parseobjtypegen         ; does the object match?
3468          bra LD5A7                    ; finish up the loop
3469 LD5A3      lda 9,x                    ; get object type (specific) requested
3470          cmpa parseobjtype            ; does it match requested object type?
3471 LD5A7      bne LD593                  ; brif not matching object
3472          ldd ,x                      ; get next pointer
3473          std ,y                      ; put in previous next pointer (remove from backpack)
3474          stx ,u                      ; save object in the specified hand
3475 LD5AF      clra                       ; set up NULL pointer
3476          clrb
3477          cmpx curtorch                ; is this object the current torch?
3478          bne LD5B7                    ; brif not
3479          std curtorch                 ; turn off current torch
3480 LD5B7      updatestatus               ; update status line to reflect new hand contents
3481          updatedungeon               ; update the dungeon display
3482          rts                         ; return to caller
3483 cmd_incant  ldx #kwlist_adj           ; point to object types list
3484          jsr LCBEC                    ; look up object

```

```

3485      ble LD5EF                                ; brif not found in list or no type specified
3486      tst kwexact                              ; was it a complete match?
3487      beq LD5EF                                ; brif not
3488      std parseobjtype                          ; save object type
3489      ldu lefthand                             ; get left hand object
3490      bsr LD5D0                                ; check if matching object is there
3491      ldu righthand                            ; get right hand object and continue
3492 LD5D0   beq LD5EF                                ; brif no object carried
3493      lda 10,u                                ; get general type
3494      cmpa #1                                  ; is it a ring?
3495      bne LD5EF                                ; brif not
3496      lda 7,u                                  ; get incant to type
3497      beq LD5EF                                ; brif there isn't one
3498      cmpa parseobjtype                        ; does it match the one we incanted?
3499      bne LD5EF                                ; brif not
3500      sta 9,u                                  ; set new type to the incanted type
3501      setobjectspecc                           ; reset object specs
3502      playsoundimm $0D                         ; play the ring sound
3503      updatestatus                             ; update the status area
3504      clr 7,u                                  ; mark ring as incanted
3505      cmpa #$12                                ; is it the FINAL ring?
3506      beq LD5F0                                ; brif so
3507 LD5EF   rts                                    ; return to caller
3508 LD5F0   ldx #img_goodwiz                      ; point to good wizard image
3509      dec enablefadesound                     ; enable fade sound effect
3510      fadeinclrst                             ; fade in the wizard
3511      renderstrimmp                           ; display victory message line 1
3512      fcb $ff,$c4,$54,$3d                     ; packed string victory message line 1

3513      fcb $84,$d8,$08,$59
3514      fcb $D1,$2e,$c8,$03
3515      fcb $70,$a6,$93,$05
3516      fcb $10,$50,$20,$2e
3517      fcb $20
3518      renderstrimmp                           ; dispaly victory message line 2
3519      fcb $c8,$00,$00,$00                     ; packed string victory message line 2
3520      fcb $00,$03,$cc,$00
3521      fcb $81,$c5,$b8,$2e
3522      fcb $9d,$06,$44,$f7
3523      fcb $bc
3524 LD621   bra LD621                            ; Do nothing until IRQ decides something should happen
3525 cmd_reveal jsr LCC31                         ; parse a hand and get pointer to hand
3526      ldu ,u                                  ; is there an object there?
3527      beq LD63E                                ; brif not
3528      lda 11,u                                ; has object been revealed?
3529      beq LD63E                                ; brif so
3530      ldb #$19                                ; add multiplier to get needed power to reveal it

```

```

3531      mul                                ; multiply out
3532      cmpd powerlevel                    ; is player strong enough?
3533      bgt LD63E                          ; brif not
3534      lda 9,u                            ; fetch specific object type
3535 LD638  setobjectspecs                   ; update specs to revealed type
3536      clr 11,u                           ; mark object as revealed
3537      updatetestatus                     ; update the status area
3538 LD63E  rts                             ; return to caller
3539 cmd_turn  ldx #kwlist_dir               ; point to direction list
3540          jsr LCBEC                       ; look up word in list
3541          ble LD693                       ; brif no match or no word
3542          ldb facing                      ; get current direction
3543          cmpa #0                         ; TURN LEFT?
3544          bne LD654                       ; brif not
3545          decb                            ; rotate counter clockwise
3546          bsr LD66D                       ; normalize direction and update display
3547          bsr LD674                       ; sweep right
3548          bra LD669                       ; finish up
3549 LD654  cmpa #1                         ; TURN RIGHT?
3550          bne LD65D                       ; brif not
3551          incb                            ; rotate clockwise
3552          bsr LD66D                       ; normalize direction and update display
3553          bra LD667                       ; sweep left and finish up
3554 LD65D  cmpa #3                         ; TURN AROUND?
3555          bne LD693                       ; brif not
3556          addb #2                         ; turn 180
3557          bsr LD66D                       ; normalize direction and update display
3558          bsr LD684                       ; sweep left and fall through
3559 LD667  bsr LD684                       ; sweep left
3560 LD669  dec pageswap                     ; set graphic swap required
3561          sync                            ; wait for swap to happen
3562          rts                             ; return to caller
3563 LD66D  andb #3                         ; normalize direction to 0-3
3564          stb facing                      ; save new direction faced
3565          jmp LC660                       ; go update display and return
3566 LD674  bsr LD696                       ; draw outline and set up for a vertical line
3567          bne LD683                       ; brif not displaying anything
3568          ldd #8                         ; start at column 8
3569 LD67B  bsr LD6BA                       ; draw and erase vertical line
3570          addd #$20                      ; move right 32 pixels
3571          tsta                            ; did we wrap?
3572          beq LD67B                       ; brif not - keep going
3573 LD683  rts                             ; return to caller
3574 LD684  bsr LD696                       ; set up for drawing the sweep
3575          bne LD692                       ; brif we aren't drawing anything
3576          ldd #$f8                       ; start at X coord 248

```

```

3577 LD68B      bsr LD68A      ; draw and undraw the line
3578          subd #$20        ; move left 32 pixels
3579          bpl LD68B        ; brif we haven't wrapped yet - do another
3580 LD692      rts            ; return to caller
3581 LD693      jmp badcommand ; carp about a bad command
3582 LD696      ldu displayptr ; get display pointer
3583          cmpu #LCE66      ; is it the regular dungeon display
3584          bne LD6B9        ; brif not - don't show turning
3585          ldx #$8080        ; scale factors of 1.0
3586          stx horizscale   ; set horizontal and vertical scale factors to 1.0
3587          clr renderdist   ; set render distance to 0 (immediate)
3588          setlighting      ; set light level for rendering
3589          cleargfxl        ; clear screen
3590          ldx #LD6C6       ; point to outline graphic
3591          drawgraphic      ; draw it
3592          ldx #$11         ; * set start Y coord to 17
3593          stx ybeg         ; *
3594          ldx #$87         ; = set end Y coord to 135
3595          stx yend        ; =
3596          clra            ; clear Z
3597 LD6B9      rts            ; return to caller
3598 LD6BA      std xbeg       ; set start X coord
3599          std xend        ; set end X coord
3600          bsr LD6C0       ; draw the line and invert mask
3601 LD6C0      jsr drawline   ; draw the line again
3602          com levbgmask    ; invert mask
3603          rts            ; return to caller
3604 ; This is top and bottom lines during a turn sweep
3605 LD6C6      fcb 16,0
3606          fcb 16,255
3607          fcb $ff
3608          fcb 136,0
3609          fcb 136,255
3610          fcb $fe
3611 cmd_move   ldx #kwlist_dir ; point to direction list
3612          jsr LCBEC        ; look up direction
3613          blt LD693        ; brif bad direction
3614          bgt LD6E3        ; brif there is a direction
3615          dec movehalf     ; mark half step
3616          updatedungeon   ; update display
3617          clrb            ; set direction to forward
3618          clr movehalf     ; set to normal display
3619          bra LD6EF        ; go finish up
3620 LD6E3      cmpa #2        ; is it MOVE BACK?
3621          bne LD6F3        ; brif not
3622          dec movebackhalf ; set half step back

```



```

3623          updatedungeon          ; go update display
3624          ldb #2                  ; set direction to backward
3625          clr movebackhalf        ; set normal display
3626 LD6EF      bsr LD720             ; update position
3627          bra LD70E               ; go calculate movement cost, etc.
3628 LD6F3      cmpa #1              ; is it MOVE RIGHT?
3629          bne LD701              ; brif not
3630          ldb #1                  ; set direction to right
3631          bsr LD720             ; update position
3632          bne LD70E             ; brif movement failed
3633          bsr LD684             ; do a sweep left
3634          bra LD70E             ; calculate movement cost, etc.
3635 LD701      cmpa #0              ; is it LEFT?
3636          bne LD693             ; brif not
3637          ldb #3                  ; set direction to left
3638          bsr LD720             ; update position
3639          bne LD70E             ; brif movement failed
3640          jsr LD674             ; do a sweep right
3641 LD70E      ldd carryweight      ; get current carry weight
3642          jsr asrd3             ; divide by 8
3643          addd #3                ; add 3 for player weight
3644          addd damagelevel      ; add to damage level
3645          std damagelevel      ; save new damage level
3646          checkdamage          ; check for pasing out
3647          dec pageswap         ; set graphics swap required
3648          sync                 ; wait for swap to happen
3649          rts                  ; return to caller
3650 LD720      pshs a,b           ; save registers
3651          clr ,-s               ; make a temp
3652          addb facing           ; add direction to current facing direction
3653          andb #3              ; normalize to 0-3
3654          stb curdir           ; save move direction
3655          ldd playerloc        ; get current player location
3656          jsr LD136            ; calculate movement
3657          beq LD738            ; brif movement succeeds
3658          playsoundimm $14     ; play the "hit the wall" sound
3659          dec ,s              ; flag failed movement
3660          ldd playerloc        ; get current location as result
3661 LD738      std playerloc      ; save new location
3662          jsr LC660            ; go update the display
3663          tst ,s+              ; set flags for did movement succeed?
3664          puls a,b,pc          ; restore registers and return
3665 cmd_use    jsr LCC31          ; fetch pointer to object in specified hand
3666          beq LD767            ; brif nothing in the hand
3667          ldd 9,x              ; fetch object type and subtype
3668          cmpb #5              ; is it a torch?

```

```

3669                bne LD757                ; brif not
3670                stx curtorch              ; set object as currently mounted
3671                jsr LD57D                  ; go place the object in the backpack
3672                playsoundimm $11           ; play the torch sound
3673                updatedungeon              ; update dungeon with new lighting
3674                rts                        ; return to caller
3675 LD757            tfr x,u                  ; save object pointer
3676                ldx #LD76B                ; point to jump table
3677 LD75C            cmpa ,x                  ; does the sub type match?
3678                beq LD768                  ; brif so
3679                leax 3,x                   ; move to next entry
3680                cmpx #LD77A                ; end of table?
3681                blo LD75C                  ; brif not - try another
3682 LD767            rts                      ; no match - do nothing
3683 LD768            jmp [1,x]                ; transfer control to specified routine
3684 LD76B            fcb $05                  ; "THEWS" (thews flask)
3685                fdb LD77A
3686                fcb $09                    ; "HALE" (hale flask)
3687                fdb LD783
3688                fcb $08                    ; "ABYE" (abye flask)
3689                fdb LD787
3690                fcb $04                    ; "SEER" (seer scroll)
3691                fdb LD7A2
3692                fcb $07                    ; "VISION" (vision scroll)
3693                fdb LD7A0
3694 LD77A            ldd #1000                ; thews increases player power by 1000
3695                addd powerlevel            ; add to existing power value
3696                std powerlevel            ; save new power value
3697                bra LD792                  ; go empty the flask and update things
3698 LD783            clra                      ; new damage level = 0
3699                clrb
3700                bra LD790                  ; go set damage level and clean up flask
3701 LD787            ldx powerlevel            ; fetch player power level
3702                lda #$66                  ; roughly 0.8
3703                jsr applyscale            ; go calculate 80% of player power level
3704                addd damagelevel          ; add that to the current damage level
3705 LD790            std damagelevel          ; save new damage level
3706 LD792            ldb #$17                ; type for "EMPTY"
3707                stb 9,u                    ; change flask to EMPTY
3708                clr 11,u                  ; mark flask as revealed
3709                playsoundimm $0c          ; play the flask sound
3710                updatestatus              ; update status line to reflect changed flask state
3711                checkdamage              ; check the damage level and recovery interval
3712                rts                        ; return to caller
3713 LD7A0            clra                      ; flag for not showing creatures
3714                skip2                     ; skip over next instruction

```

```

3715 LD7A2      lda #$ff                ; flag for do show creatures
3716          sta showseer             ; set creature display flag
3717          tst ll,u                  ; is flask revealed?
3718          bne LD7B6                 ; brif not - do nothing
3719          playsoundimm $0e          ; play the scroll sound
3720          clr hidestatus            ; flag command processor to do a "restart"
3721          ldx #displayscroll        ; point to scroll display routine
3722          stx displayptr            ; set the display handler
3723          updatedungeon             ; update display with scroll
3724 LD7B6      rts                     ; return to caller
3725 cmd_zload   bsr LD7BC              ; parse the file name
3726          dec loadsaveflag          ; flag ZLOAD
3727          rts                       ; return to caller
3728 LD7BC      ldx #wordbuff           ; get start address to set to $ff
3729          leau $20,x                ; set $20 bytes
3730          setblock                  ; go clear block to $ff
3731          jmp LCB96                 ; go parse a word off command
3732 cmd_zsave   bsr LD7BC              ; parse the file name
3733          stx CBUFAD                ; point buffer to file name
3734          ldd #$0f                  ; * set block type to header, length to 15
3735          std BLKTYP                ; *
3736          inc loadsaveflag          ; flag ZSAVE
3737          rts                       ; return to caller
3738 ; Objects in backpack for demo game
3739 startobjdemo fcb 13                ; iron sword
3740          fcb 15                    ; pine torch
3741          fcb 16                    ; leather shield
3742          fcb $ff                   ; end of list
3743 ; Objects in backpack for normal game
3744 startobj     fcb 17                ; wooden sword
3745          fcb 15                    ; pine torch
3746          fcb $ff                   ; end of list
3747 ; This is the list of routines that get scheduling entries by default.
3748 LD7DC      fdb LD1EB               ; keyboard input processing
3749          fdb LD1C2                 ; dungeon display update
3750          fdb LD1D5                 ; damage healing tick
3751          fdb LD19B                 ; tick down torch life
3752          fdb LD027                 ; add the "revenge" monsters for the current level
3753          fdb 0                    ; end of routine list
3754 ; cold start variable initializers
3755 LD7E8      fcb 12
3756          fdb $103
3757          jmp swi2svc               ; SWI2 handler
3758          jmp swisvc               ; SWI handler
3759          jmp irqsvc               ; NMI handler (why??)
3760          jmp irqsvc               ; IRQ handler

```

```

3761      fcb $17
3762      fdb V202
3763      fcb $01                ; V202 - apparently unused
3764      fdb $ffff             ; allones - 16 bit all ones value, or -1
3765      fdb 128                ; horizcent
3766      fdb 76                 ; vertcent
3767      fdb LD870              ; screenvis - pointer to primary display screen info
3768      fdb LD876              ; screendraw - pointer to secondary display screen info
3769      fdb demogame           ; demoseqptr - pointer to demo game command sequence
3770      fdb objecttab          ; objectfree - next free object entry
3771      fdb linebuff           ; linebuffptr - the line input buffer pointer
3772      fcb 12,22              ; playerloc - starting coordinates in maze (y, x)
3773      fdb $23                ; carryweight - the weight of objects the player is carrying
3774      fdb $17a0              ; powerlevel - player power level
3775      fcb $54
3776      fdb infoarea
3777      fdb $1000              ; infoarea - text area starts at top of screen
3778      fdb $0260              ; infoarea+2 - text area ends after 19 lines
3779      fdb 0                  ; infoarea+4 - text cursor position at top of screen
3780      fcb 0                  ; infoarea+6 - black background
3781      fcb $ff                ; infoarea+7 - do not render on secondary screen
3782      fdb $2300              ; statusarea - text area starts at row 19 on screen
3783      fdb $40                ; statusarea+2 - text area goes for two lines
3784      fdb 0                  ; statusarea+4 - text cursor is at top of area
3785      fcb $ff                ; statusarea+6 - background is white
3786      fcb 0                  ; statusarea+7 - do render on secondary screen
3787      fdb $2400              ; commandarea - text area starts at row 20 on screen
3788      fdb $80                ; commandarea+2 - text area goes for four lines
3789      fdb 0                  ; commandarea+4 - text cursor is at top of area
3790      fcb 0                  ; commandarea+6 - background is black
3791      fcb 0                  ; commandarea+7 - do render on secondary screen
3792      fcb 9,9,4,2,0,0,0,0,0,0,0,0,0,0,0,0 ; initial creature counts for level 1
3793      fcb 2,4,0,6,6,6,0,0,0,0,0,0,0,0,0,0 ; initial creature counts for level 2
3794      fcb 0,0,0,4,0,6,8,4,0,0,1,0,0,0,0,0 ; initial creature counts for level 3
3795      fcb 0,0,0,0,0,0,8,6,6,4,0,0,0,0,0,0 ; initial creature counts for level 4
3796      fcb 2,2,2,2,2,2,2,4,4,8,0,1,0,0,0,0 ; initial creature counts for level 5
3797      fcb 4
3798      fdb emptyhand+10
3799      fcb $04,$00,$00,$05    ; empty hand attack data
3800      fcb 0
3801
3802      ; these tables are used for clearing and otherwise setting up the graphics screens
3803      LD870      fdb $1000      ; primary screen start address
3804      fdb $2300      ; primary screen gfx area end address
3805      fdb $2046      ; primary screen SAM register value
3806      LD876      fdb $2800      ; secondary screen start address

```

```

3807          fdb $3b00          ; secondary screen gfx area end address
3808          fdb $20a6          ; secondary screen SAM register value
3809 LD87C      fdb $2300          ; start address of status line on first screen
3810          fdb $2400          ; end address of status line on first screen
3811          fdb 0              ; dummy (SAM register setting)
3812          fdb $3b00          ; start address of status line on second screen
3813          fdb $3c00          ; end address of status line on second screen
3814          fdb 0              ; dummy (SAM register setting)
3815 LD888      fdb $2400          ; start address of command area on first screen
3816          fdb $2800          ; end address of command area on first screen
3817          fdb 0              ; dummy (SAM register setting)
3818          fdb $3c00          ; start address of command area on second screen
3819          fdb $4000          ; end address of command area on second screen
3820          fdb 0              ; dummy (SAM register setting)
3821
3822 ; This is the keyword table used for command parsing. Each keyword is stored in packed format.
3823 ; Each keyword is preceded by a value which indicates the object type. Where the object type is
3824 ; not relevant, that value will be zero. The value is shown in parentheses below.
3825 kwlist_cmd  fcb 15          ; 15 keywords in the command list
3826 kw_attack   fcb $30,$03,$4a,$04,$6b      ; "ATTACK" keyword
3827          fcb $28,$06,$c4,$b4,$40          ; "CLIMB" keyword
3828          fcb $20,$09,$27,$c0              ; "DROP" keyword
3829 kw_examine  fcb $38,$0b,$80,$b5,$2e,$28    ; "EXAMINE" keyword
3830          fcb $18,$0e,$5a,$00              ; "GET" keyword
3831          fcb $30,$12,$e1,$85,$d4          ; "INCANT" keyword
3832 kw_look     fcb $20,$18,$f7,$ac           ; "LOOK" keyword
3833 kw_move     fcb $20,$1a,$fb,$14           ; "MOVE" keyword
3834 kw_pull     fcb $20,$21,$56,$30           ; "PULL" keyword
3835          fcb $30,$24,$5b,$14,$2c          ; "REVEAL" keyword
3836          fcb $20,$27,$47,$dc             ; "STOW" keyword
3837 kw_turn     fcb $20,$29,$59,$38           ; "TURN" keyword
3838 kw_use      fcb $18,$2b,$32,$80           ; "USE" keyword
3839          fcb $28,$34,$c7,$84,$80          ; "ZLOAD" keyword
3840          fcb $28,$35,$30,$d8,$a0          ; "ZSAVE" keyword
3841 kwlist_dir  fcb 6              ; 6 keywords in direction list
3842 kw_left     fcb $20,$18,$53,$50           ; "LEFT" keyword
3843 kw_right    fcb $28,$24,$93,$a2,$80       ; "RIGHT" keyword
3844          fcb $20,$04,$11,$ac             ; "BACK" keyword
3845          fcb $30,$03,$27,$d5,$c4          ; "AROUND" keyword
3846          fcb $10,$2b,$00                 ; "UP" keyword
3847          fcb $20,$08,$fb,$b8             ; "DOWN" keyword
3848 kwlist_adj  fcb 25             ; 25 keywords in the misc keywords list
3849 kw_supreme  fcb $38,$67,$58,$48,$ad,$28   ; "SUPREME" keyword (1)
3850          fcb $28,$54,$fa,$b0,$a0         ; "JOULE" keyword (1)
3851          fcb $31,$0a,$cb,$26,$68         ; "ELVISH" keyword (4)
3852          fcb $38,$da,$9a,$22,$49,$60     ; "MITHRIL" keyword (3)

```

```

3853      fcb $20,$a6,$52,$c8      ; "SEER" keyword (2)
3854      fcb $28,$28,$82,$de,$60  ; "THEWS" keyword (0)
3855      fcb $20,$64,$96,$94      ; "RIME" keyword (1)
3856      fcb $30,$ac,$99,$a5,$ee  ; "VISION" keyword (2)
3857      fcb $20,$02,$2c,$94      ; "ABYE" keyword (0)
3858      fcb $20,$10,$16,$14      ; "HALE" keyword (0)
3859      fcb $29,$66,$f6,$06,$40  ; "SOLAR" keyword (5)
3860      fcb $30,$c5,$27,$bb,$45  ; "BRONZE" keyword (3)
3861      fcb $30,$6d,$56,$0c,$2e  ; "VULCAN" keyword (1)
3862      fcb $21,$13,$27,$b8      ; "IRON" keyword (4)
3863      fcb $29,$59,$57,$06,$40  ; "LUNAR" keyword (5)
3864      fcb $21,$60,$97,$14      ; "PINE" keyword (5)
3865      fcb $38,$d8,$50,$d1,$05,$90 ; "LEATHER" keyword (3)
3866      fcb $31,$2e,$f7,$90,$ae  ; "WOODEN" keyword (4)
3867      fcb $28,$4c,$97,$05,$80  ; "FINAL" keyword (1)
3868      fcb $30,$4a,$e2,$c8,$f9  ; "ENERGY" keyword (1)
3869      fcb $18,$52,$32,$80      ; "ICE" keyword (1)
3870      fcb $20,$4c,$99,$14      ; "FIRE" keyword (1)
3871      fcb $20,$4e,$f6,$10      ; "GOLD" keyword (1)
3872      fcb $28,$0a,$d8,$53,$20  ; "EMPTY" keyword (0)
3873      fcb $21,$48,$50,$90      ; "DEAD" keyword (5)
3874  kwlist_obj      fcb 6      ; 6 object types in the following list
3875  kw_flask         fcb $28,$0c,$c0,$cd,$60  ; "FLASK" keyword (0)
3876                  fcb $20,$64,$97,$1c      ; "RING" keyword (1)
3877                  fcb $30,$a6,$39,$3d,$8c    ; "SCROLL" keyword (2)
3878  kw_shield        fcb $30,$e6,$84,$95,$84  ; "SHIELD" keyword (3)
3879  kw_sword         fcb $29,$27,$77,$c8,$80  ; "SWORD" keyword (4)
3880  kw_torch         fcb $29,$68,$f9,$0d,$00  ; "TORCH" keyword (5)

3881 ; The following is the sequence of commands used in the demo game
3882 demogame          fcb 1      ; EXAMINE
3883                  fdb kw_examine
3884                  fcb 3      ; PULL RIGHT TORCH
3885                  fdb kw_pull
3886                  fdb kw_right
3887                  fdb kw_torch
3888                  fcb 2      ; USE RIGHT
3889                  fdb kw_use
3890                  fdb kw_right
3891                  fcb 1      ; LOOK
3892                  fdb kw_look
3893                  fcb 1      ; MOVE
3894                  fdb kw_move
3895                  fcb 3      ; PULL LEFT SHIELD
3896                  fdb kw_pull
3897                  fdb kw_left
3898                  fdb kw_shield

```

```

3899          fcb 3                                ; PULL RIGHT SWORD
3900          fdb kw_pull
3901          fdb kw_right
3902          fdb kw_sword
3903          fcb 1                                ; MOVE
3904          fdb kw_move
3905          fcb 1                                ; MOVE
3906          fdb kw_move
3907          fcb 2                                ; ATTACK RIGHT
3908          fdb kw_attack
3909          fdb kw_right
3910          fcb 2                                ; TURN RIGHT
3911          fdb kw_turn
3912          fdb kw_right
3913          fcb 1                                ; MOVE
3914          fdb kw_move
3915          fcb 1                                ; MOVE
3916          fdb kw_move
3917          fcb 1                                ; MOVE
3918          fdb kw_move
3919          fcb 2                                ; TURN RIGHT
3920          fdb kw_turn
3921          fdb kw_right
3922          fcb 1                                ; MOVE
3923          fdb kw_move
3924          fcb 1                                ; MOVE
3925          fdb kw_move
3926          fcb $ff
3927          ; jump table for commands
3928 LD9D0      fdb cmd_attack                      ; ATTACK
3929          fdb cmd_climb                        ; CLIMB
3930          fdb cmd_drop                         ; DROP
3931          fdb cmd_examine                     ; EXAMINE
3932          fdb cmd_get                         ; GET
3933          fdb cmd_incant                      ; INCANT
3934          fdb cmd_look                        ; LOOK
3935          fdb cmd_move                        ; MOVE
3936          fdb cmd_pull                        ; PULL
3937          fdb cmd_reveal                      ; REVEAL
3938          fdb cmd_stow                        ; STOW
3939          fdb cmd_turn                        ; TURN
3940          fdb cmd_use                         ; USE
3941          fdb cmd_zload                       ; ZLOAD
3942          fdb cmd_zsave                       ; ZSAVE
3943          ; pointers to the image data for object types
3944 LD9EE      fdb img_flask                      ; flask

```

```

3945          fdb img_ring          ; ring
3946          fdb img_scroll        ; scroll
3947          fdb img_shield        ; shield
3948          fdb img_sword         ; sword
3949          fdb img_torch         ; torch
3950
3951 LD9FA          fcb $05,$01
3952
3953 LD9FC          fcb $0A,$19,$19,$0A
3954 ; This is the object data table. Each entry is four bytes as follows:
3955 ; 0      object type
3956 ; 1      reveal strength required
3957 ; 2      magical offense multiplier
3958 ; 3      physical offense multiplier
3959 objspecs      fcb $01,$FF,$00,$05          ; supreme ring
3960          fcb $01,$AA,$00,$05          ; joule ring
3961          fcb $04,$96,$40,$40          ; elvish sword
3962          fcb $03,$8C,$0D,$1A          ; mithril shield
3963          fcb $02,$82,$00,$05          ; seer scroll
3964          fcb $00,$46,$00,$05          ; thews flask
3965          fcb $01,$34,$00,$05          ; rime ring
3966          fcb $02,$32,$00,$05          ; vision scroll
3967          fcb $00,$30,$00,$05          ; abye flask
3968          fcb $00,$28,$00,$05          ; hale flask
3969          fcb $05,$46,$00,$05          ; solar torch
3970          fcb $03,$19,$00,$1A          ; bronze shield
3971          fcb $01,$0D,$00,$05          ; vulcan ring
3972          fcb $04,$0D,$00,$28          ; iron sword
3973
3974          fcb $05,$19,$00,$05          ; lunar torch
3975          fcb $05,$05,$00,$05          ; pine torch
3976          fcb $03,$05,$00,$0A          ; leather shield
3977          fcb $04,$05,$00,$10          ; wooden sword
3978          fcb $01,$00,$00,$00          ; final ring
3979          fcb $01,$00,$FF,$FF          ; energy ring
3980          fcb $01,$00,$FF,$FF          ; ice ring
3981          fcb $01,$00,$FF,$FF          ; fire ring
3982          fcb $01,$00,$00,$05          ; gold ring
3983          fcb $00,$00,$00,$05          ; empty flask
3984          fcb $05,$05,$00,$05          ; dead torch
3985 ; This table has additional object data including ring charges, etc, organized as follows:
3986 ; 0      object number
3987 ; 1      burn time (torch), charges (ring), magical defense (shield)
3988 ; 2      physical light (torch), physical defense (shield)
3989 ; 3      magical ight (torch)
3989 objextraspecs fcb $00,$03,$12,$00          ; supreme ring
3990          fcb $01,$03,$13,$00          ; joule ring

```



```

3991          fcb $03,$40,$40,$00          ; mithril shield
3992          fcb $06,$03,$14,$00          ; rime ring
3993          fcb $0A,$3C,$0D,$0B          ; solar torch
3994          fcb $0B,$60,$80,$00          ; bronze shield
3995          fcb $0C,$03,$15,$00          ; vulcan ring
3996          fcb $0E,$1E,$0A,$04          ; lunar torch
3997          fcb $0F,$0F,$07,$00          ; pine torch
3998          fcb $10,$6C,$80,$00          ; leather shield
3999          fcb $18,$00,$00,$00          ; dead torch
4000          fcb $FF                      ; end of table
4001 ; This is the table of objects to create for a game. Each entry corresponds to
4002 ; a single object type. The first nibble is the minimum level number on which it
4003 ; appears. The second nibble is the number of objects of that type to generate.
4004 ; Generation starts at the specified level and creates one object assigned to
4005 ; that level. Then it creates another assigned to the next level, and so on.
4006 ; If it gets to level 5, it will reset to the minimum level. It cycles like this
4007 ; until there are the specified number of objects in the entire game.
4008 LDA91          fcb $41                  ; 1 supreme ring, level 5
4009          fcb $31                  ; 1 joule ring, level 4
4010          fcb $31                  ; 1 elvish sword, level 4
4011          fcb $32                  ; 1 mithril shield each, level 4 and 5
4012          fcb $23                  ; 1 seer scroll each, level 3-5
4013          fcb $23                  ; 1 thews flask each, level 3-5
4014          fcb $11                  ; 1 rime ring, level 2
4015          fcb $13                  ; 1 vision scrool each, level 2-4
4016          fcb $16                  ; 2 aby flask each, level 2-3; 1 aby flask each level 4-5
4017          fcb $14                  ; 1 hale flask each, level 2-5
4018          fcb $14                  ; 1 solar torch each, level 2-5
4019          fcb $16                  ; 2 bronze shield each, level 2-3; 1 bronze shield each, level
4-5
4020          fcb $01                  ; 1 vulcan ring, level 1
4021          fcb $04                  ; 1 iron sword each, level 1-4
4022          fcb $08                  ; 2 lunar torch each, level 1-3; 1 lunar torch each, level 4-5
4023          fcb $08                  ; 2 pine torch each, level 1-3; 1 pine torch each, level 4-5
4024          fcb $03                  ; 1 leather shield each, level 1-3
4025          fcb $04                  ; 1 wooden sword each, level 1-4
4026 ; pointers to creature images
4027 LDAA3          fdb LDE26              ; spider
4028          fdb LDFCA                ; viper
4029          fdb LDD41                ; club giant
4030          fdb LDE59                ; blob
4031          fdb LDE82                ; knight
4032          fdb LDD51                ; axe giant
4033          fdb LDE3F                ; scorpion
4034          fdb LDE9D                ; shield knight
4035          fdb LDE07                ; wraith

```

```

4036          fdb LDDA3                      ; galdrog
4037          fdb img_wizardgen              ; wizard's image
4038          fdb img_wizard                 ; wizard
4039 ; This is the creature data table. Each entry is 8 bytes organized as follows:
4040 ; 0,1   creature power level
4041 ; 2     creature magical attack strength
4042 ; 3     creature magical defense strength
4043 ; 4     creature physical attack strength
4044 ; 5     creature physical defense strength
4045 ; 6     creature scheduling speed (movement) (in tenths of a second)
4046 ; 7     creature scheduling speed (attack) (in tenths of a second)
4047 LDABB          fcb $00,$20,$00,$FF,$80,$FF,$17,$0B ; spider
4048          fcb $00,$38,$00,$FF,$50,$80,$0F,$07 ; viper
4049          fcb $00,$C8,$00,$FF,$34,$C0,$1D,$17 ; club giant
4050          fcb $01,$30,$00,$FF,$60,$A7,$1F,$1F ; blob
4051          fcb $01,$F8,$00,$80,$60,$3C,$0D,$07 ; knight
4052          fcb $02,$C0,$00,$80,$80,$30,$11,$0D ; axe giant
4053          fcb $01,$90,$FF,$80,$FF,$80,$05,$04 ; scorpion
4054          fcb $03,$20,$00,$40,$FF,$08,$0D,$07 ; shield knight
4055          fcb $03,$20,$C0,$10,$C0,$08,$03,$03 ; wraith
4056          fcb $03,$E8,$FF,$05,$FF,$03,$04,$03 ; galdrog
4057          fcb $03,$E8,$FF,$06,$FF,$00,$0D,$07 ; wizard's image
4058          fcb $1F,$40,$FF,$06,$FF,$00,$0D,$07 ; wizard
4059 ; This is the text font - these values are in packed format
4060 LDB1B          fcb $30,$00,$00,$00,$00          ; char code 0 - space
4061          fcb $31,$15,$18,$fe,$31                ; char code 1 - A
4062          fcb $37,$a3,$1f,$46,$3e                ; char code 2 - B
4063          fcb $33,$a3,$08,$42,$2e                ; char code 3 - C
4064          fcb $37,$a3,$18,$c6,$3e                ; char code 4 - D
4065          fcb $37,$e1,$0f,$42,$1f                ; char code 5 - E
4066          fcb $37,$e1,$0f,$42,$10                ; char code 6 - F
4067          fcb $33,$e3,$08,$4e,$2f                ; char code 7 - G
4068          fcb $34,$63,$1f,$c6,$31                ; char code 8 - H
4069          fcb $33,$88,$42,$10,$8e                ; char code 9 - I
4070          fcb $30,$42,$10,$86,$2e                ; char code 10 - J
4071          fcb $34,$65,$4c,$52,$51                ; char code 11 - K
4072          fcb $34,$21,$08,$42,$1f                ; char code 12 - L
4073          fcb $34,$77,$5a,$d6,$31                ; char code 13 - M
4074          fcb $34,$63,$9a,$ce,$31                ; char code 14 - N
4075          fcb $33,$a3,$18,$c6,$2e                ; char code 15 - O
4076          fcb $37,$a3,$1f,$42,$10                ; char code 16 - P
4077          fcb $33,$a3,$18,$d6,$4d                ; char code 17 - Q
4078          fcb $37,$a3,$1f,$52,$51                ; char code 18 - R
4079          fcb $33,$a3,$07,$06,$2e                ; char code 19 - S
4080          fcb $37,$ea,$42,$10,$84                ; char code 20 - T
4081          fcb $34,$63,$18,$c6,$2e                ; char code 21 - U

```

```

4082          fcb $34,$63,$15,$28,$84          ; char code 22 - V
4083          fcb $34,$63,$1a,$d7,$71          ; char code 23 - W
4084          fcb $34,$62,$a2,$2a,$31          ; char code 24 - X
4085          fcb $34,$62,$a2,$10,$84          ; char code 25 - Y
4086          fcb $37,$c2,$22,$22,$1f          ; char code 26 - Z
4087          fcb $31,$08,$42,$10,$04          ; char code 27 - !
4088          fcb $30,$00,$00,$00,$1f          ; char code 28 - underscore
4089          fcb $33,$a2,$13,$10,$04          ; char code 29 - ?
4090          fcb $30,$00,$00,$00,$04          ; char code 30 - .
4091 ; some special glyphs
4092 LDBB6          fcb $00,$00,$01,$01,$00,$00,$00 ; char code 32 - left part of contracted heart
4093          fcb $00,$a0,$f0,$f0,$e0,$40,$00 ; char code 33 - right part of contracted heart
4094          fcb $00,$01,$03,$03,$01,$00,$00 ; char code 34 - left half of expanded heart
4095          fcb $00,$b0,$f8,$f8,$f0,$e0,$40 ; char code 35 - right part of expanded heart
4096
4097 ; These two entries are related to sound generation.
4098 LDBD2          fcb $00,$80,$00,$01,$00,$50,$00,$04 ; for the "wizard fade out" sound and the walk into
wall sound
4099 LDBDA          fcb $00,$50,$00,$05          ; for the create death sound
4100
4101 ; This table is for rendering walls in specific directions. There is one entry each
4102 ; for left, right, and forward. Each entry has four pointers to graphics, for no door,
4103 ; physical door, magical door, and solid wall.
4104 LDBDE          fcb 3
4105          fdb LDC4F
4106          fdb LDC6B
4107          fdb LDC9B
4108          fdb LDC33
4109
4110          fcb 0
4111          fdb LDC6A
4112          fdb LDC8B
4113          fdb LDCA9
4114          fdb LDC45
4115          fcb 1
4116          fdb LDC5D
4117          fdb LDC7B
4118          fdb LDCA2
4119          fdb LDC3C
4120          fcb $ff
4121
4121 ; image data for a shield
4122 img_shield      fcb 134,172
4123          fcb 128,192
4124          fcb 122,186
4125          fcb 128,168
4126          fcb $fc

```

```
4127             fcb $3e,$04,$00
4128             fcb $fe
4129 ; image data for a torch
4130 img_torch      fcb 118,60
4131             fcb $fc
4132             fcb $f7,$ff,$2a,$00
4133             fcb $fe
4134 ; image data for a sword
4135 img_sword      fcb 114,80
4136             fcb 124,100
4137             fcb $ff
4138             fcb 118,82
4139             fcb 114,86
4140             fcb $fe
4141
4142 ; image data for a flask
4143 img_flask      fcb 110,162
4144             fcb $fc
4145             fcb $51,$0e,$b1,$00
4146             fcb $fe
4147 ; image data for a ring
4148 img_ring       fcb 122,60
4149             fcb $fc
4150             fcb $11,$1f,$ff,$f1,$00
4151             fcb $fe
4152 ; image data for a scroll
4153 img_scroll     fcb 118,194
4154             fcb $fc
4155             fcb $1f,$34,$f1,$dc,$00
4156             fcb $fe
4157
4158 ; Creature around corner to the left indicator graphic
4159 LDC33          fcb 16,27
4160             fcb 38,64
4161             fcb 114,64
4162             fcb 136,27
4163             fcb $fe
4164 ; Creature around corner to the right indicator graphic
4165 LDC3C          fcb 16,229
4166             fcb 38,192
4167             fcb 114,192
4168             fcb 136,229
4169             fcb $fe
4170 LDC45          fcb 38,64
4171             fcb 38,192
4172             fcb $ff
```

4173		fcf 114,64
4174		fcf 114,192
4175		fcf \$fe
4176	LDC4F	fcf 38,29
4177		fcf 38,64
4178		fcf 114,64
4179		fcf 114,27
4180		fcf \$ff
4181		fcf 16,27
4182		fcf 38,64
4183		fcf \$fe
4184	LDC5D	fcf 38,229
4185		fcf 38,192
4186		fcf 114,192
4187		fcf 114,229
4188		fcf \$ff
4189		fcf 16,229
4190		fcf 38,192
4191	LDC6A	fcf \$fe
4192	LDC6B	fcf 128,40
4193		fcf 65,40
4194		fcf 68,56
4195		fcf 119,56
4196		fcf \$ff
4197		fcf 92,48
4198		fcf 93,52
4199		fcf \$fd
4200		fdb LDC33
4201	LDC7B	fcf 128,216
4202		fcf 65,216
4203		fcf 68,200
4204		fcf 119,200
4205		fcf \$ff
4206		fcf 92,208
4207		fcf 93,204
4208		fcf \$fd
4209		fdb LDC3C
4210	LDC8B	fcf 114,108
4211		fcf 67,108
4212		fcf 67,148
4213		fcf 114,148
4214		fcf \$ff
4215		fcf 94,126
4216		fcf 94,130
4217		fcf \$fd
4218		fdb LDC45

```

4219 LDC9B      fcb 128,40
4220          fcb 66,50
4221          fcb 117,58
4222          fcb $fe
4223 LDCA2      fcb 128,216
4224          fcb 66,206
4225          fcb 117,198
4226          fcb $fe
4227 LDCA9      fcb 113,108
4228          fcb 67,128
4229          fcb 114,148
4230          fcb $fe
4231 LDCB0      fcb 100,28
4232          fcb $fc
4233          fcb $44,$2e,$42,$4c,$00
4234          fcb $fe
4235 LDCB9      fcb 100,228
4236          fcb $fc
4237          fcb $4c,$22,$4e,$44,$00
4238          fcb $fe
4239 ; Table of pointers to hole/ladder graphics
4240 LDCC2      fdb LDD0E
4241          fdb LDCCA
4242          fdb LDD2A
4243          fdb LDCD0
4244 LDCCA      fcb $fb
4245          fdb LDCD6
4246          fcb $fd
4247          fdb LDD0E
4248 LDCD0      fcb $fb
4249          fdb LDCD6
4250          fcb $fd
4251          fdb LDD2A
4252 LDCD6      fcb 24,116
4253          fcb 128,116
4254          fcb $ff
4255          fcb 24,140
4256          fcb 128,140
4257          fcb $ff
4258          fcb 28,116
4259          fcb 28,140
4260          fcb $ff
4261          fcb 40,116
4262          fcb 40,140
4263          fcb $ff
4264          fcb 52,116

```

4265		fc b 52,140
4266		fc b \$ff
4267		fc b 64,116
4268		fc b 64,140
4269		fc b \$ff
4270		fc b 76,116
4271		fc b 76,140
4272		fc b \$ff
4273		fc b 88,116
4274		fc b 88,140
4275		fc b \$ff
4276		fc b 100,116
4277		fc b 100,140
4278		fc b \$ff
4279		fc b 112,116
4280		fc b 112,140
4281		fc b \$ff
4282		fc b 123,116
4283		fc b 123,140
4284		fc b \$ff
4285		fc b \$fa
4286	LDD0E	fc b 34,100
4287		fc b 24,92
4288		fc b 24,164
4289		fc b 34,156
4290		fc b 34,100
4291		fc b 24,100
4292		fc b \$ff
4293		fc b 34,156
4294		fc b 24,156
4295		fc b \$ff
4296		fc b 28,47
4297		fc b 28,96
4298		fc b \$ff
4299		fc b 28,161
4300		fc b 28,210
4301		fc b \$fe
4302	LDD2A	fc b 118,100
4303		fc b 128,92
4304		fc b 128,164
4305		fc b 118,156
4306		fc b 118,100
4307		fc b 128,100
4308		fc b \$ff
4309		fc b 118,156
4310		fc b 128,156

4311		fcf \$ff
4312	LDD3C	fcf 28,47
4313		fcf 28,210
4314		fcf \$fe
4315	LDD41	fcf 104,98
4316		fcf \$fc
4317		fcf \$d7,\$d4,\$14,\$12,\$30,\$1d,\$0d,\$fd
4318		fcf \$29,\$00
4319		fcf \$fd
4320		fdb LDD62
4321	LDD51	fcf 104,98
4322		fcf 94,124
4323		fcf 96,126
4324		fcf 106,100
4325		fcf \$ff
4326		fcf 102,132
4327		fcf 92,114
4328		fcf 102,118
4329		fcf 110,114
4330	LDD62	fcf 102,132
4331		fcf \$fc
4332		fcf \$02,\$56,\$56,\$17,\$ee,\$02,\$ea,\$bb
4333		fcf \$bb,\$ea,\$ea,\$00
4334		fcf 78,92
4335		fcf \$fc
4336		fcf \$c2,\$51,\$3e,\$cf,\$fc,\$42,\$13,\$00
4337		fcf 106,90
4338		fcf \$fc
4339		fcf \$1e,\$11,\$f3,\$62,\$39,\$e2,\$0c,\$e4
4340		fcf \$8a,\$e2,\$00
4341		fcf 86,84
4342		fcf \$fc
4343		fcf \$54,\$65,\$2e,\$ca,\$ba,\$a1,\$d4,\$ee
4344		fcf \$12,\$d2,\$13,\$e1,\$20,\$f6,\$24,\$72
4345		fcf \$58,\$ee,\$c5,\$be,\$00
4346		fcf \$fe
4347	LDDA3	fcf 80,124
4348		fcf 94,114
4349		fcf 110,120
4350		fcf 132,112
4351		fcf 104,78
4352		fcf 132,48
4353		fcf 68,72
4354		fcf 84,32
4355		fcf 22,88
4356		fcf 52,114

4357	fc	b	92,128	
4358	fc	b	52,142	
4359	fc	b	22,168	
4360	fc	b	88,224	
4361	fc	b	68,184	
4362	fc	b	132,208	
4363	fc	b	112,178	
4364	fc	b	132,144	
4365	fc	b	110,136	
4366	fc	b	94,142	
4367	fc	b	80,132	
4368	fc	b	\$ff	
4369	fc	b	132,112	
4370	fc	b	\$fc	
4371	fc	b	\$c5,\$92,\$be,\$c3,\$43,\$5e,\$72,\$45	
4372	fc	b	\$00	
4373	fc	b	82,122	
4374	fc	b	\$fc	
4375	fc	b	\$78,\$e9,\$8d,\$ec,\$33,\$0c,\$24,\$72	
4376	fc	b	\$47,\$e7,\$00	
4377	fc	b	22,168	
4378	fc	b	\$fc	
4379	fc	b	\$2d,\$c2,\$3d,\$30,\$4b,\$4b,\$ed,\$b2	
4380	fc	b	\$9d,\$71,\$3d,\$dd,\$91,\$7d,\$52,\$63	
4381	fc	b	\$a3,\$2d,\$ed,\$2d,\$cb,\$cb,\$d0,\$dd	
4382	fc	b	\$42,\$ed,\$00	
4383	fc	b	\$fe	
4384	LDE07	fc	b	62,68
4385		fc	b	68,88
4386		fc	b	56,100
4387		fc	b	\$ff
4388		fc	b	74,90
4389		fc	b	70,74
4390		fc	b	\$fc
4391		fc	b	\$33,\$f5,\$f5,\$c1,\$5a,\$62,\$0e,\$00
4392		fc	b	100,80
4393		fc	b	\$fc
4394		fc	b	\$b3,\$17,\$34,\$eb,\$0a,\$3d,\$00
4395		fc	b	\$fe
4396	LDE26	fc	b	124,160
4397		fc	b	\$fc
4398		fc	b	\$c2,\$22,\$e4,\$24,\$2c,\$ec,\$04,\$04
4399		fc	b	\$e2,\$42,\$00
4400		fc	b	124,168
4401		fc	b	\$fc
4402		fc	b	\$c1,\$21,\$12,\$f2,\$e1,\$41,\$00

4403		fc	b	\$	f	e
4404	LDE3F	fc	b	112,74		
4405		fc	b	\$	f	c
4406		fc	b	\$e0,\$ee,\$2c,\$42,\$14,\$14,\$20,\$0c		
4407		fc	b	\$cc,\$22,\$0c,\$22,\$00		
4408		fc	b	124,90		
4409		fc	b	\$	f	c
4410		fc	b	\$e0,\$0c,\$2c,\$20,\$04,\$00		
4411		fc	b	\$	f	e
4412	LDE59	fc	b	82,130		
4413		fc	b	\$	f	c
4414		fc	b	\$28,\$7d,\$5f,\$50,\$5b,\$f5,\$2f,\$d5		
4415		fc	b	\$17,\$17,\$f3,\$22,\$e1,\$14,\$dd,\$8f		
4416		fc	b	\$8d,\$db,\$ec,\$00		
4417		fc	b	86,130		
4418		fc	b	\$	f	c
4419		fc	b	\$33,\$31,\$1b,\$91,\$3b,\$5f,\$f5,\$00		
4420		fc	b	108,116		
4421		fc	b	114,118		
4422		fc	b	120,144		
4423		fc	b	\$	f	e
4424	LDE82	fc	b	34,124		
4425		fc	b	\$	f	c
4426		fc	b	\$04,\$1f,\$0e,\$ff,\$00		
4427		fc	b	80,142		
4428		fc	b	64,136		
4429		fc	b	46,146		
4430		fc	b	64,156		
4431		fc	b	82,140		
4432		fc	b	76,136		
4433		fc	b	64,146		
4434		fc	b	58,140		
4435		fc	b	\$	f	d
4436		f	d	b	LDEB3	
4437	LDE9D	fc	b	30,126		
4438		fc	b	\$	f	c
4439		fc	b	\$50,\$0f,\$e0,\$00		
4440		fc	b	44,150		
4441		fc	b	52,166		
4442		fc	b	76,164		
4443		fc	b	92,150		
4444		fc	b	76,136		
4445		fc	b	52,134		
4446		fc	b	44,150		
4447		fc	b	\$	f	f
4448	LDEB3	fc	b	80,140		

```
4449      fcb 128,152
4450      fcb 132,160
4451      fcb 132,144
4452      fcb 126,144
4453      fcb 84,130
4454      fcb $ff
4455      fcb 84,126
4456      fcb 126,110
4457      fcb 132,110
4458      fcb 132,92
4459      fcb 128,102
4460      fcb 80,116
4461      fcb $ff
4462      fcb 80,140
4463      fcb $fc
4464      fcb $3a,$d9,$83,$de,$ad,$e6,$a1,$e2
4465      fcb $22,$61,$26,$ea,$20,$3d,$dd,$e0
4466      fcb $00
4467      fcb 52,128
4468      fcb 20,128
4469      fcb $fc
4470      fcb $0e,$21,$02,$e1,$0e,$00
4471      fcb 74,102
4472      fcb $fc
4473      fcb $e0,$02,$d0,$08,$30,$02,$20,$01
4474      fcb $30,$02,$d0,$01,$87,$00
4475      fcb 46,110
4476      fcb 64,102
4477      fcb 64,100
4478      fcb 30,102
4479      fcb 20,98
4480      fcb 30,94
4481      fcb 64,96
4482      fcb 64,98
4483      fcb 20,98
4484      fcb $FE
4485      ; Image for the Wizard
4486      img_wizard fcb 46,98
4487      fcb $fc
4488      fcb $21,$2f,$2d,$fd,$ce,$c2,$f2,$12
4489      fcb $0f,$1e,$3f,$21,$12,$e3,$e0,$00
4490      fcb 104,154
4491      fcb $fc
4492      fcb $21,$2f,$2d,$fd,$ce,$c2,$f2,$12
4493      fcb $0f,$1e,$3f,$22,$12,$e2,$e0,$00
4494      fcb $fd
```

```
4495          fdb img_wizardgen
4496 ; Image for the "good" wizard
4497 img_goodwiz fcb 40,86
4498          fcb 64,92
4499          fcb 42,100
4500          fcb 54,82
4501          fcb 56,104
4502          fcb 40,86
4503          fcb $ff
4504          fcb 66,140
4505          fcb $fc
4506          fcb $70,$ad,$35,$1b,$b3,$00
4507          fcb 96,146
4508          fcb 120,148
4509          fcb 100,136
4510          fcb 106,154
4511          fcb 116,138
4512          fcb 96,146
4513          fcb $ff
4514          fcb 80,116
4515          fcb $fc
4516          fcb $53,$ec,$e4,$4d,$b0,$00
4517 img_wizardgen fcb 64,124
4518          fcb $fc
4519          fcb $4e,$c0,$7b,$9c,$d4,$e4,$e1,$e1
4520          fcb $dd,$1c,$96,$03,$00
4521          fcb 28,130
4522          fcb $fc
4523          fcb $03,$45,$71,$da,$1e,$11,$e1,$00
4524          fcb 48,134
4525          fcb 54,142
4526          fcb 116,164
4527          fcb 132,132
4528          fcb 130,118
4529          fcb 120,94
4530          fcb 90,110
4531          fcb 132,132
4532          fcb 72,106
4533          fcb $ff
4534          fcb 64,102
4535          fcb $fc
4536          fcb $1f,$bd,$f1,$53,$00
4537          fcb 66,102
4538          fcb $fc
4539          fcb $1e,$32,$11,$73,$00
4540          fcb 88,112
```

4541		fcB 72,120
4542		fcB \$ff
4543		fcB 62,132
4544		fcB 20,128
4545		fcB 52,122
4546		fcB 64,122
4547		fcB 60,124
4548		fcB 114,128
4549		fcB 80,130
4550		fcB 68,130
4551		fcB 62,132
4552		fcB \$ff
4553		fcB 40,130
4554		fcB \$fc
4555		fcB \$ff,\$1e,\$11,\$f2,\$3f,\$20,\$0f,\$c0
4556		fcB \$ff,\$31,\$00
4557		fcB \$fe
4558	LDFCA	fcB 132,130
4559		fcB 112,122
4560		fcB 92,124
4561		fcB 94,126
4562		fcB 94,130
4563		fcB 92,132
4564		fcB 112,130
4565		fcB 128,140
4566		fcB 132,136
4567		fcB 132,114
4568		fcB 120,108
4569		fcB 106,118
4570		fcB 120,112
4571		fcB 124,116
4572		fcB 124,126
4573		fcB \$ff
4574		fcB 100,120
4575		fcB \$fc
4576		fcB \$e0,\$e2,\$ee,\$e0,\$f1,\$22,\$ee,\$06
4577		fcB \$2e,\$e2,\$11,\$20,\$2e,\$22,\$20,\$00
4578		fcB \$fe
4579		fcc 'KSK'
